NAME

perl - Practical Extraction and Report Language 



SYNOPSIS

perl [ -sTuU ] [ -hv ] [ -V[:configvar] ] [ -cw ] [ -d[:debugger] ] [ -D[number/list] ] [ -pna ] [ -Fpattern ] [ -l[octal] ] [ -0[octal] ] [ -Idir ] [ -m[-]module ] [ -M[-]'module...' ] [ -P ] [ -S ] [ -x[dir] ] [ -i[extension] ] [ -e 'command' ] [ -- ] [ programfile ] [ argument ]... 

For ease of access, the Perl manual has been split up into a number of sections: 

    perl                Perl overview (this section)

    perldelta           Perl changes since previous version

    perl5.004delta      Perl changes between 5.003 and 5.004

    perlfaq             Perl frequently asked questions

    perldata            Perl data structures

    perlsyn             Perl syntax

    perlop              Perl operators and precedence

    perlre              Perl regular expressions

    perlrun             Perl execution and options

    perlfunc            Perl builtin functions

    perlvar             Perl predefined variables

    perlsub             Perl subroutines

    perlmod             Perl modules: how they work

    perlmodlib          Perl modules: how to write and use

    perlmodinstall      Perl modules: how to install from CPAN

    perlform            Perl formats

    perllocale          Perl locale support

    perlref             Perl references

    perldsc             Perl data structures intro

    perllol             Perl data structures: lists of lists

    perltoot            Perl OO tutorial

    perlobj             Perl objects

    perltie             Perl objects hidden behind simple variables

    perlbot             Perl OO tricks and examples

    perlipc             Perl interprocess communication

    perldebug           Perl debugging

    perldiag            Perl diagnostic messages

    perlsec             Perl security

    perltrap            Perl traps for the unwary

    perlport            Perl portability guide

    perlstyle           Perl style guide

    perlpod             Perl plain old documentation

    perlbook            Perl book information

    perlembed           Perl ways to embed perl in your C or C++ application

    perlapio            Perl internal IO abstraction interface

    perlxs              Perl XS application programming interface

    perlxstut           Perl XS tutorial

    perlguts            Perl internal functions for those doing extensions

    perlcall            Perl calling conventions from C

    perlhist            Perl history records

(If you're intending to read these straight through for the first time, the suggested order will tend to reduce the number of forward references.) 

By default, all of the above manpages are installed in the /usr/local/man/ directory. 

Extensive additional documentation for Perl modules is available. The default configuration for perl will place this additional documentation in the /usr/local/lib/perl5/man directory (or else in the man subdirectory of the Perl library directory). Some of this additional documentation is distributed standard with Perl, but you'll also find documentation for third-party modules there. 

You should be able to view Perl's documentation with your man(1) program by including the proper directories in the appropriate start-up files, or in the MANPATH environment variable. To find out where the configuration has installed the manpages, type: 

    perl -V:man.dir

If the directories have a common stem, such as /usr/local/man/man1 and /usr/local/man/man3, you need only to add that stem (/usr/local/man) to your man(1) configuration files or your MANPATH environment variable. If they do not share a stem, you'll have to add both stems. 

If that doesn't work for some reason, you can still use the supplied perldoc script to view module information. You might also look into getting a replacement man program. 

If something strange has gone wrong with your program and you're not sure where you should look for help, try the -w switch first. It will often point out exactly where the trouble is. 



DESCRIPTION

Perl is a language optimized for scanning arbitrary text files, extracting information from those text files, and printing reports based on that information. It's also a good language for many system management tasks. The language is intended to be practical (easy to use, efficient, complete) rather than beautiful (tiny, elegant, minimal). 

Perl combines (in the author's opinion, anyway) some of the best features of C, sed, awk, and sh, so people familiar with those languages should have little difficulty with it. (Language historians will also note some vestiges of csh, Pascal, and even BASIC-PLUS.) Expression syntax corresponds quite closely to C expression syntax. Unlike most Unix utilities, Perl does not arbitrarily limit the size of your data--if you've got the memory, Perl can slurp in your whole file as a single string. Recursion is of unlimited depth. And the tables used by hashes (previously called ``associative arrays'') grow as necessary to prevent degraded performance. Perl uses sophisticated pattern matching techniques to scan large amounts of data very quickly. Although optimized for scanning text, Perl can also deal with binary data, and can make dbm files look like hashes. Setuid Perl scripts are safer than C programs through a dataflow tracing mechanism which prevents many stupid security holes. 

If you have a problem that would ordinarily use sed or awk or sh, but it exceeds their capabilities or must run a little faster, and you don't want to write the silly thing in C, then Perl may be for you. There are also translators to turn your sed and awk scripts into Perl scripts. 

But wait, there's more... 

Perl version 5 is nearly a complete rewrite, and provides the following additional benefits: 

· Many usability enhancements 

It is now possible to write much more readable Perl code (even within regular expressions). Formerly cryptic variable names can be replaced by mnemonic identifiers. Error messages are more informative, and the optional warnings will catch many of the mistakes a novice might make. This cannot be stressed enough. Whenever you get mysterious behavior, try the -w switch!!! Whenever you don't get mysterious behavior, try using -w anyway. 

· Simplified grammar The new yacc grammar is one half the size of the old one. Many of the arbitrary grammar rules have been regularized. The number of reserved words has been cut by 2/3. Despite this, nearly all old Perl scripts will continue to work unchanged. 

· Lexical scoping Perl variables may now be declared within a lexical scope, like ``auto'' variables in C. Not only is this more efficient, but it contributes to better privacy for ``programming in the large''. Anonymous subroutines exhibit deep binding of lexical variables (closures). 

· Arbitrarily nested data structures Any scalar value, including any array element, may now contain a reference to any other variable or subroutine. You can easily create anonymous variables and subroutines. Perl manages your reference counts for you. 

· Modularity and reusability The Perl library is now defined in terms of modules which can be easily shared among various packages. A package may choose to import all or a portion of a module's published interface. Pragmas (that is, compiler directives) are defined and used by the same mechanism. 

· Object-oriented programming A package can function as a class. Dynamic multiple inheritance and virtual methods are supported in a straightforward manner and with very little new syntax. Filehandles may now be treated as objects. 

· Embeddable and Extensible Perl may now be embedded easily in your C or C++ application, and can either call or be called by your routines through a documented interface. The XS preprocessor is provided to make it easy to glue your C or C++ routines into Perl. Dynamic loading of modules is supported, and Perl itself can be made into a dynamic library. 

· POSIX compliant A major new module is the POSIX module, which provides access to all available POSIX routines and definitions, via object classes where appropriate. 

· Package constructors and destructors The new BEGIN and END blocks provide means to capture control as a package is being compiled, and after the program exits. As a degenerate case they work just like awk's BEGIN and END when you use the -p or -n switches. 

· Multiple simultaneous DBM implementations A Perl program may now access DBM, NDBM, SDBM, GDBM, and Berkeley DB files from the same script simultaneously. In fact, the old dbmopen interface has been generalized to allow any variable to be tied to an object class which defines its access methods. 

· Subroutine definitions may now be autoloaded In fact, the AUTOLOAD mechanism also allows you to define any arbitrary semantics for undefined subroutine calls. It's not for just autoloading. 

· Regular expression enhancements You can now specify nongreedy quantifiers. You can now do grouping without creating a backreference. You can now write regular expressions with embedded whitespace and comments for readability. A consistent extensibility mechanism has been added that is upwardly compatible with all old regular expressions. 

· Innumerable Unbundled Modules The Comprehensive Perl Archive Network described in the perlmodlib manpage contains hundreds of plug-and-play modules full of reusable code. See http://www.perl.com/CPAN for a site near you. 

· Compilability While not yet in full production mode, a working perl-to-C compiler does exist. It can generate portable byte code, simple C, or optimized C code. 

Okay, that's definitely enough hype. 



ENVIRONMENT

See the perlrun manpage. 



AUTHOR

Larry Wall <larry@wall.org>, with the help of oodles of other folks. 

If your Perl success stories and testimonials may be of help to others who wish to advocate the use of Perl in their applications, or if you wish to simply express your gratitude to Larry and the Perl developers, please write to <perl-thanks@perl.org>. 



FILES

 "/tmp/perl-e$$"        temporary file for -e commands

 "@INC"                 locations of perl libraries



SEE ALSO

 a2p    awk to perl translator

 s2p    sed to perl translator



DIAGNOSTICS

The -w switch produces some lovely diagnostics. 

See the perldiag manpage for explanations of all Perl's diagnostics. The use diagnostics pragma automatically turns Perl's normally terse warnings and errors into these longer forms. 

Compilation errors will tell you the line number of the error, with an indication of the next token or token type that was to be examined. (In the case of a script passed to Perl via -e switches, each -e is counted as one line.) 

Setuid scripts have additional constraints that can produce error messages such as ``Insecure dependency''. See the perlsec manpage. 

Did we mention that you should definitely consider using the -w switch? 



BUGS

The -w switch is not mandatory. 

Perl is at the mercy of your machine's definitions of various operations such as type casting, atof(), and floating-point output with sprintf(). 

If your stdio requires a seek or eof between reads and writes on a particular stream, so does Perl. (This doesn't apply to sysread() and syswrite().) 

While none of the built-in data types have any arbitrary size limits (apart from memory size), there are still a few arbitrary limits: a given variable name may not be longer than 255 characters, and no component of your PATH may be longer than 255 if you use -S. A regular expression may not compile to more than 32767 bytes internally. 

You may mail your bug reports (be sure to include full configuration information as output by the myconfig program in the perl source tree, or by perl -V) to <perlbug@perl.com>. If you've succeeded in compiling perl, the perlbug script in the utils/ subdirectory can be used to help mail in a bug report. 

Perl actually stands for Pathologically Eclectic Rubbish Lister, but don't tell anyone I said that. 



NOTES

The Perl motto is ``There's more than one way to do it.'' Divining how many more is left as an exercise to the reader. 

The three principal virtues of a programmer are Laziness, Impatience, and Hubris. See the Camel Book for why. 



DISCLAIMER

We are painfully aware that these documents may contain incorrect links and misformatted HTML. Such bugs lie in the automatic translation process that automatically created the hundreds and hundreds of separate documents that you find here. Please do not report link or formatting bugs, because we cannot fix per-document problems. The only bug reports that will help us are those that supply working patches to the installhtml or pod2html programs, or to the Pod::HTML module itself, for which I and the entire Perl community will shower you with thanks and praises. 

If rather than formatting bugs, you encounter substantive content errors in these documents, such as mistakes in the explanations or code, please use the perlbug utility included with the Perl distribution. 

--Tom Christiansen, Perl Documentation Compiler and Editor

NAME

perlsyn - Perl syntax 



DESCRIPTION

A Perl script consists of a sequence of declarations and statements. The only things that need to be declared in Perl are report formats and subroutines. See the sections below for more information on those declarations. All uninitialized user-created objects are assumed to start with a null or 0 value until they are defined by some explicit operation such as assignment. (Though you can get warnings about the use of undefined values if you like.) The sequence of statements is executed just once, unlike in sed and awk scripts, where the sequence of statements is executed for each input line. While this means that you must explicitly loop over the lines of your input file (or files), it also means you have much more control over which files and which lines you look at. (Actually, I'm lying--it is possible to do an implicit loop with either the -n or -p switch. It's just not the mandatory default like it is in sed and awk.) 



Declarations

Perl is, for the most part, a free-form language. (The only exception to this is format declarations, for obvious reasons.) Comments are indicated by the "#" character, and extend to the end of the line. If you attempt to use /* */ C-style comments, it will be interpreted either as division or pattern matching, depending on the context, and C++ // comments just look like a null regular expression, so don't do that. 

A declaration can be put anywhere a statement can, but has no effect on the execution of the primary sequence of statements--declarations all take effect at compile time. Typically all the declarations are put at the beginning or the end of the script. However, if you're using lexically-scoped private variables created with my(), you'll have to make sure your format or subroutine definition is within the same block scope as the my if you expect to be able to access those private variables. 

Declaring a subroutine allows a subroutine name to be used as if it were a list operator from that point forward in the program. You can declare a subroutine without defining it by saying sub name, thus: 

    sub myname;

    $me = myname $0             or die "can't get myname";

Note that it functions as a list operator, not as a unary operator; so be careful to use or instead of || in this case. However, if you were to declare the subroutine as sub myname ($), then myname would function as a unary operator, so either or or || would work. 

Subroutines declarations can also be loaded up with the require statement or both loaded and imported into your namespace with a use statement. See the perlmod manpage for details on this. 

A statement sequence may contain declarations of lexically-scoped variables, but apart from declaring a variable name, the declaration acts like an ordinary statement, and is elaborated within the sequence of statements as if it were an ordinary statement. That means it actually has both compile-time and run-time effects. 



Simple statements

The only kind of simple statement is an expression evaluated for its side effects. Every simple statement must be terminated with a semicolon, unless it is the final statement in a block, in which case the semicolon is optional. (A semicolon is still encouraged there if the block takes up more than one line, because you may eventually add another line.) Note that there are some operators like eval {} and do {} that look like compound statements, but aren't (they're just TERMs in an expression), and thus need an explicit termination if used as the last item in a statement. 

Any simple statement may optionally be followed by a SINGLE modifier, just before the terminating semicolon (or block ending). The possible modifiers are: 

    if EXPR

    unless EXPR

    while EXPR

    until EXPR

    foreach EXPR

The if and unless modifiers have the expected semantics, presuming you're a speaker of English. The foreach modifier is an iterator: For each value in EXPR, it aliases $_ to the value and executes the statement. The while and until modifiers have the usual ``while loop'' semantics (conditional evaluated first), except when applied to a do -BLOCK (or to the now-deprecated do -SUBROUTINE statement), in which case the block executes once before the conditional is evaluated. This is so that you can write loops like: 

    do {

        $line = <STDIN>;

        ...

    } until $line  eq ".\n";

See do. Note also that the loop control statements described later will NOT work in this construct, because modifiers don't take loop labels. Sorry. You can always put another block inside of it (for next) or around it (for last) to do that sort of thing. For next, just double the braces: 

    do {{

        next if $x == $y;

        # do something here

    }} until $x++ > $z;

For last, you have to be more elaborate: 

    LOOP: { 

            do {

                last if $x = $y**2;

                # do something here

            } while $x++ <= $z;

    }



Compound statements

In Perl, a sequence of statements that defines a scope is called a block. Sometimes a block is delimited by the file containing it (in the case of a required file, or the program as a whole), and sometimes a block is delimited by the extent of a string (in the case of an eval). 

But generally, a block is delimited by curly brackets, also known as braces. We will call this syntactic construct a BLOCK. 

The following compound statements may be used to control flow: 

    if (EXPR) BLOCK

    if (EXPR) BLOCK else BLOCK

    if (EXPR) BLOCK elsif (EXPR) BLOCK ... else BLOCK

    LABEL while (EXPR) BLOCK

    LABEL while (EXPR) BLOCK continue BLOCK

    LABEL for (EXPR; EXPR; EXPR) BLOCK

    LABEL foreach VAR (LIST) BLOCK

    LABEL BLOCK continue BLOCK

Note that, unlike C and Pascal, these are defined in terms of BLOCKs, not statements. This means that the curly brackets are required--no dangling statements allowed. If you want to write conditionals without curly brackets there are several other ways to do it. The following all do the same thing: 

    if (!open(FOO)) { die "Can't open $FOO: $!"; }

    die "Can't open $FOO: $!" unless open(FOO);

    open(FOO) or die "Can't open $FOO: $!";     # FOO or bust!

    open(FOO) ? 'hi mom' : die "Can't open $FOO: $!";

                        # a bit exotic, that last one

The if statement is straightforward. Because BLOCKs are always bounded by curly brackets, there is never any ambiguity about which if an else goes with. If you use unless in place of if, the sense of the test is reversed. 

The while statement executes the block as long as the expression is true (does not evaluate to the null string ("") or 0 or "0"). The LABEL is optional, and if present, consists of an identifier followed by a colon. The LABEL identifies the loop for the loop control statements next, last, and redo. If the LABEL is omitted, the loop control statement refers to the innermost enclosing loop. This may include dynamically looking back your call-stack at run time to find the LABEL. Such desperate behavior triggers a warning if you use the -w flag. 

If there is a continue BLOCK, it is always executed just before the conditional is about to be evaluated again, just like the third part of a for loop in C. Thus it can be used to increment a loop variable, even when the loop has been continued via the next statement (which is similar to the C continue statement). 



Loop Control

The next command is like the continue statement in C; it starts the next iteration of the loop: 

    LINE: while (<STDIN>) {

        next LINE if /^#/;      # discard comments

        ...

    }

The last command is like the break statement in C (as used in loops); it immediately exits the loop in question. The continue block, if any, is not executed: 

    LINE: while (<STDIN>) {

        last LINE if /^$/;      # exit when done with header

        ...

    }

The redo command restarts the loop block without evaluating the conditional again. The continue block, if any, is not executed. This command is normally used by programs that want to lie to themselves about what was just input. 

For example, when processing a file like /etc/termcap. If your input lines might end in backslashes to indicate continuation, you want to skip ahead and get the next record. 

    while (<>) {

        chomp;

        if (s/\\$//) {

            $_ .= <>;

            redo unless eof();

        }

        # now process $_

    }

which is Perl short-hand for the more explicitly written version: 

    LINE: while (defined($line = <ARGV>)) {

        chomp($line);

        if ($line =~ s/\\$//) {

            $line .= <ARGV>;

            redo LINE unless eof(); # not eof(ARGV)!

        }

        # now process $line

    }

Note that if there were a continue block on the above code, it would get executed even on discarded lines. This is often used to reset line counters or ?pat? one-time matches. 

    # inspired by :1,$g/fred/s//WILMA/

    while (<>) {

        ?(fred)?    && s//WILMA $1 WILMA/;

        ?(barney)?  && s//BETTY $1 BETTY/;

        ?(homer)?   && s//MARGE $1 MARGE/;

    } continue {

        print "$ARGV $.: $_";

        close ARGV  if eof();           # reset $.

        reset       if eof();           # reset ?pat?

    }

If the word while is replaced by the word until, the sense of the test is reversed, but the conditional is still tested before the first iteration. 

The loop control statements don't work in an if or unless, since they aren't loops. You can double the braces to make them such, though. 

    if (/pattern/) {{

        next if /fred/;

        next if /barney/;

        # so something here

    }}

The form while/if BLOCK BLOCK, available in Perl 4, is no longer available. Replace any occurrence of if BLOCK by if (do BLOCK). 



For Loops

Perl's C-style for loop works exactly like the corresponding while loop; that means that this: 

    for ($i = 1; $i < 10; $i++) {

        ...

    }

is the same as this: 

    $i = 1;

    while ($i < 10) {

        ...

    } continue {

        $i++;

    }

(There is one minor difference: The first form implies a lexical scope for variables declared with my in the initialization expression.) 

Besides the normal array index looping, for can lend itself to many other interesting applications. Here's one that avoids the problem you get into if you explicitly test for end-of-file on an interactive file descriptor causing your program to appear to hang. 

    $on_a_tty = -t STDIN && -t STDOUT;

    sub prompt { print "yes? " if $on_a_tty }

    for ( prompt(); <STDIN>; prompt() ) {

        # do something

    }



Foreach Loops

The foreach loop iterates over a normal list value and sets the variable VAR to be each element of the list in turn. If the variable is preceded with the keyword my, then it is lexically scoped, and is therefore visible only within the loop. Otherwise, the variable is implicitly local to the loop and regains its former value upon exiting the loop. If the variable was previously declared with my, it uses that variable instead of the global one, but it's still localized to the loop. (Note that a lexically scoped variable can cause problems if you have subroutine or format declarations within the loop which refer to it.) 

The foreach keyword is actually a synonym for the for keyword, so you can use foreach for readability or for for brevity. (Or because the Bourne shell is more familiar to you than csh, so writing for comes more naturally.) If VAR is omitted, $_ is set to each value. If any element of LIST is an lvalue, you can modify it by modifying VAR inside the loop. That's because the foreach loop index variable is an implicit alias for each item in the list that you're looping over. 

If any part of LIST is an array, foreach will get very confused if you add or remove elements within the loop body, for example with splice. So don't do that. 

foreach probably won't do what you expect if VAR is a tied or other special variable. Don't do that either. 

Examples: 

    for (@ary) { s/foo/bar/ }

    foreach my $elem (@elements) {

        $elem *= 2;

    }

    for $count (10,9,8,7,6,5,4,3,2,1,'BOOM') {

        print $count, "\n"; sleep(1);

    }

    for (1..15) { print "Merry Christmas\n"; }

    foreach $item (split(/:[\\\n:]*/, $ENV{TERMCAP})) {

        print "Item: $item\n";

    }

Here's how a C programmer might code up a particular algorithm in Perl: 

    for (my $i = 0; $i < @ary1; $i++) {

        for (my $j = 0; $j < @ary2; $j++) {

            if ($ary1[$i] > $ary2[$j]) {

                last; # can't go to outer :-(

            }

            $ary1[$i] += $ary2[$j];

        }

        # this is where that last takes me

    }

Whereas here's how a Perl programmer more comfortable with the idiom might do it: 

    OUTER: foreach my $wid (@ary1) {

    INNER:   foreach my $jet (@ary2) {

                next OUTER if $wid > $jet;

                $wid += $jet;

             }

          }

See how much easier this is? It's cleaner, safer, and faster. It's cleaner because it's less noisy. It's safer because if code gets added between the inner and outer loops later on, the new code won't be accidentally executed. The next explicitly iterates the other loop rather than merely terminating the inner one. And it's faster because Perl executes a foreach statement more rapidly than it would the equivalent for loop. 



Basic BLOCKs and Switch Statements

A BLOCK by itself (labeled or not) is semantically equivalent to a loop that executes once. Thus you can use any of the loop control statements in it to leave or restart the block. (Note that this is NOT true in eval{}, sub{}, or contrary to popular belief do{} blocks, which do NOT count as loops.) The continue block is optional. 

The BLOCK construct is particularly nice for doing case structures. 

    SWITCH: {

        if (/^abc/) { $abc = 1; last SWITCH; }

        if (/^def/) { $def = 1; last SWITCH; }

        if (/^xyz/) { $xyz = 1; last SWITCH; }

        $nothing = 1;

    }

There is no official switch statement in Perl, because there are already several ways to write the equivalent. In addition to the above, you could write 

    SWITCH: {

        $abc = 1, last SWITCH  if /^abc/;

        $def = 1, last SWITCH  if /^def/;

        $xyz = 1, last SWITCH  if /^xyz/;

        $nothing = 1;

    }

(That's actually not as strange as it looks once you realize that you can use loop control ``operators'' within an expression, That's just the normal C comma operator.) 

or 

    SWITCH: {

        /^abc/ && do { $abc = 1; last SWITCH; };

        /^def/ && do { $def = 1; last SWITCH; };

        /^xyz/ && do { $xyz = 1; last SWITCH; };

        $nothing = 1;

    }

or formatted so it stands out more as a ``proper'' switch statement: 

    SWITCH: {

        /^abc/      && do {

                            $abc = 1;

                            last SWITCH;

                       };

        /^def/      && do {

                            $def = 1;

                            last SWITCH;

                       };

        /^xyz/      && do {

                            $xyz = 1;

                            last SWITCH;

                        };

        $nothing = 1;

    }

or 

    SWITCH: {

        /^abc/ and $abc = 1, last SWITCH;

        /^def/ and $def = 1, last SWITCH;

        /^xyz/ and $xyz = 1, last SWITCH;

        $nothing = 1;

    }

or even, horrors, 

    if (/^abc/)

        { $abc = 1 }

    elsif (/^def/)

        { $def = 1 }

    elsif (/^xyz/)

        { $xyz = 1 }

    else

        { $nothing = 1 }

A common idiom for a switch statement is to use foreach's aliasing to make a temporary assignment to $_ for convenient matching: 

    SWITCH: for ($where) {

                /In Card Names/     && do { push @flags, '-e'; last; };

                /Anywhere/          && do { push @flags, '-h'; last; };

                /In Rulings/        && do {                    last; };

                die "unknown value for form variable where: `$where'";

            }

Another interesting approach to a switch statement is arrange for a do block to return the proper value: 

    $amode = do {

        if     ($flag & O_RDONLY) { "r" }       # XXX: isn't this 0?

        elsif  ($flag & O_WRONLY) { ($flag & O_APPEND) ? "a" : "w" }

        elsif  ($flag & O_RDWR)   {

            if ($flag & O_CREAT)  { "w+" }

            else                  { ($flag & O_APPEND) ? "a+" : "r+" }

        }

    };

Or 

        print do {

            ($flags & O_WRONLY) ? "write-only"          :

            ($flags & O_RDWR)   ? "read-write"          :

                                  "read-only";

        };

Or if you are certainly that all the && clauses are true, you can use something like this, which ``switches'' on the value of the HTTP_USER_AGENT envariable. 

    #!/usr/bin/perl 

    # pick out jargon file page based on browser

    $dir = 'http://www.wins.uva.nl/~mes/jargon';

    for ($ENV{HTTP_USER_AGENT}) { 

        $page  =    /Mac/            && 'm/Macintrash.html'

                 || /Win(dows )?NT/  && 'e/evilandrude.html'

                 || /Win|MSIE|WebTV/ && 'm/MicroslothWindows.html'

                 || /Linux/          && 'l/Linux.html'

                 || /HP-UX/          && 'h/HP-SUX.html'

                 || /SunOS/          && 's/ScumOS.html'

                 ||                     'a/AppendixB.html';

    }

    print "Location: $dir/$page\015\012\015\012";

That kind of switch statement only works when you know the && clauses will be true. If you don't, the previous ?: example should be used. 

You might also consider writing a hash instead of synthesizing a switch statement. 



Goto

Although not for the faint of heart, Perl does support a goto statement. A loop's LABEL is not actually a valid target for a goto; it's just the name of the loop. There are three forms: goto -LABEL, goto -EXPR, and goto -&NAME. 

The goto -LABEL form finds the statement labeled with LABEL and resumes execution there. It may not be used to go into any construct that requires initialization, such as a subroutine or a foreach loop. It also can't be used to go into a construct that is optimized away. It can be used to go almost anywhere else within the dynamic scope, including out of subroutines, but it's usually better to use some other construct such as last or die. The author of Perl has never felt the need to use this form of goto (in Perl, that is--C is another matter). 

The goto -EXPR form expects a label name, whose scope will be resolved dynamically. This allows for computed gotos per FORTRAN, but isn't necessarily recommended if you're optimizing for maintainability: 

    goto ("FOO", "BAR", "GLARCH")[$i];

The goto -&NAME form is highly magical, and substitutes a call to the named subroutine for the currently running subroutine. This is used by AUTOLOAD() subroutines that wish to load another subroutine and then pretend that the other subroutine had been called in the first place (except that any modifications to @_ in the current subroutine are propagated to the other subroutine.) After the goto, not even caller() will be able to tell that this routine was called first. 

In almost all cases like this, it's usually a far, far better idea to use the structured control flow mechanisms of next, last, or redo instead of resorting to a goto. For certain applications, the catch and throw pair of eval{} and die() for exception processing can also be a prudent approach. 



PODs: Embedded Documentation

Perl has a mechanism for intermixing documentation with source code. While it's expecting the beginning of a new statement, if the compiler encounters a line that begins with an equal sign and a word, like this 

    =head1 Here There Be Pods!

Then that text and all remaining text up through and including a line beginning with =cut will be ignored. The format of the intervening text is described in the perlpod manpage. 

This allows you to intermix your source code and your documentation text freely, as in 

    =item snazzle($)

    The snazzle() function will behave in the most spectacular

    form that you can possibly imagine, not even excepting

    cybernetic pyrotechnics.

    =cut back to the compiler, nuff of this pod stuff!

    sub snazzle($) {

        my $thingie = shift;

        .........

    }

Note that pod translators should look at only paragraphs beginning with a pod directive (it makes parsing easier), whereas the compiler actually knows to look for pod escapes even in the middle of a paragraph. This means that the following secret stuff will be ignored by both the compiler and the translators. 

    $a=3;

    =secret stuff

     warn "Neither POD nor CODE!?"

    =cut back

    print "got $a\n";

You probably shouldn't rely upon the warn() being podded out forever. Not all pod translators are well-behaved in this regard, and perhaps the compiler will become pickier. 

One may also use pod directives to quickly comment out a section of code. 



Plain Old Comments (Not!)

Much like the C preprocessor, Perl can process line directives. Using this, one can control Perl's idea of filenames and line numbers in error or warning messages (especially for strings that are processed with eval()). The syntax for this mechanism is the same as for most C preprocessors: it matches the regular expression /^#\s*line\s+(\d+)\s*(?:\s"([^"]*)")?/ with $1 being the line number for the next line, and $2 being the optional filename (specified within quotes). 

Here are some examples that you should be able to type into your command shell: 

    % perl

    # line 200 "bzzzt"

    # the `#' on the previous line must be the first char on line

    die 'foo';

    __END__

    foo at bzzzt line 201.

    % perl

    # line 200 "bzzzt"

    eval qq[\n#line 2001 ""\ndie 'foo']; print $@;

    __END__

    foo at - line 2001.

    % perl

    eval qq[\n#line 200 "foo bar"\ndie 'foo']; print $@;

    __END__

    foo at foo bar line 200.

    % perl

    # line 345 "goop"

    eval "\n#line " . __LINE__ . ' "' . __FILE__ ."\"\ndie 'foo'";

    print $@;

    __END__

    foo at goop line 345.

NAME

perldata - Perl data types 



DESCRIPTION



Variable names

Perl has three data structures: scalars, arrays of scalars, and associative arrays of scalars, known as ``hashes''. Normal arrays are indexed by number, starting with 0. (Negative subscripts count from the end.) Hash arrays are indexed by string. 

Values are usually referred to by name (or through a named reference). The first character of the name tells you to what sort of data structure it refers. The rest of the name tells you the particular value to which it refers. Most often, it consists of a single identifier, that is, a string beginning with a letter or underscore, and containing letters, underscores, and digits. In some cases, it may be a chain of identifiers, separated by :: (or by ', but that's deprecated); all but the last are interpreted as names of packages, to locate the namespace in which to look up the final identifier (see Packages for details). It's possible to substitute for a simple identifier an expression that produces a reference to the value at runtime; this is described in more detail below, and in the perlref manpage. 

There are also special variables whose names don't follow these rules, so that they don't accidentally collide with one of your normal variables. Strings that match parenthesized parts of a regular expression are saved under names containing only digits after the $ (see the perlop manpage and the perlre manpage). In addition, several special variables that provide windows into the inner working of Perl have names containing punctuation characters (see the perlvar manpage). 

Scalar values are always named with '$', even when referring to a scalar that is part of an array. It works like the English word ``the''. Thus we have: 

    $days               # the simple scalar value "days"

    $days[28]           # the 29th element of array @days

    $days{'Feb'}        # the 'Feb' value from hash %days

    $#days              # the last index of array @days

but entire arrays or array slices are denoted by '@', which works much like the word ``these'' or ``those'': 

    @days               # ($days[0], $days[1],... $days[n])

    @days[3,4,5]        # same as @days[3..5]

    @days{'a','c'}      # same as ($days{'a'},$days{'c'})

and entire hashes are denoted by '%': 

    %days               # (key1, val1, key2, val2 ...)

In addition, subroutines are named with an initial '&', though this is optional when it's otherwise unambiguous (just as ``do'' is often redundant in English). Symbol table entries can be named with an initial '*', but you don't really care about that yet. 

Every variable type has its own namespace. You can, without fear of conflict, use the same name for a scalar variable, an array, or a hash (or, for that matter, a filehandle, a subroutine name, or a label). This means that $foo and @foo are two different variables. It also means that $foo[1] is a part of @foo, not a part of $foo. This may seem a bit weird, but that's okay, because it is weird. 

Because variable and array references always start with '$', '@', or '%', the ``reserved'' words aren't in fact reserved with respect to variable names. (They ARE reserved with respect to labels and filehandles, however, which don't have an initial special character. You can't have a filehandle named ``log'', for instance. Hint: you could say open(LOG,'logfile') rather than open(log,'logfile'). Using uppercase filehandles also improves readability and protects you from conflict with future reserved words.) Case IS significant--``FOO'', ``Foo'', and ``foo'' are all different names. Names that start with a letter or underscore may also contain digits and underscores. 

It is possible to replace such an alphanumeric name with an expression that returns a reference to an object of that type. For a description of this, see the perlref manpage. 

Names that start with a digit may contain only more digits. Names that do not start with a letter, underscore, or digit are limited to one character, e.g., $% or $$. (Most of these one character names have a predefined significance to Perl. For instance, $$ is the current process id.) 



Context

The interpretation of operations and values in Perl sometimes depends on the requirements of the context around the operation or value. There are two major contexts: scalar and list. Certain operations return list values in contexts wanting a list, and scalar values otherwise. (If this is true of an operation it will be mentioned in the documentation for that operation.) In other words, Perl overloads certain operations based on whether the expected return value is singular or plural. (Some words in English work this way, like ``fish'' and ``sheep''.) 

In a reciprocal fashion, an operation provides either a scalar or a list context to each of its arguments. For example, if you say 

    int( <STDIN> )

the integer operation provides a scalar context for the <STDIN> operator, which responds by reading one line from STDIN and passing it back to the integer operation, which will then find the integer value of that line and return that. If, on the other hand, you say 

    sort( <STDIN> )

then the sort operation provides a list context for <STDIN>, which will proceed to read every line available up to the end of file, and pass that list of lines back to the sort routine, which will then sort those lines and return them as a list to whatever the context of the sort was. 

Assignment is a little bit special in that it uses its left argument to determine the context for the right argument. Assignment to a scalar evaluates the righthand side in a scalar context, while assignment to an array or array slice evaluates the righthand side in a list context. Assignment to a list also evaluates the righthand side in a list context. 

User defined subroutines may choose to care whether they are being called in a scalar or list context, but most subroutines do not need to care, because scalars are automatically interpolated into lists. See wantarray. 



Scalar values

All data in Perl is a scalar or an array of scalars or a hash of scalars. Scalar variables may contain various kinds of singular data, such as numbers, strings, and references. In general, conversion from one form to another is transparent. (A scalar may not contain multiple values, but may contain a reference to an array or hash containing multiple values.) Because of the automatic conversion of scalars, operations, and functions that return scalars don't need to care (and, in fact, can't care) whether the context is looking for a string or a number. 

Scalars aren't necessarily one thing or another. There's no place to declare a scalar variable to be of type ``string'', or of type ``number'', or type ``filehandle'', or anything else. Perl is a contextually polymorphic language whose scalars can be strings, numbers, or references (which includes objects). While strings and numbers are considered pretty much the same thing for nearly all purposes, references are strongly-typed uncastable pointers with builtin reference-counting and destructor invocation. 

A scalar value is interpreted as TRUE in the Boolean sense if it is not the null string or the number 0 (or its string equivalent, ``0''). The Boolean context is just a special kind of scalar context. 

There are actually two varieties of null scalars: defined and undefined. Undefined null scalars are returned when there is no real value for something, such as when there was an error, or at end of file, or when you refer to an uninitialized variable or element of an array. An undefined null scalar may become defined the first time you use it as if it were defined, but prior to that you can use the defined() operator to determine whether the value is defined or not. 

To find out whether a given string is a valid nonzero number, it's usually enough to test it against both numeric 0 and also lexical ``0'' (although this will cause -w noises). That's because strings that aren't numbers count as 0, just as they do in awk: 

    if ($str == 0 && $str ne "0")  {

        warn "That doesn't look like a number";

    }

That's usually preferable because otherwise you won't treat IEEE notations like NaN or Infinity properly. At other times you might prefer to use the POSIX::strtod function or a regular expression to check whether data is numeric. See the perlre manpage for details on regular expressions. 

    warn "has nondigits"        if     /\D/;

    warn "not a natural number" unless /^\d+$/;             # rejects -3

    warn "not an integer"       unless /^-?\d+$/;           # rejects +3

    warn "not an integer"       unless /^[+-]?\d+$/;

    warn "not a decimal number" unless /^-?\d+\.?\d*$/;     # rejects .2

    warn "not a decimal number" unless /^-?(?:\d+(?:\.\d*)?|\.\d+)$/;

    warn "not a C float"

        unless /^([+-]?)(?=\d|\.\d)\d*(\.\d*)?([Ee]([+-]?\d+))?$/;

The length of an array is a scalar value. You may find the length of array @days by evaluating $#days, as in csh. (Actually, it's not the length of the array, it's the subscript of the last element, because there is (ordinarily) a 0th element.) Assigning to $#days changes the length of the array. Shortening an array by this method destroys intervening values. Lengthening an array that was previously shortened NO LONGER recovers the values that were in those elements. (It used to in Perl 4, but we had to break this to make sure destructors were called when expected.) You can also gain some miniscule measure of efficiency by pre-extending an array that is going to get big. (You can also extend an array by assigning to an element that is off the end of the array.) You can truncate an array down to nothing by assigning the null list () to it. The following are equivalent: 

    @whatever = ();

    $#whatever = -1;

If you evaluate a named array in a scalar context, it returns the length of the array. (Note that this is not true of lists, which return the last value, like the C comma operator, nor of built-in functions, which return whatever they feel like returning.) The following is always true: 

    scalar(@whatever) == $#whatever - $[ + 1;

Version 5 of Perl changed the semantics of $[: files that don't set the value of $[ no longer need to worry about whether another file changed its value. (In other words, use of $[ is deprecated.) So in general you can assume that 

    scalar(@whatever) == $#whatever + 1;

Some programmers choose to use an explicit conversion so nothing's left to doubt: 

    $element_count = scalar(@whatever);

If you evaluate a hash in a scalar context, it returns a value that is true if and only if the hash contains any key/value pairs. (If there are any key/value pairs, the value returned is a string consisting of the number of used buckets and the number of allocated buckets, separated by a slash. This is pretty much useful only to find out whether Perl's (compiled in) hashing algorithm is performing poorly on your data set. For example, you stick 10,000 things in a hash, but evaluating %HASH in scalar context reveals ``1/16'', which means only one out of sixteen buckets has been touched, and presumably contains all 10,000 of your items. This isn't supposed to happen.) 

You can preallocate space for a hash by assigning to the keys() function. This rounds up the allocated bucked to the next power of two: 

    keys(%users) = 1000;                # allocate 1024 buckets



Scalar value constructors

Numeric literals are specified in any of the customary floating point or integer formats: 

    12345

    12345.67

    .23E-10

    0xffff              # hex

    0377                # octal

    4_294_967_296       # underline for legibility

String literals are usually delimited by either single or double quotes. They work much like shell quotes: double-quoted string literals are subject to backslash and variable substitution; single-quoted strings are not (except for ``\''' and ``\\''). The usual Unix backslash rules apply for making characters such as newline, tab, etc., as well as some more exotic forms. See Quote and Quotelike Operators for a list. 

Octal or hex representations in string literals (e.g. '0xffff') are not automatically converted to their integer representation. The hex() and oct() functions make these conversions for you. See hex and oct for more details. 

You can also embed newlines directly in your strings, i.e., they can end on a different line than they begin. This is nice, but if you forget your trailing quote, the error will not be reported until Perl finds another line containing the quote character, which may be much further on in the script. Variable substitution inside strings is limited to scalar variables, arrays, and array slices. (In other words, names beginning with $ or @, followed by an optional bracketed expression as a subscript.) The following code segment prints out ``The price is $100.'' 

    $Price = '$100';    # not interpreted

    print "The price is $Price.\n";     # interpreted

As in some shells, you can put curly brackets around the name to delimit it from following alphanumerics. In fact, an identifier within such curlies is forced to be a string, as is any single identifier within a hash subscript. Our earlier example, 

    $days{'Feb'}

can be written as 

    $days{Feb}

and the quotes will be assumed automatically. But anything more complicated in the subscript will be interpreted as an expression. 

Note that a single-quoted string must be separated from a preceding word by a space, because single quote is a valid (though deprecated) character in a variable name (see Packages). 

Three special literals are __FILE__, __LINE__, and __PACKAGE__, which represent the current filename, line number, and package name at that point in your program. They may be used only as separate tokens; they will not be interpolated into strings. If there is no current package (due to an empty package; directive), __PACKAGE__ is the undefined value. 

The tokens __END__ and __DATA__ may be used to indicate the logical end of the script before the actual end of file. Any following text is ignored, but may be read via a DATA filehandle: main::DATA for __END__, or PACKNAME::DATA (where PACKNAME is the current package) for __DATA__. The two control characters ^D and ^Z are synonyms for __END__ (or __DATA__ in a module). See the SelfLoader manpage for more description of __DATA__, and an example of its use. Note that you cannot read from the DATA filehandle in a BEGIN block: the BEGIN block is executed as soon as it is seen (during compilation), at which point the corresponding __DATA__ (or __END__) token has not yet been seen. 

A word that has no other interpretation in the grammar will be treated as if it were a quoted string. These are known as ``barewords''. As with filehandles and labels, a bareword that consists entirely of lowercase letters risks conflict with future reserved words, and if you use the -w switch, Perl will warn you about any such words. Some people may wish to outlaw barewords entirely. If you say 

    use strict 'subs';

then any bareword that would NOT be interpreted as a subroutine call produces a compile-time error instead. The restriction lasts to the end of the enclosing block. An inner block may countermand this by saying no strict 'subs'. 

Array variables are interpolated into double-quoted strings by joining all the elements of the array with the delimiter specified in the $" variable ($LIST_SEPARATOR in English), space by default. The following are equivalent: 

    $temp = join($",@ARGV);

    system "echo $temp";

    system "echo @ARGV";

Within search patterns (which also undergo double-quotish substitution) there is a bad ambiguity: Is /$foo[bar]/ to be interpreted as /${foo}[bar]/ (where [bar] is a character class for the regular expression) or as /${foo[bar]}/ (where [bar] is the subscript to array @foo)? If @foo doesn't otherwise exist, then it's obviously a character class. If @foo exists, Perl takes a good guess about [bar], and is almost always right. If it does guess wrong, or if you're just plain paranoid, you can force the correct interpretation with curly brackets as above. 

A line-oriented form of quoting is based on the shell ``here-doc'' syntax. Following a << you specify a string to terminate the quoted material, and all lines following the current line down to the terminating string are the value of the item. The terminating string may be either an identifier (a word), or some quoted text. If quoted, the type of quotes you use determines the treatment of the text, just as in regular quoting. An unquoted identifier works like double quotes. There must be no space between the << and the identifier. (If you put a space it will be treated as a null identifier, which is valid, and matches the first empty line.) The terminating string must appear by itself (unquoted and with no surrounding whitespace) on the terminating line. 

        print <<EOF;

    The price is $Price.

    EOF

        print <<"EOF";  # same as above

    The price is $Price.

    EOF

        print <<`EOC`;  # execute commands

    echo hi there

    echo lo there

    EOC

        print <<"foo", <<"bar"; # you can stack them

    I said foo.

    foo

    I said bar.

    bar

        myfunc(<<"THIS", 23, <<'THAT');

    Here's a line

    or two.

    THIS

    and here's another.

    THAT

Just don't forget that you have to put a semicolon on the end to finish the statement, as Perl doesn't know you're not going to try to do this: 

        print <<ABC

    179231

    ABC

        + 20;



List value constructors

List values are denoted by separating individual values by commas (and enclosing the list in parentheses where precedence requires it): 

    (LIST)

In a context not requiring a list value, the value of the list literal is the value of the final element, as with the C comma operator. For example, 

    @foo = ('cc', '-E', $bar);

assigns the entire list value to array foo, but 

    $foo = ('cc', '-E', $bar);

assigns the value of variable bar to variable foo. Note that the value of an actual array in a scalar context is the length of the array; the following assigns the value 3 to $foo: 

    @foo = ('cc', '-E', $bar);

    $foo = @foo;                # $foo gets 3

You may have an optional comma before the closing parenthesis of a list literal, so that you can say: 

    @foo = (

        1,

        2,

        3,

    );

LISTs do automatic interpolation of sublists. That is, when a LIST is evaluated, each element of the list is evaluated in a list context, and the resulting list value is interpolated into LIST just as if each individual element were a member of LIST. Thus arrays and hashes lose their identity in a LIST--the list 

    (@foo,@bar,&SomeSub,%glarch)

contains all the elements of @foo followed by all the elements of @bar, followed by all the elements returned by the subroutine named SomeSub called in a list context, followed by the key/value pairs of %glarch. To make a list reference that does NOT interpolate, see the perlref manpage. 

The null list is represented by (). Interpolating it in a list has no effect. Thus ((),(),()) is equivalent to (). Similarly, interpolating an array with no elements is the same as if no array had been interpolated at that point. 

A list value may also be subscripted like a normal array. You must put the list in parentheses to avoid ambiguity. For example: 

    # Stat returns list value.

    $time = (stat($file))[8];

    # SYNTAX ERROR HERE.

    $time = stat($file)[8];  # OOPS, FORGOT PARENTHESES

    # Find a hex digit.

    $hexdigit = ('a','b','c','d','e','f')[$digit-10];

    # A "reverse comma operator".

    return (pop(@foo),pop(@foo))[0];

You may assign to undef in a list. This is useful for throwing away some of the return values of a function: 

    ($dev, $ino, undef, undef, $uid, $gid) = stat($file);

Lists may be assigned to if and only if each element of the list is legal to assign to: 

    ($a, $b, $c) = (1, 2, 3);

    ($map{'red'}, $map{'blue'}, $map{'green'}) = (0x00f, 0x0f0, 0xf00);

Array assignment in a scalar context returns the number of elements produced by the expression on the right side of the assignment: 

    $x = (($foo,$bar) = (3,2,1));       # set $x to 3, not 2

    $x = (($foo,$bar) = f());           # set $x to f()'s return count

This is very handy when you want to do a list assignment in a Boolean context, because most list functions return a null list when finished, which when assigned produces a 0, which is interpreted as FALSE. 

The final element may be an array or a hash: 

    ($a, $b, @rest) = split;

    my($a, $b, %rest) = @_;

You can actually put an array or hash anywhere in the list, but the first one in the list will soak up all the values, and anything after it will get a null value. This may be useful in a local() or my(). 

A hash literal contains pairs of values to be interpreted as a key and a value: 

    # same as map assignment above

    %map = ('red',0x00f,'blue',0x0f0,'green',0xf00);

While literal lists and named arrays are usually interchangeable, that's not the case for hashes. Just because you can subscript a list value like a normal array does not mean that you can subscript a list value as a hash. Likewise, hashes included as parts of other lists (including parameters lists and return lists from functions) always flatten out into key/value pairs. That's why it's good to use references sometimes. 

It is often more readable to use the => operator between key/value pairs. The => operator is mostly just a more visually distinctive synonym for a comma, but it also arranges for its left-hand operand to be interpreted as a string--if it's a bareword that would be a legal identifier. This makes it nice for initializing hashes: 

    %map = (

                 red   => 0x00f,

                 blue  => 0x0f0,

                 green => 0xf00,

   );

or for initializing hash references to be used as records: 

    $rec = {

                witch => 'Mable the Merciless',

                cat   => 'Fluffy the Ferocious',

                date  => '10/31/1776',

    };

or for using call-by-named-parameter to complicated functions: 

   $field = $query->radio_group(

               name      => 'group_name',

               values    => ['eenie','meenie','minie'],

               default   => 'meenie',

               linebreak => 'true',

               labels    => \%labels

   );

Note that just because a hash is initialized in that order doesn't mean that it comes out in that order. See sort for examples of how to arrange for an output ordering. 



Typeglobs and Filehandles

Perl uses an internal type called a typeglob to hold an entire symbol table entry. The type prefix of a typeglob is a *, because it represents all types. This used to be the preferred way to pass arrays and hashes by reference into a function, but now that we have real references, this is seldom needed. 

The main use of typeglobs in modern Perl is create symbol table aliases. This assignment: 

    *this = *that;

makes $this an alias for $that, @this an alias for @that, %this an alias for %that, &this an alias for &that, etc. Much safer is to use a reference. This: 

    local *Here::blue = \$There::green;

temporarily makes $Here::blue an alias for $There::green, but doesn't make @Here::blue an alias for @There::green, or %Here::blue an alias for %There::green, etc. See Symbol Tables for more examples of this. Strange though this may seem, this is the basis for the whole module import/export system. 

Another use for typeglobs is to to pass filehandles into a function or to create new filehandles. If you need to use a typeglob to save away a filehandle, do it this way: 

    $fh = *STDOUT;

or perhaps as a real reference, like this: 

    $fh = \*STDOUT;

See the perlsub manpage for examples of using these as indirect filehandles in functions. 

Typeglobs are also a way to create a local filehandle using the local() operator. These last until their block is exited, but may be passed back. For example: 

    sub newopen {

        my $path = shift;

        local *FH;  # not my!

        open   (FH, $path)          or  return undef;

        return *FH;

    }

    $fh = newopen('/etc/passwd');

Now that we have the *foo{THING} notation, typeglobs aren't used as much for filehandle manipulations, although they're still needed to pass brand new file and directory handles into or out of functions. That's because *HANDLE{IO} only works if HANDLE has already been used as a handle. In other words, *FH can be used to create new symbol table entries, but *foo{THING} cannot. 

Another way to create anonymous filehandles is with the IO::Handle module and its ilk. These modules have the advantage of not hiding different types of the same name during the local(). See the bottom of open() for an example. 

See the perlref manpage, the perlsub manpage, and Symbol Tables for more discussion on typeglobs and the *foo{THING} syntax. 

NAME

perlsub - Perl subroutines 



SYNOPSIS

To declare subroutines: 

    sub NAME;             # A "forward" declaration.

    sub NAME(PROTO);      #  ditto, but with prototypes

    sub NAME BLOCK        # A declaration and a definition.

    sub NAME(PROTO) BLOCK #  ditto, but with prototypes

To define an anonymous subroutine at runtime: 

    $subref = sub BLOCK;            # no proto

    $subref = sub (PROTO) BLOCK;    # with proto

To import subroutines: 

    use PACKAGE qw(NAME1 NAME2 NAME3);

To call subroutines: 

    NAME(LIST);    # & is optional with parentheses.

    NAME LIST;     # Parentheses optional if predeclared/imported.

    &NAME;         # Makes current @_ visible to called subroutine.



DESCRIPTION

Like many languages, Perl provides for user-defined subroutines. These may be located anywhere in the main program, loaded in from other files via the do, require, or use keywords, or even generated on the fly using eval or anonymous subroutines (closures). You can even call a function indirectly using a variable containing its name or a CODE reference to it. 

The Perl model for function call and return values is simple: all functions are passed as parameters one single flat list of scalars, and all functions likewise return to their caller one single flat list of scalars. Any arrays or hashes in these call and return lists will collapse, losing their identities--but you may always use pass-by-reference instead to avoid this. Both call and return lists may contain as many or as few scalar elements as you'd like. (Often a function without an explicit return statement is called a subroutine, but there's really no difference from the language's perspective.) 

Any arguments passed to the routine come in as the array @_. Thus if you called a function with two arguments, those would be stored in $_[0] and $_[1]. The array @_ is a local array, but its elements are aliases for the actual scalar parameters. In particular, if an element $_[0] is updated, the corresponding argument is updated (or an error occurs if it is not updatable). If an argument is an array or hash element which did not exist when the function was called, that element is created only when (and if) it is modified or if a reference to it is taken. (Some earlier versions of Perl created the element whether or not it was assigned to.) Note that assigning to the whole array @_ removes the aliasing, and does not update any arguments. 

The return value of the subroutine is the value of the last expression evaluated. Alternatively, a return statement may be used to exit the subroutine, optionally specifying the returned value, which will be evaluated in the appropriate context (list, scalar, or void) depending on the context of the subroutine call. If you specify no return value, the subroutine will return an empty list in a list context, an undefined value in a scalar context, or nothing in a void context. If you return one or more arrays and/or hashes, these will be flattened together into one large indistinguishable list. 

Perl does not have named formal parameters, but in practice all you do is assign to a my() list of these. Any variables you use in the function that aren't declared private are global variables. For the gory details on creating private variables, see Private Variables via my() and Temporary Values via local(). To create protected environments for a set of functions in a separate package (and probably a separate file), see Packages. 

Example: 

    sub max {

        my $max = shift(@_);

        foreach $foo (@_) {

            $max = $foo if $max < $foo;

        }

        return $max;

    }

    $bestday = max($mon,$tue,$wed,$thu,$fri);

Example: 

    # get a line, combining continuation lines

    #  that start with whitespace

    sub get_line {

        $thisline = $lookahead;  # GLOBAL VARIABLES!!

        LINE: while (defined($lookahead = <STDIN>)) {

            if ($lookahead =~ /^[ \t]/) {

                $thisline .= $lookahead;

            }

            else {

                last LINE;

            }

        }

        $thisline;

    }

    $lookahead = <STDIN>;       # get first line

    while ($_ = get_line()) {

        ...

    }

Use array assignment to a local list to name your formal arguments: 

    sub maybeset {

        my($key, $value) = @_;

        $Foo{$key} = $value unless $Foo{$key};

    }

This also has the effect of turning call-by-reference into call-by-value, because the assignment copies the values. Otherwise a function is free to do in-place modifications of @_ and change its caller's values. 

    upcase_in($v1, $v2);  # this changes $v1 and $v2

    sub upcase_in {

        for (@_) { tr/a-z/A-Z/ }

    }

You aren't allowed to modify constants in this way, of course. If an argument were actually literal and you tried to change it, you'd take a (presumably fatal) exception. For example, this won't work: 

    upcase_in("frederick");

It would be much safer if the upcase_in() function were written to return a copy of its parameters instead of changing them in place: 

    ($v3, $v4) = upcase($v1, $v2);  # this doesn't

    sub upcase {

        return unless defined wantarray;  # void context, do nothing

        my @parms = @_;

        for (@parms) { tr/a-z/A-Z/ }

        return wantarray ? @parms : $parms[0];

    }

Notice how this (unprototyped) function doesn't care whether it was passed real scalars or arrays. Perl will see everything as one big long flat @_ parameter list. This is one of the ways where Perl's simple argument-passing style shines. The upcase() function would work perfectly well without changing the upcase() definition even if we fed it things like this: 

    @newlist   = upcase(@list1, @list2);

    @newlist   = upcase( split /:/, $var );

Do not, however, be tempted to do this: 

    (@a, @b)   = upcase(@list1, @list2);

Because like its flat incoming parameter list, the return list is also flat. So all you have managed to do here is stored everything in @a and made @b an empty list. See Pass by Reference for alternatives. 

A subroutine may be called using the ``&'' prefix. The ``&'' is optional in modern Perls, and so are the parentheses if the subroutine has been predeclared. (Note, however, that the ``&'' is NOT optional when you're just naming the subroutine, such as when it's used as an argument to defined() or undef(). Nor is it optional when you want to do an indirect subroutine call with a subroutine name or reference using the &$subref() or &{$subref}() constructs. See the perlref manpage for more on that.) 

Subroutines may be called recursively. If a subroutine is called using the ``&'' form, the argument list is optional, and if omitted, no @_ array is set up for the subroutine: the @_ array at the time of the call is visible to subroutine instead. This is an efficiency mechanism that new users may wish to avoid. 

    &foo(1,2,3);        # pass three arguments

    foo(1,2,3);         # the same

    foo();              # pass a null list

    &foo();             # the same

    &foo;               # foo() get current args, like foo(@_) !!

    foo;                # like foo() IFF sub foo predeclared, else "foo"

Not only does the ``&'' form make the argument list optional, but it also disables any prototype checking on the arguments you do provide. This is partly for historical reasons, and partly for having a convenient way to cheat if you know what you're doing. See the section on Prototypes below. 

Function whose names are in all upper case are reserved to the Perl core, just as are modules whose names are in all lower case. A function in all capitals is a loosely-held convention meaning it will be called indirectly by the run-time system itself. Functions that do special, pre-defined things are BEGIN, END, AUTOLOAD, and DESTROY--plus all the functions mentioned in the perltie manpage. The 5.005 release adds INIT to this list. 



Private Variables via my()
Synopsis: 

    my $foo;            # declare $foo lexically local

    my (@wid, %get);    # declare list of variables local

    my $foo = "flurp";  # declare $foo lexical, and init it

    my @oof = @bar;     # declare @oof lexical, and init it

A ``my'' declares the listed variables to be confined (lexically) to the enclosing block, conditional (if/unless/elsif/else), loop (for/foreach/while/until/continue), subroutine, eval, or do/require/use'd file. If more than one value is listed, the list must be placed in parentheses. All listed elements must be legal lvalues. Only alphanumeric identifiers may be lexically scoped--magical builtins like $/ must currently be localize with ``local'' instead. 

Unlike dynamic variables created by the ``local'' operator, lexical variables declared with ``my'' are totally hidden from the outside world, including any called subroutines (even if it's the same subroutine called from itself or elsewhere--every call gets its own copy). 

This doesn't mean that a my() variable declared in a statically enclosing lexical scope would be invisible. Only the dynamic scopes are cut off. For example, the bumpx() function below has access to the lexical $x variable because both the my and the sub occurred at the same scope, presumably the file scope. 

    my $x = 10;

    sub bumpx { $x++ } 

(An eval(), however, can see the lexical variables of the scope it is being evaluated in so long as the names aren't hidden by declarations within the eval() itself. See the perlref manpage.) 

The parameter list to my() may be assigned to if desired, which allows you to initialize your variables. (If no initializer is given for a particular variable, it is created with the undefined value.) Commonly this is used to name the parameters to a subroutine. Examples: 

    $arg = "fred";        # "global" variable

    $n = cube_root(27);

    print "$arg thinks the root is $n\n";

 fred thinks the root is 3

    sub cube_root {

        my $arg = shift;  # name doesn't matter

        $arg **= 1/3;

        return $arg;

    }

The ``my'' is simply a modifier on something you might assign to. So when you do assign to the variables in its argument list, the ``my'' doesn't change whether those variables are viewed as a scalar or an array. So 

    my ($foo) = <STDIN>;                # WRONG?

    my @FOO = <STDIN>;

both supply a list context to the right-hand side, while 

    my $foo = <STDIN>;

supplies a scalar context. But the following declares only one variable: 

    my $foo, $bar = 1;                  # WRONG

That has the same effect as 

    my $foo;

    $bar = 1;

The declared variable is not introduced (is not visible) until after the current statement. Thus, 

    my $x = $x;

can be used to initialize the new $x with the value of the old $x, and the expression 

    my $x = 123 and $x == 123

is false unless the old $x happened to have the value 123. 

Lexical scopes of control structures are not bounded precisely by the braces that delimit their controlled blocks; control expressions are part of the scope, too. Thus in the loop 

    while (defined(my $line = <>)) {

        $line = lc $line;

    } continue {

        print $line;

    }

the scope of $line extends from its declaration throughout the rest of the loop construct (including the continue clause), but not beyond it. Similarly, in the conditional 

    if ((my $answer = <STDIN>) =~ /^yes$/i) {

        user_agrees();

    } elsif ($answer =~ /^no$/i) {

        user_disagrees();

    } else {

        chomp $answer;

        die "'$answer' is neither 'yes' nor 'no'";

    }

the scope of $answer extends from its declaration throughout the rest of the conditional (including elsif and else clauses, if any), but not beyond it. 

(None of the foregoing applies to if/unless or while/until modifiers appended to simple statements. Such modifiers are not control structures and have no effect on scoping.) 

The foreach loop defaults to scoping its index variable dynamically (in the manner of local; see below). However, if the index variable is prefixed with the keyword ``my'', then it is lexically scoped instead. Thus in the loop 

    for my $i (1, 2, 3) {

        some_function();

    }

the scope of $i extends to the end of the loop, but not beyond it, and so the value of $i is unavailable in some_function(). 

Some users may wish to encourage the use of lexically scoped variables. As an aid to catching implicit references to package variables, if you say 

    use strict 'vars';

then any variable reference from there to the end of the enclosing block must either refer to a lexical variable, or must be fully qualified with the package name. A compilation error results otherwise. An inner block may countermand this with "no strict 'vars'". 

A my() has both a compile-time and a run-time effect. At compile time, the compiler takes notice of it; the principle usefulness of this is to quiet "use strict 'vars'". The actual initialization is delayed until run time, so it gets executed appropriately; every time through a loop, for example. 

Variables declared with ``my'' are not part of any package and are therefore never fully qualified with the package name. In particular, you're not allowed to try to make a package variable (or other global) lexical: 

    my $pack::var;      # ERROR!  Illegal syntax

    my $_;              # also illegal (currently)

In fact, a dynamic variable (also known as package or global variables) are still accessible using the fully qualified :: notation even while a lexical of the same name is also visible: 

    package main;

    local $x = 10;

    my    $x = 20;

    print "$x and $::x\n";

That will print out 20 and 10. 

You may declare ``my'' variables at the outermost scope of a file to hide any such identifiers totally from the outside world. This is similar to C's static variables at the file level. To do this with a subroutine requires the use of a closure (anonymous function with lexical access). If a block (such as an eval(), function, or package) wants to create a private subroutine that cannot be called from outside that block, it can declare a lexical variable containing an anonymous sub reference: 

    my $secret_version = '1.001-beta';

    my $secret_sub = sub { print $secret_version };

    &$secret_sub();

As long as the reference is never returned by any function within the module, no outside module can see the subroutine, because its name is not in any package's symbol table. Remember that it's not REALLY called $some_pack::secret_version or anything; it's just $secret_version, unqualified and unqualifiable. 

This does not work with object methods, however; all object methods have to be in the symbol table of some package to be found. 



Peristent Private Variables

Just because a lexical variable is lexically (also called statically) scoped to its enclosing block, eval, or do FILE, this doesn't mean that within a function it works like a C static. It normally works more like a C auto, but with implicit garbage collection. 

Unlike local variables in C or C++, Perl's lexical variables don't necessarily get recycled just because their scope has exited. If something more permanent is still aware of the lexical, it will stick around. So long as something else references a lexical, that lexical won't be freed--which is as it should be. You wouldn't want memory being free until you were done using it, or kept around once you were done. Automatic garbage collection takes care of this for you. 

This means that you can pass back or save away references to lexical variables, whereas to return a pointer to a C auto is a grave error. It also gives us a way to simulate C's function statics. Here's a mechanism for giving a function private variables with both lexical scoping and a static lifetime. If you do want to create something like C's static variables, just enclose the whole function in an extra block, and put the static variable outside the function but in the block. 

    {

        my $secret_val = 0;

        sub gimme_another {

            return ++$secret_val;

        }

    }

    # $secret_val now becomes unreachable by the outside

    # world, but retains its value between calls to gimme_another

If this function is being sourced in from a separate file via require or use, then this is probably just fine. If it's all in the main program, you'll need to arrange for the my() to be executed early, either by putting the whole block above your main program, or more likely, placing merely a BEGIN sub around it to make sure it gets executed before your program starts to run: 

    sub BEGIN {

        my $secret_val = 0;

        sub gimme_another {

            return ++$secret_val;

        }

    }

See Package Constructors and Destructors about the BEGIN function. 

If declared at the outermost scope, the file scope, then lexicals work someone like C's file statics. They are available to all functions in that same file declared below them, but are inaccessible from outside of the file. This is sometimes used in modules to create private variables for the whole module. 



Temporary Values via local()

NOTE: In general, you should be using ``my'' instead of ``local'', because it's faster and safer. Exceptions to this include the global punctuation variables, filehandles and formats, and direct manipulation of the Perl symbol table itself. Format variables often use ``local'' though, as do other variables whose current value must be visible to called subroutines. 

Synopsis: 

    local $foo;                 # declare $foo dynamically local

    local (@wid, %get);         # declare list of variables local

    local $foo = "flurp";       # declare $foo dynamic, and init it

    local @oof = @bar;          # declare @oof dynamic, and init it

    local *FH;                  # localize $FH, @FH, %FH, &FH  ...

    local *merlyn = *randal;    # now $merlyn is really $randal, plus

                                #     @merlyn is really @randal, etc

    local *merlyn = 'randal';   # SAME THING: promote 'randal' to *randal

    local *merlyn = \$randal;   # just alias $merlyn, not @merlyn etc

A local() modifies its listed variables to be ``local'' to the enclosing block, eval, or do FILE--and to any subroutine called from within that block. A local() just gives temporary values to global (meaning package) variables. It does not create a local variable. This is known as dynamic scoping. Lexical scoping is done with ``my'', which works more like C's auto declarations. 

If more than one variable is given to local(), they must be placed in parentheses. All listed elements must be legal lvalues. This operator works by saving the current values of those variables in its argument list on a hidden stack and restoring them upon exiting the block, subroutine, or eval. This means that called subroutines can also reference the local variable, but not the global one. The argument list may be assigned to if desired, which allows you to initialize your local variables. (If no initializer is given for a particular variable, it is created with an undefined value.) Commonly this is used to name the parameters to a subroutine. Examples: 

    for $i ( 0 .. 9 ) {

        $digits{$i} = $i;

    }

    # assume this function uses global %digits hash

    parse_num();

    # now temporarily add to %digits hash

    if ($base12) {

        # (NOTE: not claiming this is efficient!)

        local %digits  = (%digits, 't' => 10, 'e' => 11);

        parse_num();  # parse_num gets this new %digits!

    }

    # old %digits restored here

Because local() is a run-time command, it gets executed every time through a loop. In releases of Perl previous to 5.0, this used more stack storage each time until the loop was exited. Perl now reclaims the space each time through, but it's still more efficient to declare your variables outside the loop. 

A local is simply a modifier on an lvalue expression. When you assign to a localized variable, the local doesn't change whether its list is viewed as a scalar or an array. So 

    local($foo) = <STDIN>;

    local @FOO = <STDIN>;

both supply a list context to the right-hand side, while 

    local $foo = <STDIN>;

supplies a scalar context. 

A note about local() and composite types is in order. Something like local(%foo) works by temporarily placing a brand new hash in the symbol table. The old hash is left alone, but is hidden ``behind'' the new one. 

This means the old variable is completely invisible via the symbol table (i.e. the hash entry in the *foo typeglob) for the duration of the dynamic scope within which the local() was seen. This has the effect of allowing one to temporarily occlude any magic on composite types. For instance, this will briefly alter a tied hash to some other implementation: 

    tie %ahash, 'APackage';

    [...]

    {

       local %ahash;

       tie %ahash, 'BPackage';

       [..called code will see %ahash tied to 'BPackage'..]

       {

          local %ahash;

          [..%ahash is a normal (untied) hash here..]

       }

    }

    [..%ahash back to its initial tied self again..]

As another example, a custom implementation of %ENV might look like this: 

    {

        local %ENV;

        tie %ENV, 'MyOwnEnv';

        [..do your own fancy %ENV manipulation here..]

    }

    [..normal %ENV behavior here..]

It's also worth taking a moment to explain what happens when you localize a member of a composite type (i.e. an array or hash element). In this case, the element is localized by name. This means that when the scope of the local() ends, the saved value will be restored to the hash element whose key was named in the local(), or the array element whose index was named in the local(). If that element was deleted while the local() was in effect (e.g. by a delete() from a hash or a shift() of an array), it will spring back into existence, possibly extending an array and filling in the skipped elements with undef. For instance, if you say 

    %hash = ( 'This' => 'is', 'a' => 'test' );

    @ary  = ( 0..5 );

    {

         local($ary[5]) = 6;

         local($hash{'a'}) = 'drill';

         while (my $e = pop(@ary)) {

             print "$e . . .\n";

             last unless $e > 3;

         }

         if (@ary) {

             $hash{'only a'} = 'test';

             delete $hash{'a'};

         }

    }

    print join(' ', map { "$_ $hash{$_}" } sort keys %hash),".\n";

    print "The array has ",scalar(@ary)," elements: ",

          join(', ', map { defined $_ ? $_ : 'undef' } @ary),"\n";

Perl will print 

    6 . . .

    4 . . .

    3 . . .

    This is a test only a test.

    The array has 6 elements: 0, 1, 2, undef, undef, 5



Passing Symbol Table Entries (typeglobs)

[Note: The mechanism described in this section was originally the only way to simulate pass-by-reference in older versions of Perl. While it still works fine in modern versions, the new reference mechanism is generally easier to work with. See below.] 

Sometimes you don't want to pass the value of an array to a subroutine but rather the name of it, so that the subroutine can modify the global copy of it rather than working with a local copy. In perl you can refer to all objects of a particular name by prefixing the name with a star: *foo. This is often known as a ``typeglob'', because the star on the front can be thought of as a wildcard match for all the funny prefix characters on variables and subroutines and such. 

When evaluated, the typeglob produces a scalar value that represents all the objects of that name, including any filehandle, format, or subroutine. When assigned to, it causes the name mentioned to refer to whatever ``*'' value was assigned to it. Example: 

    sub doubleary {

        local(*someary) = @_;

        foreach $elem (@someary) {

            $elem *= 2;

        }

    }

    doubleary(*foo);

    doubleary(*bar);

Note that scalars are already passed by reference, so you can modify scalar arguments without using this mechanism by referring explicitly to $_[0] etc. You can modify all the elements of an array by passing all the elements as scalars, but you have to use the * mechanism (or the equivalent reference mechanism) to push, pop, or change the size of an array. It will certainly be faster to pass the typeglob (or reference). 

Even if you don't want to modify an array, this mechanism is useful for passing multiple arrays in a single LIST, because normally the LIST mechanism will merge all the array values so that you can't extract out the individual arrays. For more on typeglobs, see Typeglobs and Filehandles. 



When to Still Use local()

Despite the existence of my(), there are still three places where the local() operator still shines. In fact, in these three places, you must use local instead of my. 

1. You need to give a global variable a temporary value, especially $_. 

The global variables, like @ARGV or the punctuation variables, must be localized with local(). This block reads in /etc/motd, and splits it up into chunks separated by lines of equal signs, which are placed in @Fields. 

    {

        local @ARGV = ("/etc/motd");

        local $/ = undef;

        local $_ = <>;  

        @Fields = split /^\s*=+\s*$/;

    } 

It particular, it's important to localize $_ in any routine that assigns to it. Look out for implicit assignments in while conditionals. 

2. You need to create a local file or directory handle or a local function. A function that needs a filehandle of its own must use local() uses local() on complete typeglob. This can be used to create new symbol table entries: 

3.     sub ioqueue {

4.         local  (*READER, *WRITER);    # not my!

5.         pipe    (READER,  WRITER);    or die "pipe: $!";

6.         return (*READER, *WRITER);

7.     }

8.     ($head, $tail) = ioqueue();

See the Symbol module for a way to create anonymous symbol table entries. 

Because assignment of a reference to a typeglob creates an alias, this can be used to create what is effectively a local function, or at least, a local alias. 

    {

        local *grow = \&shrink; # only until this block exists

        grow();                 # really calls shrink()

        move();                 # if move() grow()s, it shrink()s too

    }

    grow();                     # get the real grow() again

See Function Templates for more about manipulating functions by name in this way. 

9. You want to temporarily change just one element of an array or hash. You can localize just one element of an aggregate. Usually this is done on dynamics: 

10.     {

11.         local $SIG{INT} = 'IGNORE';

12.         funct();                            # uninterruptible

13.     } 

14.     # interruptibility automatically restored here

But it also works on lexically declared aggregates. Prior to 5.005, this operation could on occasion misbehave. 



Pass by Reference

If you want to pass more than one array or hash into a function--or return them from it--and have them maintain their integrity, then you're going to have to use an explicit pass-by-reference. Before you do that, you need to understand references as detailed in the perlref manpage. This section may not make much sense to you otherwise. 

Here are a few simple examples. First, let's pass in several arrays to a function and have it pop all of then, return a new list of all their former last elements: 

    @tailings = popmany ( \@a, \@b, \@c, \@d );

    sub popmany {

        my $aref;

        my @retlist = ();

        foreach $aref ( @_ ) {

            push @retlist, pop @$aref;

        }

        return @retlist;

    }

Here's how you might write a function that returns a list of keys occurring in all the hashes passed to it: 

    @common = inter( \%foo, \%bar, \%joe );

    sub inter {

        my ($k, $href, %seen); # locals

        foreach $href (@_) {

            while ( $k = each %$href ) {

                $seen{$k}++;

            }

        }

        return grep { $seen{$_} == @_ } keys %seen;

    }

So far, we're using just the normal list return mechanism. What happens if you want to pass or return a hash? Well, if you're using only one of them, or you don't mind them concatenating, then the normal calling convention is ok, although a little expensive. 

Where people get into trouble is here: 

    (@a, @b) = func(@c, @d);

or

    (%a, %b) = func(%c, %d);

That syntax simply won't work. It sets just @a or %a and clears the @b or %b. Plus the function didn't get passed into two separate arrays or hashes: it got one long list in @_, as always. 

If you can arrange for everyone to deal with this through references, it's cleaner code, although not so nice to look at. Here's a function that takes two array references as arguments, returning the two array elements in order of how many elements they have in them: 

    ($aref, $bref) = func(\@c, \@d);

    print "@$aref has more than @$bref\n";

    sub func {

        my ($cref, $dref) = @_;

        if (@$cref > @$dref) {

            return ($cref, $dref);

        } else {

            return ($dref, $cref);

        }

    }

It turns out that you can actually do this also: 

    (*a, *b) = func(\@c, \@d);

    print "@a has more than @b\n";

    sub func {

        local (*c, *d) = @_;

        if (@c > @d) {

            return (\@c, \@d);

        } else {

            return (\@d, \@c);

        }

    }

Here we're using the typeglobs to do symbol table aliasing. It's a tad subtle, though, and also won't work if you're using my() variables, because only globals (well, and local()s) are in the symbol table. 

If you're passing around filehandles, you could usually just use the bare typeglob, like *STDOUT, but typeglobs references would be better because they'll still work properly under use strict 'refs'. For example: 

    splutter(\*STDOUT);

    sub splutter {

        my $fh = shift;

        print $fh "her um well a hmmm\n";

    }

    $rec = get_rec(\*STDIN);

    sub get_rec {

        my $fh = shift;

        return scalar <$fh>;

    }

Another way to do this is using *HANDLE{IO}, see the perlref manpage for usage and caveats. 

If you're planning on generating new filehandles, you could do this: 

    sub openit {

        my $name = shift;

        local *FH;

        return open (FH, $path) ? *FH : undef;

    }

Although that will actually produce a small memory leak. See the bottom of open() for a somewhat cleaner way using the IO::Handle package. 



Prototypes

As of the 5.002 release of perl, if you declare 

    sub mypush (\@@)

then mypush() takes arguments exactly like push() does. The declaration of the function to be called must be visible at compile time. The prototype affects only the interpretation of new-style calls to the function, where new-style is defined as not using the & character. In other words, if you call it like a builtin function, then it behaves like a builtin function. If you call it like an old-fashioned subroutine, then it behaves like an old-fashioned subroutine. It naturally falls out from this rule that prototypes have no influence on subroutine references like \&foo or on indirect subroutine calls like &{$subref}. 

Method calls are not influenced by prototypes either, because the function to be called is indeterminate at compile time, because it depends on inheritance. 

Because the intent is primarily to let you define subroutines that work like builtin commands, here are the prototypes for some other functions that parse almost exactly like the corresponding builtins. 

    Declared as                 Called as

    sub mylink ($$)          mylink $old, $new

    sub myvec ($$$)          myvec $var, $offset, 1

    sub myindex ($$;$)       myindex &getstring, "substr"

    sub mysyswrite ($$$;$)   mysyswrite $buf, 0, length($buf) - $off, $off

    sub myreverse (@)        myreverse $a, $b, $c

    sub myjoin ($@)          myjoin ":", $a, $b, $c

    sub mypop (\@)           mypop @array

    sub mysplice (\@$$@)     mysplice @array, @array, 0, @pushme

    sub mykeys (\%)          mykeys %{$hashref}

    sub myopen (*;$)         myopen HANDLE, $name

    sub mypipe (**)          mypipe READHANDLE, WRITEHANDLE

    sub mygrep (&@)          mygrep { /foo/ } $a, $b, $c

    sub myrand ($)           myrand 42

    sub mytime ()            mytime

Any backslashed prototype character represents an actual argument that absolutely must start with that character. The value passed to the subroutine (as part of @_) will be a reference to the actual argument given in the subroutine call, obtained by applying \ to that argument. 

Unbackslashed prototype characters have special meanings. Any unbackslashed @ or % eats all the rest of the arguments, and forces list context. An argument represented by $ forces scalar context. An & requires an anonymous subroutine, which, if passed as the first argument, does not require the ``sub'' keyword or a subsequent comma. A * does whatever it has to do to turn the argument into a reference to a symbol table entry. 

A semicolon separates mandatory arguments from optional arguments. (It is redundant before @ or %.) 

Note how the last three examples above are treated specially by the parser. mygrep() is parsed as a true list operator, myrand() is parsed as a true unary operator with unary precedence the same as rand(), and mytime() is truly without arguments, just like time(). That is, if you say 

    mytime +2;

you'll get mytime() + 2, not mytime(2), which is how it would be parsed without the prototype. 

The interesting thing about & is that you can generate new syntax with it: 

    sub try (&@) {

        my($try,$catch) = @_;

        eval { &$try };

        if ($@) {

            local $_ = $@;

            &$catch;

        }

    }

    sub catch (&) { $_[0] }

    try {

        die "phooey";

    } catch {

        /phooey/ and print "unphooey\n";

    };

That prints "unphooey". (Yes, there are still unresolved issues having to do with the visibility of @_. I'm ignoring that question for the moment. (But note that if we make @_ lexically scoped, those anonymous subroutines can act like closures... (Gee, is this sounding a little Lispish? (Never mind.)))) 

And here's a reimplementation of grep: 

    sub mygrep (&@) {

        my $code = shift;

        my @result;

        foreach $_ (@_) {

            push(@result, $_) if &$code;

        }

        @result;

    }

Some folks would prefer full alphanumeric prototypes. Alphanumerics have been intentionally left out of prototypes for the express purpose of someday in the future adding named, formal parameters. The current mechanism's main goal is to let module writers provide better diagnostics for module users. Larry feels the notation quite understandable to Perl programmers, and that it will not intrude greatly upon the meat of the module, nor make it harder to read. The line noise is visually encapsulated into a small pill that's easy to swallow. 

It's probably best to prototype new functions, not retrofit prototyping into older ones. That's because you must be especially careful about silent impositions of differing list versus scalar contexts. For example, if you decide that a function should take just one parameter, like this: 

    sub func ($) {

        my $n = shift;

        print "you gave me $n\n";

    }

and someone has been calling it with an array or expression returning a list: 

    func(@foo);

    func( split /:/ );

Then you've just supplied an automatic scalar() in front of their argument, which can be more than a bit surprising. The old @foo which used to hold one thing doesn't get passed in. Instead, the func() now gets passed in 1, that is, the number of elements in @foo. And the split() gets called in a scalar context and starts scribbling on your @_ parameter list. 

This is all very powerful, of course, and should be used only in moderation to make the world a better place. 



Constant Functions

Functions with a prototype of () are potential candidates for inlining. If the result after optimization and constant folding is either a constant or a lexically-scoped scalar which has no other references, then it will be used in place of function calls made without & or do. Calls made using & or do are never inlined. (See constant.pm for an easy way to declare most constants.) 

The following functions would all be inlined: 

    sub pi ()           { 3.14159 }             # Not exact, but close.

    sub PI ()           { 4 * atan2 1, 1 }      # As good as it gets,

                                                # and it's inlined, too!

    sub ST_DEV ()       { 0 }

    sub ST_INO ()       { 1 }

    sub FLAG_FOO ()     { 1 << 8 }

    sub FLAG_BAR ()     { 1 << 9 }

    sub FLAG_MASK ()    { FLAG_FOO | FLAG_BAR }

    sub OPT_BAZ ()      { not (0x1B58 & FLAG_MASK) }

    sub BAZ_VAL () {

        if (OPT_BAZ) {

            return 23;

        }

        else {

            return 42;

        }

    }

    sub N () { int(BAZ_VAL) / 3 }

    BEGIN {

        my $prod = 1;

        for (1..N) { $prod *= $_ }

        sub N_FACTORIAL () { $prod }

    }

If you redefine a subroutine that was eligible for inlining, you'll get a mandatory warning. (You can use this warning to tell whether or not a particular subroutine is considered constant.) The warning is considered severe enough not to be optional because previously compiled invocations of the function will still be using the old value of the function. If you need to be able to redefine the subroutine you need to ensure that it isn't inlined, either by dropping the () prototype (which changes the calling semantics, so beware) or by thwarting the inlining mechanism in some other way, such as 

    sub not_inlined () {

        23 if $];

    }



Overriding Builtin Functions

Many builtin functions may be overridden, though this should be tried only occasionally and for good reason. Typically this might be done by a package attempting to emulate missing builtin functionality on a non-Unix system. 

Overriding may be done only by importing the name from a module--ordinary predeclaration isn't good enough. However, the subs pragma (compiler directive) lets you, in effect, predeclare subs via the import syntax, and these names may then override the builtin ones: 

    use subs 'chdir', 'chroot', 'chmod', 'chown';

    chdir $somewhere;

    sub chdir { ... }

To unambiguously refer to the builtin form, one may precede the builtin name with the special package qualifier CORE::. For example, saying CORE::open() will always refer to the builtin open(), even if the current package has imported some other subroutine called &open() from elsewhere. 

Library modules should not in general export builtin names like ``open'' or ``chdir'' as part of their default @EXPORT list, because these may sneak into someone else's namespace and change the semantics unexpectedly. Instead, if the module adds the name to the @EXPORT_OK list, then it's possible for a user to import the name explicitly, but not implicitly. That is, they could say 

    use Module 'open';

and it would import the open override, but if they said 

    use Module;

they would get the default imports without the overrides. 

The foregoing mechanism for overriding builtins is restricted, quite deliberately, to the package that requests the import. There is a second method that is sometimes applicable when you wish to override a builtin everywhere, without regard to namespace boundaries. This is achieved by importing a sub into the special namespace CORE::GLOBAL::. Here is an example that quite brazenly replaces the glob operator with something that understands regular expressions. 

    package REGlob;

    require Exporter;

    @ISA = 'Exporter';

    @EXPORT_OK = 'glob';

    sub import {

        my $pkg = shift;

        return unless @_;

        my $sym = shift;

        my $where = ($sym =~ s/^GLOBAL_// ? 'CORE::GLOBAL' : caller(0));

        $pkg->export($where, $sym, @_);

    }

    sub glob {

        my $pat = shift;

        my @got;

        local(*D);

        if (opendir D, '.') { @got = grep /$pat/, readdir D; closedir D; }

        @got;

    }

    1;

And here's how it could be (ab)used: 

    #use REGlob 'GLOBAL_glob';      # override glob() in ALL namespaces

    package Foo;

    use REGlob 'glob';              # override glob() in Foo:: only

    print for <^[a-z_]+\.pm\$>;     # show all pragmatic modules

Note that the initial comment shows a contrived, even dangerous example. By overriding glob globally, you would be forcing the new (and subversive) behavior for the glob operator for every namespace, without the complete cognizance or cooperation of the modules that own those namespaces. Naturally, this should be done with extreme caution--if it must be done at all. 

The REGlob example above does not implement all the support needed to cleanly override perl's glob operator. The builtin glob has different behaviors depending on whether it appears in a scalar or list context, but our REGlob doesn't. Indeed, many perl builtins have such context sensitive behaviors, and these must be adequately supported by a properly written override. For a fully functional example of overriding glob, study the implementation of File::DosGlob in the standard library. 



Autoloading

If you call a subroutine that is undefined, you would ordinarily get an immediate fatal error complaining that the subroutine doesn't exist. (Likewise for subroutines being used as methods, when the method doesn't exist in any base class of the class package.) If, however, there is an AUTOLOAD subroutine defined in the package or packages that were searched for the original subroutine, then that AUTOLOAD subroutine is called with the arguments that would have been passed to the original subroutine. The fully qualified name of the original subroutine magically appears in the $AUTOLOAD variable in the same package as the AUTOLOAD routine. The name is not passed as an ordinary argument because, er, well, just because, that's why... 

Most AUTOLOAD routines will load in a definition for the subroutine in question using eval, and then execute that subroutine using a special form of ``goto'' that erases the stack frame of the AUTOLOAD routine without a trace. (See the standard AutoLoader module, for example.) But an AUTOLOAD routine can also just emulate the routine and never define it. For example, let's pretend that a function that wasn't defined should just call system() with those arguments. All you'd do is this: 

    sub AUTOLOAD {

        my $program = $AUTOLOAD;

        $program =~ s/.*:://;

        system($program, @_);

    }

    date();

    who('am', 'i');

    ls('-l');

In fact, if you predeclare the functions you want to call that way, you don't even need the parentheses: 

    use subs qw(date who ls);

    date;

    who "am", "i";

    ls -l;

A more complete example of this is the standard Shell module, which can treat undefined subroutine calls as calls to Unix programs. 

Mechanisms are available for modules writers to help split the modules up into autoloadable files. See the standard AutoLoader module described in the AutoLoader manpage and in the AutoSplit manpage, the standard SelfLoader modules in the SelfLoader manpage, and the document on adding C functions to perl code in the perlxs manpage. 



SEE ALSO

See the perlref manpage for more about references and closures. See the perlxs manpage if you'd like to learn about calling C subroutines from perl. See the perlmod manpage to learn about bundling up your functions in separate files. 



DISCLAIMER

We are painfully aware that these documents may contain incorrect links and misformatted HTML. Such bugs lie in the automatic translation process that automatically created the hundreds and hundreds of separate documents that you find here. Please do not report link or formatting bugs, because we cannot fix per-document problems. The only bug reports that will help us are those that supply working patches to the installhtml or pod2html programs, or to the Pod::HTML module itself, for which I and the entire Perl community will shower you with thanks and praises. 

If rather than formatting bugs, you encounter substantive content errors in these documents, such as mistakes in the explanations or code, please use the perlbug utility included with the Perl distribution. 

--Tom Christiansen, Perl Documentation Compiler and Editor

NAME

perlop - Perl operators and precedence 



SYNOPSIS

Perl operators have the following associativity and precedence, listed from highest precedence to lowest. Note that all operators borrowed from C keep the same precedence relationship with each other, even where C's precedence is slightly screwy. (This makes learning Perl easier for C folks.) With very few exceptions, these all operate on scalar values only, not array values. 

    left        terms and list operators (leftward)

    left        ->

    nonassoc    ++ --

    right       **

    right       ! ~ \ and unary + and -

    left        =~ !~

    left        * / % x

    left        + - .

    left        << >>

    nonassoc    named unary operators

    nonassoc    < > <= >= lt gt le ge

    nonassoc    == != <=> eq ne cmp

    left        &

    left        | ^

    left        &&

    left        ||

    nonassoc    ..  ...

    right       ?:

    right       = += -= *= etc.

    left        , =>

    nonassoc    list operators (rightward)

    right       not

    left        and

    left        or xor

In the following sections, these operators are covered in precedence order. 

Many operators can be overloaded for objects. See the overload manpage. 



DESCRIPTION



Terms and List Operators (Leftward)

A TERM has the highest precedence in Perl. They includes variables, quote and quote-like operators, any expression in parentheses, and any function whose arguments are parenthesized. Actually, there aren't really functions in this sense, just list operators and unary operators behaving as functions because you put parentheses around the arguments. These are all documented in the perlfunc manpage. 

If any list operator (print(), etc.) or any unary operator (chdir(), etc.) is followed by a left parenthesis as the next token, the operator and arguments within parentheses are taken to be of highest precedence, just like a normal function call. 

In the absence of parentheses, the precedence of list operators such as print, sort, or chmod is either very high or very low depending on whether you are looking at the left side or the right side of the operator. For example, in 

    @ary = (1, 3, sort 4, 2);

    print @ary;         # prints 1324

the commas on the right of the sort are evaluated before the sort, but the commas on the left are evaluated after. In other words, list operators tend to gobble up all the arguments that follow them, and then act like a simple TERM with regard to the preceding expression. Note that you have to be careful with parentheses: 

    # These evaluate exit before doing the print:

    print($foo, exit);  # Obviously not what you want.

    print $foo, exit;   # Nor is this.

    # These do the print before evaluating exit:

    (print $foo), exit; # This is what you want.

    print($foo), exit;  # Or this.

    print ($foo), exit; # Or even this.

Also note that 

    print ($foo & 255) + 1, "\n";

probably doesn't do what you expect at first glance. See Named Unary Operators for more discussion of this. 

Also parsed as terms are the do {} and eval {} constructs, as well as subroutine and method calls, and the anonymous constructors [] and {}. 

See also Quote and Quote-like Operators toward the end of this section, as well as I/O Operators. 



The Arrow Operator

Just as in C and C++, ``->'' is an infix dereference operator. If the right side is either a [...] or {...} subscript, then the left side must be either a hard or symbolic reference to an array or hash (or a location capable of holding a hard reference, if it's an lvalue (assignable)). See the perlref manpage. 

Otherwise, the right side is a method name or a simple scalar variable containing the method name, and the left side must either be an object (a blessed reference) or a class name (that is, a package name). See the perlobj manpage. 



Auto-increment and Auto-decrement

``++'' and ``--'' work as in C. That is, if placed before a variable, they increment or decrement the variable before returning the value, and if placed after, increment or decrement the variable after returning the value. 

The auto-increment operator has a little extra builtin magic to it. If you increment a variable that is numeric, or that has ever been used in a numeric context, you get a normal increment. If, however, the variable has been used in only string contexts since it was set, and has a value that is not the empty string and matches the pattern /^[a-zA-Z]*[0-9]*$/, the increment is done as a string, preserving each character within its range, with carry: 

    print ++($foo = '99');      # prints '100'

    print ++($foo = 'a0');      # prints 'a1'

    print ++($foo = 'Az');      # prints 'Ba'

    print ++($foo = 'zz');      # prints 'aaa'

The auto-decrement operator is not magical. 



Exponentiation

Binary ``**'' is the exponentiation operator. Note that it binds even more tightly than unary minus, so -2**4 is -(2**4), not (-2)**4. (This is implemented using C's pow(3) function, which actually works on doubles internally.) 



Symbolic Unary Operators

Unary ``!'' performs logical negation, i.e., ``not''. See also not for a lower precedence version of this. 

Unary ``-'' performs arithmetic negation if the operand is numeric. If the operand is an identifier, a string consisting of a minus sign concatenated with the identifier is returned. Otherwise, if the string starts with a plus or minus, a string starting with the opposite sign is returned. One effect of these rules is that -bareword is equivalent to "-bareword". 

Unary ``~'' performs bitwise negation, i.e., 1's complement. For example, 0666 &~ 027 is 0640. (See also Integer Arithmetic and Bitwise String Operators.) 

Unary ``+'' has no effect whatsoever, even on strings. It is useful syntactically for separating a function name from a parenthesized expression that would otherwise be interpreted as the complete list of function arguments. (See examples above under Terms and List Operators (Leftward).) 

Unary ``\'' creates a reference to whatever follows it. See the perlref manpage. Do not confuse this behavior with the behavior of backslash within a string, although both forms do convey the notion of protecting the next thing from interpretation. 



Binding Operators

Binary ``=~'' binds a scalar expression to a pattern match. Certain operations search or modify the string $_ by default. This operator makes that kind of operation work on some other string. The right argument is a search pattern, substitution, or transliteration. The left argument is what is supposed to be searched, substituted, or transliterated instead of the default $_. The return value indicates the success of the operation. (If the right argument is an expression rather than a search pattern, substitution, or transliteration, it is interpreted as a search pattern at run time. This can be is less efficient than an explicit search, because the pattern must be compiled every time the expression is evaluated. 

Binary ``!~'' is just like ``=~'' except the return value is negated in the logical sense. 



Multiplicative Operators

Binary ``*'' multiplies two numbers. 

Binary ``/'' divides two numbers. 

Binary ``%'' computes the modulus of two numbers. Given integer operands $a and $b: If $b is positive, then $a % $b is $a minus the largest multiple of $b that is not greater than $a. If $b is negative, then $a % $b is $a minus the smallest multiple of $b that is not less than $a (i.e. the result will be less than or equal to zero). Note than when use integer is in scope, ``%'' give you direct access to the modulus operator as implemented by your C compiler. This operator is not as well defined for negative operands, but it will execute faster. 

Binary ``x'' is the repetition operator. In scalar context, it returns a string consisting of the left operand repeated the number of times specified by the right operand. In list context, if the left operand is a list in parentheses, it repeats the list. 

    print '-' x 80;             # print row of dashes

    print "\t" x ($tab/8), ' ' x ($tab%8);      # tab over

    @ones = (1) x 80;           # a list of 80 1's

    @ones = (5) x @ones;        # set all elements to 5



Additive Operators

Binary ``+'' returns the sum of two numbers. 

Binary ``-'' returns the difference of two numbers. 

Binary ``.'' concatenates two strings. 



Shift Operators

Binary ``<<`` returns the value of its left argument shifted left by the number of bits specified by the right argument. Arguments should be integers. (See also Integer Arithmetic.) 

Binary ``>>'' returns the value of its left argument shifted right by the number of bits specified by the right argument. Arguments should be integers. (See also Integer Arithmetic.) 



Named Unary Operators

The various named unary operators are treated as functions with one argument, with optional parentheses. These include the filetest operators, like -f, -M, etc. See the perlfunc manpage. 

If any list operator (print(), etc.) or any unary operator (chdir(), etc.) is followed by a left parenthesis as the next token, the operator and arguments within parentheses are taken to be of highest precedence, just like a normal function call. Examples: 

    chdir $foo    || die;       # (chdir $foo) || die

    chdir($foo)   || die;       # (chdir $foo) || die

    chdir ($foo)  || die;       # (chdir $foo) || die

    chdir +($foo) || die;       # (chdir $foo) || die

but, because * is higher precedence than ||: 

    chdir $foo * 20;    # chdir ($foo * 20)

    chdir($foo) * 20;   # (chdir $foo) * 20

    chdir ($foo) * 20;  # (chdir $foo) * 20

    chdir +($foo) * 20; # chdir ($foo * 20)

    rand 10 * 20;       # rand (10 * 20)

    rand(10) * 20;      # (rand 10) * 20

    rand (10) * 20;     # (rand 10) * 20

    rand +(10) * 20;    # rand (10 * 20)

See also Terms and List Operators (Leftward). 



Relational Operators

Binary ``<'' returns true if the left argument is numerically less than the right argument. 

Binary ``>'' returns true if the left argument is numerically greater than the right argument. 

Binary ``<='' returns true if the left argument is numerically less than or equal to the right argument. 

Binary ``>='' returns true if the left argument is numerically greater than or equal to the right argument. 

Binary ``lt'' returns true if the left argument is stringwise less than the right argument. 

Binary ``gt'' returns true if the left argument is stringwise greater than the right argument. 

Binary ``le'' returns true if the left argument is stringwise less than or equal to the right argument. 

Binary ``ge'' returns true if the left argument is stringwise greater than or equal to the right argument. 



Equality Operators

Binary ``=='' returns true if the left argument is numerically equal to the right argument. 

Binary ``!='' returns true if the left argument is numerically not equal to the right argument. 

Binary ``<=>'' returns -1, 0, or 1 depending on whether the left argument is numerically less than, equal to, or greater than the right argument. 

Binary ``eq'' returns true if the left argument is stringwise equal to the right argument. 

Binary ``ne'' returns true if the left argument is stringwise not equal to the right argument. 

Binary ``cmp'' returns -1, 0, or 1 depending on whether the left argument is stringwise less than, equal to, or greater than the right argument. 

``lt'', ``le'', ``ge'', ``gt'' and ``cmp'' use the collation (sort) order specified by the current locale if use locale is in effect. See the perllocale manpage. 



Bitwise And

Binary ``&'' returns its operators ANDed together bit by bit. (See also Integer Arithmetic and Bitwise String Operators.) 



Bitwise Or and Exclusive Or

Binary ``|'' returns its operators ORed together bit by bit. (See also Integer Arithmetic and Bitwise String Operators.) 

Binary ``^'' returns its operators XORed together bit by bit. (See also Integer Arithmetic and Bitwise String Operators.) 



C-style Logical And

Binary ``&&'' performs a short-circuit logical AND operation. That is, if the left operand is false, the right operand is not even evaluated. Scalar or list context propagates down to the right operand if it is evaluated. 



C-style Logical Or

Binary ``||'' performs a short-circuit logical OR operation. That is, if the left operand is true, the right operand is not even evaluated. Scalar or list context propagates down to the right operand if it is evaluated. 

The || and && operators differ from C's in that, rather than returning 0 or 1, they return the last value evaluated. Thus, a reasonably portable way to find out the home directory (assuming it's not ``0'') might be: 

    $home = $ENV{'HOME'} || $ENV{'LOGDIR'} ||

        (getpwuid($<))[7] || die "You're homeless!\n";

In particular, this means that you shouldn't use this for selecting between two aggregates for assignment: 

    @a = @b || @c;              # this is wrong

    @a = scalar(@b) || @c;      # really meant this

    @a = @b ? @b : @c;          # this works fine, though

As more readable alternatives to && and || when used for control flow, Perl provides and and or operators (see below). The short-circuit behavior is identical. The precedence of ``and'' and ``or'' is much lower, however, so that you can safely use them after a list operator without the need for parentheses: 

    unlink "alpha", "beta", "gamma"

            or gripe(), next LINE;

With the C-style operators that would have been written like this: 

    unlink("alpha", "beta", "gamma")

            || (gripe(), next LINE);

Use ``or'' for assignment is unlikely to do what you want; see below. 



Range Operators

Binary ``..'' is the range operator, which is really two different operators depending on the context. In list context, it returns an array of values counting (by ones) from the left value to the right value. This is useful for writing foreach (1..10) loops and for doing slice operations on arrays. In the current implementation, no temporary array is created when the range operator is used as the expression in foreach loops, but older versions of Perl might burn a lot of memory when you write something like this: 

    for (1 .. 1_000_000) {

        # code

    }

In scalar context, ``..'' returns a boolean value. The operator is bistable, like a flip-flop, and emulates the line-range (comma) operator of sed, awk, and various editors. Each ``..'' operator maintains its own boolean state. It is false as long as its left operand is false. Once the left operand is true, the range operator stays true until the right operand is true, AFTER which the range operator becomes false again. (It doesn't become false till the next time the range operator is evaluated. It can test the right operand and become false on the same evaluation it became true (as in awk), but it still returns true once. If you don't want it to test the right operand till the next evaluation (as in sed), use three dots (``...'') instead of two.) The right operand is not evaluated while the operator is in the ``false'' state, and the left operand is not evaluated while the operator is in the ``true'' state. The precedence is a little lower than || and &&. The value returned is either the empty string for false, or a sequence number (beginning with 1) for true. The sequence number is reset for each range encountered. The final sequence number in a range has the string ``E0'' appended to it, which doesn't affect its numeric value, but gives you something to search for if you want to exclude the endpoint. You can exclude the beginning point by waiting for the sequence number to be greater than 1. If either operand of scalar ``..'' is a constant expression, that operand is implicitly compared to the $. variable, the current line number. Examples: 

As a scalar operator: 

    if (101 .. 200) { print; }  # print 2nd hundred lines

    next line if (1 .. /^$/);   # skip header lines

    s/^/> / if (/^$/ .. eof()); # quote body

    # parse mail messages

    while (<>) {

        $in_header =   1  .. /^$/;

        $in_body   = /^$/ .. eof();

        # do something based on those

    } continue {

        close ARGV if eof;              # reset $. each file

    }

As a list operator: 

    for (101 .. 200) { print; } # print $_ 100 times

    @foo = @foo[0 .. $#foo];    # an expensive no-op

    @foo = @foo[$#foo-4 .. $#foo];      # slice last 5 items

The range operator (in list context) makes use of the magical auto-increment algorithm if the operands are strings. You can say 

    @alphabet = ('A' .. 'Z');

to get all the letters of the alphabet, or 

    $hexdigit = (0 .. 9, 'a' .. 'f')[$num & 15];

to get a hexadecimal digit, or 

    @z2 = ('01' .. '31');  print $z2[$mday];

to get dates with leading zeros. If the final value specified is not in the sequence that the magical increment would produce, the sequence goes until the next value would be longer than the final value specified. 



Conditional Operator

Ternary ``?:'' is the conditional operator, just as in C. It works much like an if-then-else. If the argument before the ? is true, the argument before the : is returned, otherwise the argument after the : is returned. For example: 

    printf "I have %d dog%s.\n", $n,

            ($n == 1) ? '' : "s";

Scalar or list context propagates downward into the 2nd or 3rd argument, whichever is selected. 

    $a = $ok ? $b : $c;  # get a scalar

    @a = $ok ? @b : @c;  # get an array

    $a = $ok ? @b : @c;  # oops, that's just a count!

The operator may be assigned to if both the 2nd and 3rd arguments are legal lvalues (meaning that you can assign to them): 

    ($a_or_b ? $a : $b) = $c;

This is not necessarily guaranteed to contribute to the readability of your program. 

Because this operator produces an assignable result, using assignments without parentheses will get you in trouble. For example, this: 

    $a % 2 ? $a += 10 : $a += 2

Really means this: 

    (($a % 2) ? ($a += 10) : $a) += 2

Rather than this: 

    ($a % 2) ? ($a += 10) : ($a += 2)



Assignment Operators

``='' is the ordinary assignment operator. 

Assignment operators work as in C. That is, 

    $a += 2;

is equivalent to 

    $a = $a + 2;

although without duplicating any side effects that dereferencing the lvalue might trigger, such as from tie(). Other assignment operators work similarly. The following are recognized: 

    **=    +=    *=    &=    <<=    &&=

           -=    /=    |=    >>=    ||=

           .=    %=    ^=

                 x=

Note that while these are grouped by family, they all have the precedence of assignment. 

Unlike in C, the assignment operator produces a valid lvalue. Modifying an assignment is equivalent to doing the assignment and then modifying the variable that was assigned to. This is useful for modifying a copy of something, like this: 

    ($tmp = $global) =~ tr [A-Z] [a-z];

Likewise, 

    ($a += 2) *= 3;

is equivalent to 

    $a += 2;

    $a *= 3;



Comma Operator

Binary ``,'' is the comma operator. In scalar context it evaluates its left argument, throws that value away, then evaluates its right argument and returns that value. This is just like C's comma operator. 

In list context, it's just the list argument separator, and inserts both its arguments into the list. 

The => digraph is mostly just a synonym for the comma operator. It's useful for documenting arguments that come in pairs. As of release 5.001, it also forces any word to the left of it to be interpreted as a string. 



List Operators (Rightward)

On the right side of a list operator, it has very low precedence, such that it controls all comma-separated expressions found there. The only operators with lower precedence are the logical operators ``and'', ``or'', and ``not'', which may be used to evaluate calls to list operators without the need for extra parentheses: 

    open HANDLE, "filename"

        or die "Can't open: $!\n";

See also discussion of list operators in Terms and List Operators (Leftward). 



Logical Not

Unary ``not'' returns the logical negation of the expression to its right. It's the equivalent of ``!'' except for the very low precedence. 



Logical And

Binary ``and'' returns the logical conjunction of the two surrounding expressions. It's equivalent to && except for the very low precedence. This means that it short-circuits: i.e., the right expression is evaluated only if the left expression is true. 



Logical or and Exclusive Or

Binary ``or'' returns the logical disjunction of the two surrounding expressions. It's equivalent to || except for the very low precedence. This makes it useful for control flow 

    print FH $data              or die "Can't write to FH: $!";

This means that it short-circuits: i.e., the right expression is evaluated only if the left expression is false. Due to its precedence, you should probably avoid using this for assignment, only for control flow. 

    $a = $b or $c;              # bug: this is wrong

    ($a = $b) or $c;            # really means this

    $a = $b || $c;              # better written this way

However, when it's a list context assignment and you're trying to use ``||'' for control flow, you probably need ``or'' so that the assignment takes higher precedence. 

    @info = stat($file) || die;     # oops, scalar sense of stat!

    @info = stat($file) or die;     # better, now @info gets its due

Then again, you could always use parentheses. 

Binary ``xor'' returns the exclusive-OR of the two surrounding expressions. It cannot short circuit, of course. 



C Operators Missing From Perl

Here is what C has that Perl doesn't: 

unary & 

Address-of operator. (But see the ``\'' operator for taking a reference.) 

unary * 

Dereference-address operator. (Perl's prefix dereferencing operators are typed: $, @, %, and &.) 

(TYPE) 

Type casting operator. 



Quote and Quote-like Operators

While we usually think of quotes as literal values, in Perl they function as operators, providing various kinds of interpolating and pattern matching capabilities. Perl provides customary quote characters for these behaviors, but also provides a way for you to choose your quote character for any of them. In the following table, a {} represents any pair of delimiters you choose. Non-bracketing delimiters use the same character fore and aft, but the 4 sorts of brackets (round, angle, square, curly) will all nest. 

    Customary  Generic        Meaning        Interpolates

        ''       q{}          Literal             no

        ""      qq{}          Literal             yes

        ``      qx{}          Command             yes (unless '' is delimiter)

                qw{}         Word list            no

        //       m{}       Pattern match          yes

                qr{}          Pattern             yes

                 s{}{}      Substitution          yes

                tr{}{}    Transliteration         no (but see below)

Note that there can be whitespace between the operator and the quoting characters, except when # is being used as the quoting character. q#foo# is parsed as being the string foo, while q #foo# is the operator q followed by a comment. Its argument will be taken from the next line. This allows you to write: 

    s {foo}  # Replace foo

      {bar}  # with bar.

For constructs that do interpolation, variables beginning with ``$'' or ``@'' are interpolated, as are the following sequences. Within a transliteration, the first ten of these sequences may be used. 

    \t          tab             (HT, TAB)

    \n          newline         (NL)

    \r          return          (CR)

    \f          form feed       (FF)

    \b          backspace       (BS)

    \a          alarm (bell)    (BEL)

    \e          escape          (ESC)

    \033        octal char

    \x1b        hex char

    \c[         control char

    \l          lowercase next char

    \u          uppercase next char

    \L          lowercase till \E

    \U          uppercase till \E

    \E          end case modification

    \Q          quote non-word characters till \E

If use locale is in effect, the case map used by \l, \L, \u and \U is taken from the current locale. See the perllocale manpage. 

All systems use the virtual "\n" to represent a line terminator, called a ``newline''. There is no such thing as an unvarying, physical newline character. It is an illusion that the operating system, device drivers, C libraries, and Perl all conspire to preserve. Not all systems read "\r" as ASCII CR and "\n" as ASCII LF. For example, on a Mac, these are reversed, and on systems without line terminator, printing "\n" may emit no actual data. In general, use "\n" when you mean a ``newline'' for your system, but use the literal ASCII when you need an exact character. For example, most networking protocols expect and prefer a CR+LF ( "\012\015" or "\cJ\cM") for line terminators, and although they often accept just "\012", they seldom tolerate just "\015". If you get in the habit of using "\n" for networking, you may be burned some day. 

You cannot include a literal $ or @ within a \Q sequence. An unescaped $ or @ interpolates the corresponding variable, while escaping will cause the literal string \$ to be inserted. You'll need to write something like m/\Quser\E\@\Qhost/. 

Patterns are subject to an additional level of interpretation as a regular expression. This is done as a second pass, after variables are interpolated, so that regular expressions may be incorporated into the pattern from the variables. If this is not what you want, use \Q to interpolate a variable literally. 

Apart from the above, there are no multiple levels of interpolation. In particular, contrary to the expectations of shell programmers, back-quotes do NOT interpolate within double quotes, nor do single quotes impede evaluation of variables when used within double quotes. 



Regexp Quote-Like Operators

Here are the quote-like operators that apply to pattern matching and related activities. 

Most of this section is related to use of regular expressions from Perl. Such a use may be considered from two points of view: Perl handles a a string and a ``pattern'' to RE (regular expression) engine to match, RE engine finds (or does not find) the match, and Perl uses the findings of RE engine for its operation, possibly asking the engine for other matches. 

RE engine has no idea what Perl is going to do with what it finds, similarly, the rest of Perl has no idea what a particular regular expression means to RE engine. This creates a clean separation, and in this section we discuss matching from Perl point of view only. The other point of view may be found in the perlre manpage. 

?PATTERN? 

This is just like the /pattern/ search, except that it matches only once between calls to the reset() operator. This is a useful optimization when you want to see only the first occurrence of something in each file of a set of files, for instance. Only ?? patterns local to the current package are reset. 

    while (<>) {

        if (?^$?) {

                            # blank line between header and body

        }

    } continue {

        reset if eof;       # clear ?? status for next file

    }

This usage is vaguely deprecated, and may be removed in some future version of Perl. 

m/PATTERN/cgimosx 

/PATTERN/cgimosx 

Searches a string for a pattern match, and in scalar context returns true (1) or false (''). If no string is specified via the =~ or !~ operator, the $_ string is searched. (The string specified with =~ need not be an lvalue--it may be the result of an expression evaluation, but remember the =~ binds rather tightly.) See also the perlre manpage. See the perllocale manpage for discussion of additional considerations that apply when use locale is in effect. 

Options are: 

    c   Do not reset search position on a failed match when /g is in effect.

    g   Match globally, i.e., find all occurrences.

    i   Do case-insensitive pattern matching.

    m   Treat string as multiple lines.

    o   Compile pattern only once.

    s   Treat string as single line.

    x   Use extended regular expressions.

If ``/'' is the delimiter then the initial m is optional. With the m you can use any pair of non-alphanumeric, non-whitespace characters as delimiters (if single quotes are used, no interpretation is done on the replacement string. Unlike Perl 4, Perl 5 treats backticks as normal delimiters; the replacement text is not evaluated as a command). This is particularly useful for matching Unix path names that contain ``/'', to avoid LTS (leaning toothpick syndrome). If ``?'' is the delimiter, then the match-only-once rule of ?PATTERN? applies. 

PATTERN may contain variables, which will be interpolated (and the pattern recompiled) every time the pattern search is evaluated. (Note that $) and $| might not be interpolated because they look like end-of-string tests.) If you want such a pattern to be compiled only once, add a /o after the trailing delimiter. This avoids expensive run-time recompilations, and is useful when the value you are interpolating won't change over the life of the script. However, mentioning /o constitutes a promise that you won't change the variables in the pattern. If you change them, Perl won't even notice. 

If the PATTERN evaluates to the empty string, the last successfully matched regular expression is used instead. 

If the /g option is not used, m// in a list context returns a list consisting of the subexpressions matched by the parentheses in the pattern, i.e., ($1, $2, $3...). (Note that here $1 etc. are also set, and that this differs from Perl 4's behavior.) When there are no parentheses in the pattern, the return value is the list (1) for success. With or without parentheses, an empty list is returned upon failure. 

Examples: 

    open(TTY, '/dev/tty');

    <TTY> =~ /^y/i && foo();    # do foo if desired

    if (/Version: *([0-9.]*)/) { $version = $1; }

    next if m#^/usr/spool/uucp#;

    # poor man's grep

    $arg = shift;

    while (<>) {

        print if /$arg/o;       # compile only once

    }

    if (($F1, $F2, $Etc) = ($foo =~ /^(\S+)\s+(\S+)\s*(.*)/))

This last example splits $foo into the first two words and the remainder of the line, and assigns those three fields to $F1, $F2, and $Etc. The conditional is true if any variables were assigned, i.e., if the pattern matched. 

The /g modifier specifies global pattern matching--that is, matching as many times as possible within the string. How it behaves depends on the context. In list context, it returns a list of all the substrings matched by all the parentheses in the regular expression. If there are no parentheses, it returns a list of all the matched strings, as if there were parentheses around the whole pattern. 

In scalar context, each execution of m//g finds the next match, returning TRUE if it matches, and FALSE if there is no further match. The position after the last match can be read or set using the pos() function; see pos. A failed match normally resets the search position to the beginning of the string, but you can avoid that by adding the /c modifier (e.g. m//gc). Modifying the target string also resets the search position. 

You can intermix m//g matches with m/\G.../g, where \G is a zero-width assertion that matches the exact position where the previous m//g, if any, left off. The \G assertion is not supported without the /g modifier; currently, without /g, \G behaves just like \A, but that's accidental and may change in the future. 

Examples: 

    # list context

    ($one,$five,$fifteen) = (`uptime` =~ /(\d+\.\d+)/g);

    # scalar context

    $/ = ""; $* = 1;  # $* deprecated in modern perls

    while (defined($paragraph = <>)) {

        while ($paragraph =~ /[a-z]['")]*[.!?]+['")]*\s/g) {

            $sentences++;

        }

    }

    print "$sentences\n";

    # using m//gc with \G

    $_ = "ppooqppqq";

    while ($i++ < 2) {

        print "1: '";

        print $1 while /(o)/gc; print "', pos=", pos, "\n";

        print "2: '";

        print $1 if /\G(q)/gc;  print "', pos=", pos, "\n";

        print "3: '";

        print $1 while /(p)/gc; print "', pos=", pos, "\n";

    }

The last example should print: 

    1: 'oo', pos=4

    2: 'q', pos=5

    3: 'pp', pos=7

    1: '', pos=7

    2: 'q', pos=8

    3: '', pos=8

A useful idiom for lex-like scanners is /\G.../gc. You can combine several regexps like this to process a string part-by-part, doing different actions depending on which regexp matched. Each regexp tries to match where the previous one leaves off. 

 $_ = <<'EOL';

      $url = new URI::URL "http://www/";;   die if $url eq "xXx";

 EOL

 LOOP:

    {

      print(" digits"),         redo LOOP if /\G\d+\b[,.;]?\s*/gc;

      print(" lowercase"),      redo LOOP if /\G[a-z]+\b[,.;]?\s*/gc;

      print(" UPPERCASE"),      redo LOOP if /\G[A-Z]+\b[,.;]?\s*/gc;

      print(" Capitalized"),    redo LOOP if /\G[A-Z][a-z]+\b[,.;]?\s*/gc;

      print(" MiXeD"),          redo LOOP if /\G[A-Za-z]+\b[,.;]?\s*/gc;

      print(" alphanumeric"),   redo LOOP if /\G[A-Za-z0-9]+\b[,.;]?\s*/gc;

      print(" line-noise"),     redo LOOP if /\G[^A-Za-z0-9]+/gc;

      print ". That's all!\n";

    }

Here is the output (split into several lines): 

 line-noise lowercase line-noise lowercase UPPERCASE line-noise

 UPPERCASE line-noise lowercase line-noise lowercase line-noise

 lowercase lowercase line-noise lowercase lowercase line-noise

 MiXeD line-noise. That's all!

q/STRING/ 

'STRING' 

A single-quoted, literal string. A backslash represents a backslash unless followed by the delimiter or another backslash, in which case the delimiter or backslash is interpolated. 

    $foo = q!I said, "You said, 'She said it.'"!;

    $bar = q('This is it.');

    $baz = '\n';                # a two-character string

qq/STRING/ 

"STRING" 

A double-quoted, interpolated string. 

    $_ .= qq

     (*** The previous line contains the naughty word "$1".\n)

                if /(tcl|rexx|python)/;      # :-)

    $baz = "\n";                # a one-character string

qr/STRING/imosx 

A string which is (possibly) interpolated and then compiled as a regular expression. The result may be used as a pattern in a match 

    $re = qr/$pattern/;

    $string =~ /foo${re}bar/;   # can be interpolated in other patterns

    $string =~ $re;             # or used standalone

Options are: 

    i   Do case-insensitive pattern matching.

    m   Treat string as multiple lines.

    o   Compile pattern only once.

    s   Treat string as single line.

    x   Use extended regular expressions.

The benefit from this is that the pattern is precompiled into an internal representation, and does not need to be recompiled every time a match is attempted. This makes it very efficient to do something like: 

    foreach $pattern (@pattern_list) {

        my $re = qr/$pattern/;

        foreach $line (@lines) {

            if($line =~ /$re/) {

                do_something($line);

            }

        }

    }

See the perlre manpage for additional information on valid syntax for STRING, and for a detailed look at the semantics of regular expressions. 

qx/STRING/ 

`STRING` 

A string which is (possibly) interpolated and then executed as a system command with /bin/sh or its equivalent. Shell wildcards, pipes, and redirections will be honored. The collected standard output of the command is returned; standard error is unaffected. In scalar context, it comes back as a single (potentially multi-line) string. In list context, returns a list of lines (however you've defined lines with $/ or $INPUT_RECORD_SEPARATOR). 

Because backticks do not affect standard error, use shell file descriptor syntax (assuming the shell supports this) if you care to address this. To capture a command's STDERR and STDOUT together: 

    $output = `cmd 2>&1`;

To capture a command's STDOUT but discard its STDERR: 

    $output = `cmd 2>/dev/null`;

To capture a command's STDERR but discard its STDOUT (ordering is important here): 

    $output = `cmd 2>&1 1>/dev/null`;

To exchange a command's STDOUT and STDERR in order to capture the STDERR but leave its STDOUT to come out the old STDERR: 

    $output = `cmd 3>&1 1>&2 2>&3 3>&-`;

To read both a command's STDOUT and its STDERR separately, it's easiest and safest to redirect them separately to files, and then read from those files when the program is done: 

    system("program args 1>/tmp/program.stdout 2>/tmp/program.stderr");

Using single-quote as a delimiter protects the command from Perl's double-quote interpolation, passing it on to the shell instead: 

    $perl_info  = qx(ps $$);            # that's Perl's $$

    $shell_info = qx'ps $$';            # that's the new shell's $$

Note that how the string gets evaluated is entirely subject to the command interpreter on your system. On most platforms, you will have to protect shell metacharacters if you want them treated literally. This is in practice difficult to do, as it's unclear how to escape which characters. See the perlsec manpage for a clean and safe example of a manual fork() and exec() to emulate backticks safely. 

On some platforms (notably DOS-like ones), the shell may not be capable of dealing with multiline commands, so putting newlines in the string may not get you what you want. You may be able to evaluate multiple commands in a single line by separating them with the command separator character, if your shell supports that (e.g. ; on many Unix shells; & on the Windows NT cmd shell). 

Beware that some command shells may place restrictions on the length of the command line. You must ensure your strings don't exceed this limit after any necessary interpolations. See the platform-specific release notes for more details about your particular environment. 

Using this operator can lead to programs that are difficult to port, because the shell commands called vary between systems, and may in fact not be present at all. As one example, the type command under the POSIX shell is very different from the type command under DOS. That doesn't mean you should go out of your way to avoid backticks when they're the right way to get something done. Perl was made to be a glue language, and one of the things it glues together is commands. Just understand what you're getting yourself into. 

See I/O Operators for more discussion. 

qw/STRING/ 

Returns a list of the words extracted out of STRING, using embedded whitespace as the word delimiters. It is exactly equivalent to 

    split(' ', q/STRING/);

This equivalency means that if used in scalar context, you'll get split's (unfortunate) scalar context behavior, complete with mysterious warnings. 

Some frequently seen examples: 

    use POSIX qw( setlocale localeconv )

    @EXPORT = qw( foo bar baz );

A common mistake is to try to separate the words with comma or to put comments into a multi-line qw-string. For this reason the -w switch produce warnings if the STRING contains the ``,'' or the ``#'' character. 

s/PATTERN/REPLACEMENT/egimosx 

Searches a string for a pattern, and if found, replaces that pattern with the replacement text and returns the number of substitutions made. Otherwise it returns false (specifically, the empty string). 

If no string is specified via the =~ or !~ operator, the $_ variable is searched and modified. (The string specified with =~ must be scalar variable, an array element, a hash element, or an assignment to one of those, i.e., an lvalue.) 

If the delimiter chosen is single quote, no variable interpolation is done on either the PATTERN or the REPLACEMENT. Otherwise, if the PATTERN contains a $ that looks like a variable rather than an end-of-string test, the variable will be interpolated into the pattern at run-time. If you want the pattern compiled only once the first time the variable is interpolated, use the /o option. If the pattern evaluates to the empty string, the last successfully executed regular expression is used instead. See the perlre manpage for further explanation on these. See the perllocale manpage for discussion of additional considerations that apply when use locale is in effect. 

Options are: 

    e   Evaluate the right side as an expression.

    g   Replace globally, i.e., all occurrences.

    i   Do case-insensitive pattern matching.

    m   Treat string as multiple lines.

    o   Compile pattern only once.

    s   Treat string as single line.

    x   Use extended regular expressions.

Any non-alphanumeric, non-whitespace delimiter may replace the slashes. If single quotes are used, no interpretation is done on the replacement string (the /e modifier overrides this, however). Unlike Perl 4, Perl 5 treats backticks as normal delimiters; the replacement text is not evaluated as a command. If the PATTERN is delimited by bracketing quotes, the REPLACEMENT has its own pair of quotes, which may or may not be bracketing quotes, e.g., s(foo)(bar) or s<foo>/bar/. A /e will cause the replacement portion to be interpreted as a full-fledged Perl expression and eval()ed right then and there. It is, however, syntax checked at compile-time. 

Examples: 

    s/\bgreen\b/mauve/g;                # don't change wintergreen

    $path =~ s|/usr/bin|/usr/local/bin|;

    s/Login: $foo/Login: $bar/; # run-time pattern

    ($foo = $bar) =~ s/this/that/;      # copy first, then change

    $count = ($paragraph =~ s/Mister\b/Mr./g);  # get change-count

    $_ = 'abc123xyz';

    s/\d+/$&*2/e;               # yields 'abc246xyz'

    s/\d+/sprintf("%5d",$&)/e;  # yields 'abc  246xyz'

    s/\w/$& x 2/eg;             # yields 'aabbcc  224466xxyyzz'

    s/%(.)/$percent{$1}/g;      # change percent escapes; no /e

    s/%(.)/$percent{$1} || $&/ge;       # expr now, so /e

    s/^=(\w+)/&pod($1)/ge;      # use function call

    # expand variables in $_, but dynamics only, using

    # symbolic dereferencing

    s/\$(\w+)/${$1}/g;

    # /e's can even nest;  this will expand

    # any embedded scalar variable (including lexicals) in $_

    s/(\$\w+)/$1/eeg;

    # Delete (most) C comments.

    $program =~ s {

        /\*     # Match the opening delimiter.

        .*?     # Match a minimal number of characters.

        \*/     # Match the closing delimiter.

    } []gsx;

    s/^\s*(.*?)\s*$/$1/;        # trim white space in $_, expensively

    for ($variable) {           # trim white space in $variable, cheap

        s/^\s+//;

        s/\s+$//;

    }

    s/([^ ]*) *([^ ]*)/$2 $1/;  # reverse 1st two fields

Note the use of $ instead of \ in the last example. Unlike sed, we use the \<digit> form in only the left hand side. Anywhere else it's $<digit>. 

Occasionally, you can't use just a /g to get all the changes to occur. Here are two common cases: 

    # put commas in the right places in an integer

    1 while s/(.*\d)(\d\d\d)/$1,$2/g;      # perl4

    1 while s/(\d)(\d\d\d)(?!\d)/$1,$2/g;  # perl5

    # expand tabs to 8-column spacing

    1 while s/\t+/' ' x (length($&)*8 - length($`)%8)/e;

tr/SEARCHLIST/REPLACEMENTLIST/cds 

y/SEARCHLIST/REPLACEMENTLIST/cds 

Transliterates all occurrences of the characters found in the search list with the corresponding character in the replacement list. It returns the number of characters replaced or deleted. If no string is specified via the =~ or !~ operator, the $_ string is transliterated. (The string specified with =~ must be a scalar variable, an array element, a hash element, or an assignment to one of those, i.e., an lvalue.) A character range may be specified with a hyphen, so tr/A-J/0-9/ does the same replacement as tr/ACEGIBDFHJ/0246813579/. For sed devotees, y is provided as a synonym for tr. If the SEARCHLIST is delimited by bracketing quotes, the REPLACEMENTLIST has its own pair of quotes, which may or may not be bracketing quotes, e.g., tr[A-Z][a-z] or tr(+\-*/)/ABCD/. 

Options: 

    c   Complement the SEARCHLIST.

    d   Delete found but unreplaced characters.

    s   Squash duplicate replaced characters.

If the /c modifier is specified, the SEARCHLIST character set is complemented. If the /d modifier is specified, any characters specified by SEARCHLIST not found in REPLACEMENTLIST are deleted. (Note that this is slightly more flexible than the behavior of some tr programs, which delete anything they find in the SEARCHLIST, period.) If the /s modifier is specified, sequences of characters that were transliterated to the same character are squashed down to a single instance of the character. 

If the /d modifier is used, the REPLACEMENTLIST is always interpreted exactly as specified. Otherwise, if the REPLACEMENTLIST is shorter than the SEARCHLIST, the final character is replicated till it is long enough. If the REPLACEMENTLIST is empty, the SEARCHLIST is replicated. This latter is useful for counting characters in a class or for squashing character sequences in a class. 

Examples: 

    $ARGV[1] =~ tr/A-Z/a-z/;    # canonicalize to lower case

    $cnt = tr/*/*/;             # count the stars in $_

    $cnt = $sky =~ tr/*/*/;     # count the stars in $sky

    $cnt = tr/0-9//;            # count the digits in $_

    tr/a-zA-Z//s;               # bookkeeper -> bokeper

    ($HOST = $host) =~ tr/a-z/A-Z/;

    tr/a-zA-Z/ /cs;             # change non-alphas to single space

    tr [\200-\377]

       [\000-\177];             # delete 8th bit

If multiple transliterations are given for a character, only the first one is used: 

    tr/AAA/XYZ/

will transliterate any A to X. 

Note that because the transliteration table is built at compile time, neither the SEARCHLIST nor the REPLACEMENTLIST are subjected to double quote interpolation. That means that if you want to use variables, you must use an eval(): 

    eval "tr/$oldlist/$newlist/";

    die $@ if $@;

    eval "tr/$oldlist/$newlist/, 1" or die $@;



Gory details of parsing quoted constructs

When presented with something which may have several different interpretations, Perl uses the principle DWIM (expanded to Do What I Mean - not what I wrote) to pick up the most probable interpretation of the source. This strategy is so successful that Perl users usually do not suspect ambivalence of what they write. However, time to time Perl's ideas differ from what the author meant. 

The target of this section is to clarify the Perl's way of interpreting quoted constructs. The most frequent reason one may have to want to know the details discussed in this section is hairy regular expressions. However, the first steps of parsing are the same for all Perl quoting operators, so here they are discussed together. 

Some of the passes discussed below are performed concurrently, but as far as results are the same, we consider them one-by-one. For different quoting constructs Perl performs different number of passes, from one to five, but they are always performed in the same order. 

Finding the end 

First pass is finding the end of the quoted construct, be it multichar ender "\nEOF\n" of <<EOF construct, / which terminates qq/ construct, ] which terminates qq[ construct, or > which terminates a fileglob started with <. 

When searching for multichar construct no skipping is performed. When searching for one-char non-matching delimiter, such as /, combinations \\ and \/ are skipped. When searching for one-char matching delimiter, such as ], combinations \\, \] and \[ are skipped, and nested [, ] are skipped as well. 

For 3-parts constructs, s/// etc. the search is repeated once more. 

During this search no attention is paid to the semantic of the construct, thus 

    "$hash{"$foo/$bar"}"

or 

    m/ 

      bar       #  This is not a comment, this slash / terminated m//!

     /x

do not form legal quoted expressions. Note that since the slash which terminated m// was followed by a SPACE, this is not m//x, thus # was interpreted as a literal #. 

Removal of backslashes before delimiters 

During the second pass the text between the starting delimiter and the ending delimiter is copied to a safe location, and the \ is removed from combinations consisting of \ and delimiter(s) (both starting and ending delimiter if they differ). 

The removal does not happen for multi-char delimiters. 

Note that the combination \\ is left as it was! 

Starting from this step no information about the delimiter(s) is used in the parsing. 

Interpolation 

Next step is interpolation in the obtained delimiter-independent text. There are four different cases. 

C<<<'EOF'>, m'', s''', tr///, y/// 

No interpolation is performed. 

'', q// 

The only interpolation is removal of \ from pairs \\. 

"", ``, qq//, qx//, C<<file*globgt> 

\Q, \U, \u, \L, \l (possibly paired with \E) are converted to corresponding Perl constructs, thus "$foo\Qbaz$bar" is converted to 

   $foo . (quotemeta("baz" . $bar));

Other combinations of \ with following chars are substituted with appropriate expansions. 

Interpolated scalars and arrays are converted to join and . Perl constructs, thus "'@arr'" becomes 

  "'" . (join $", @arr) . "'";

Since all three above steps are performed simultaneously left-to-right, the is no way to insert a literal $ or @ inside \Q\E pair: it cannot be protected by \, since any \ (except in \E) is interpreted as a literal inside \Q\E, and any $ is interpreted as starting an interpolated scalar. 

Note also that the interpolating code needs to make decision where the interpolated scalar ends, say, whether "a $b -> {c}" means 

  "a " . $b . " -> {c}";

or 

  "a " . $b -> {c};

Most the time the decision is to take the longest possible text which does not include spaces between components and contains matching braces/brackets. 

?RE?, /RE/, m/RE/, s/RE/foo/, 

Processing of \Q, \U, \u, \L, \l and interpolation happens (almost) as with qq// constructs, but the substitution of \ followed by other chars is not performed! Moreover, inside (?{BLOCK}) no processing is performed at all. 

Interpolation has several quirks: $|, $( and $) are not interpolated, and constructs $var[SOMETHING] are voted (by several different estimators) to be an array element or $var followed by a RE alternative. This is the place where the notation ${arr[$bar]} comes handy: /${arr[0-9]}/ is interpreted as an array element -9, not as a regular expression from variable $arr followed by a digit, which is the interpretation of /$arr[0-9]/. 

Note that absence of processing of \\ creates specific restrictions on the post-processed text: if the delimiter is /, one cannot get the combination \/ into the result of this step: / will finish the regular expression, \/ will be stripped to / on the previous step, and \\/ will be left as is. Since / is equivalent to \/ inside a regular expression, this does not matter unless the delimiter is special character for the RE engine, as in s*foo*bar*, m[foo], or ?foo?. 

This step is the last one for all the constructs except regular expressions, which are processed further. 

Interpolation of regular expressions 

All the previous steps were performed during the compilation of Perl code, this one happens in run time (though it may be optimized to be calculated at compile time if appropriate). After all the preprocessing performed above (and possibly after evaluation if catenation, joining, up/down-casing and quotemeta()ing are involved) the resulting string is passed to RE engine for compilation. 

Whatever happens in the RE engine is better be discussed in the perlre manpage, but for the sake of continuity let us do it here. 

This is the first step where presence of the //x switch is relevant. The RE engine scans the string left-to-right, and converts it to a finite automaton. 

Backslashed chars are either substituted by corresponding literal strings, or generate special nodes of the finite automaton. Characters which are special to the RE engine generate corresponding nodes. (?#...) comments are ignored. All the rest is either converted to literal strings to match, or is ignored (as is whitespace and #-style comments if //x is present). 

Note that the parsing of the construct [...] is performed using absolutely different rules than the rest of the regular expression. Similarly, the (?{...}) is only checked for matching braces. 

Optimization of regular expressions 

This step is listed for completeness only. Since it does not change semantics, details of this step are not documented and are subject to change. 



I/O Operators

There are several I/O operators you should know about. A string enclosed by backticks (grave accents) first undergoes variable substitution just like a double quoted string. It is then interpreted as a command, and the output of that command is the value of the pseudo-literal, like in a shell. In scalar context, a single string consisting of all the output is returned. In list context, a list of values is returned, one for each line of output. (You can set $/ to use a different line terminator.) The command is executed each time the pseudo-literal is evaluated. The status value of the command is returned in $? (see the perlvar manpage for the interpretation of $?). Unlike in csh, no translation is done on the return data--newlines remain newlines. Unlike in any of the shells, single quotes do not hide variable names in the command from interpretation. To pass a $ through to the shell you need to hide it with a backslash. The generalized form of backticks is qx//. (Because backticks always undergo shell expansion as well, see the perlsec manpage for security concerns.) 

Evaluating a filehandle in angle brackets yields the next line from that file (newline, if any, included), or undef at end of file. Ordinarily you must assign that value to a variable, but there is one situation where an automatic assignment happens. If and ONLY if the input symbol is the only thing inside the conditional of a while or for(;;) loop, the value is automatically assigned to the variable $_. In these loop constructs, the assigned value (whether assignment is automatic or explicit) is then tested to see if it is defined. The defined test avoids problems where line has a string value that would be treated as false by perl e.g. ``'' or ``0'' with no trailing newline. (This may seem like an odd thing to you, but you'll use the construct in almost every Perl script you write.) Anyway, the following lines are equivalent to each other: 

    while (defined($_ = <STDIN>)) { print; }

    while ($_ = <STDIN>) { print; }

    while (<STDIN>) { print; }

    for (;<STDIN>;) { print; }

    print while defined($_ = <STDIN>);

    print while ($_ = <STDIN>);

    print while <STDIN>;

and this also behaves similarly, but avoids the use of $_ : 

    while (my $line = <STDIN>) { print $line }    

If you really mean such values to terminate the loop they should be tested for explicitly: 

    while (($_ = <STDIN>) ne '0') { ... }

    while (<STDIN>) { last unless $_; ... }

In other boolean contexts, <<EM>filehandle</EM>> without explicit defined test or comparison will solicit a warning if -w is in effect. 

The filehandles STDIN, STDOUT, and STDERR are predefined. (The filehandles stdin, stdout, and stderr will also work except in packages, where they would be interpreted as local identifiers rather than global.) Additional filehandles may be created with the open() function. See open() for details on this. 

If a <FILEHANDLE> is used in a context that is looking for a list, a list consisting of all the input lines is returned, one line per list element. It's easy to make a LARGE data space this way, so use with care. 

The null filehandle <> is special and can be used to emulate the behavior of sed and awk. Input from <> comes either from standard input, or from each file listed on the command line. Here's how it works: the first time <> is evaluated, the @ARGV array is checked, and if it is empty, $ARGV[0] is set to ``-'', which when opened gives you standard input. The @ARGV array is then processed as a list of filenames. The loop 

    while (<>) {

        ...                     # code for each line

    }

is equivalent to the following Perl-like pseudo code: 

    unshift(@ARGV, '-') unless @ARGV;

    while ($ARGV = shift) {

        open(ARGV, $ARGV);

        while (<ARGV>) {

            ...         # code for each line

        }

    }

except that it isn't so cumbersome to say, and will actually work. It really does shift array @ARGV and put the current filename into variable $ARGV. It also uses filehandle ARGV internally--<> is just a synonym for <ARGV>, which is magical. (The pseudo code above doesn't work because it treats <ARGV> as non-magical.) 

You can modify @ARGV before the first <> as long as the array ends up containing the list of filenames you really want. Line numbers ($.) continue as if the input were one big happy file. (But see example under eof for how to reset line numbers on each file.) 

If you want to set @ARGV to your own list of files, go right ahead. This sets @ARGV to all plain text files if no @ARGV was given: 

    @ARGV = grep { -f && -T } glob('*') unless @ARGV;

You can even set them to pipe commands. For example, this automatically filters compressed arguments through gzip: 

    @ARGV = map { /\.(gz|Z)$/ ? "gzip -dc < $_ |" : $_ } @ARGV;

If you want to pass switches into your script, you can use one of the Getopts modules or put a loop on the front like this: 

    while ($_ = $ARGV[0], /^-/) {

        shift;

        last if /^--$/;

        if (/^-D(.*)/) { $debug = $1 }

        if (/^-v/)     { $verbose++  }

        # ...           # other switches

    }

    while (<>) {

        # ...           # code for each line

    }

The <> symbol will return undef for end-of-file only once. If you call it again after this it will assume you are processing another @ARGV list, and if you haven't set @ARGV, will input from STDIN. 

If the string inside the angle brackets is a reference to a scalar variable (e.g., <$foo>), then that variable contains the name of the filehandle to input from, or its typeglob, or a reference to the same. For example: 

    $fh = \*STDIN;

    $line = <$fh>;

If what's within the angle brackets is neither a filehandle nor a simple scalar variable containing a filehandle name, typeglob, or typeglob reference, it is interpreted as a filename pattern to be globbed, and either a list of filenames or the next filename in the list is returned, depending on context. This distinction is determined on syntactic grounds alone. That means <$x> is always a readline from an indirect handle, but <$hash{key}> is always a glob. That's because $x is a simple scalar variable, but $hash{key} is not--it's a hash element. 

One level of double-quote interpretation is done first, but you can't say <$foo> because that's an indirect filehandle as explained in the previous paragraph. (In older versions of Perl, programmers would insert curly brackets to force interpretation as a filename glob: <${foo}>. These days, it's considered cleaner to call the internal function directly as glob($foo), which is probably the right way to have done it in the first place.) Example: 

    while (<*.c>) {

        chmod 0644, $_;

    }

is equivalent to 

    open(FOO, "echo *.c | tr -s ' \t\r\f' '\\012\\012\\012\\012'|");

    while (<FOO>) {

        chop;

        chmod 0644, $_;

    }

In fact, it's currently implemented that way. (Which means it will not work on filenames with spaces in them unless you have csh(1) on your machine.) Of course, the shortest way to do the above is: 

    chmod 0644, <*.c>;

Because globbing invokes a shell, it's often faster to call readdir() yourself and do your own grep() on the filenames. Furthermore, due to its current implementation of using a shell, the glob() routine may get ``Arg list too long'' errors (unless you've installed tcsh(1L) as /bin/csh). 

A glob evaluates its (embedded) argument only when it is starting a new list. All values must be read before it will start over. In a list context this isn't important, because you automatically get them all anyway. In scalar context, however, the operator returns the next value each time it is called, or a undef value if you've just run out. As for filehandles an automatic defined is generated when the glob occurs in the test part of a while or for - because legal glob returns (e.g. a file called 0) would otherwise terminate the loop. Again, undef is returned only once. So if you're expecting a single value from a glob, it is much better to say 

    ($file) = <blurch*>;

than 

    $file = <blurch*>;

because the latter will alternate between returning a filename and returning FALSE. 

It you're trying to do variable interpolation, it's definitely better to use the glob() function, because the older notation can cause people to become confused with the indirect filehandle notation. 

    @files = glob("$dir/*.[ch]");

    @files = glob($files[$i]);



Constant Folding

Like C, Perl does a certain amount of expression evaluation at compile time, whenever it determines that all arguments to an operator are static and have no side effects. In particular, string concatenation happens at compile time between literals that don't do variable substitution. Backslash interpretation also happens at compile time. You can say 

    'Now is the time for all' . "\n" .

        'good men to come to.'

and this all reduces to one string internally. Likewise, if you say 

    foreach $file (@filenames) {

        if (-s $file > 5 + 100 * 2**16) {  }

    }

the compiler will precompute the number that expression represents so that the interpreter won't have to. 



Bitwise String Operators

Bitstrings of any size may be manipulated by the bitwise operators (~ | & ^). 

If the operands to a binary bitwise op are strings of different sizes, or and xor ops will act as if the shorter operand had additional zero bits on the right, while the and op will act as if the longer operand were truncated to the length of the shorter. 

    # ASCII-based examples 

    print "j p \n" ^ " a h";            # prints "JAPH\n"

    print "JA" | "  ph\n";              # prints "japh\n"

    print "japh\nJunk" & '_____';       # prints "JAPH\n";

    print 'p N$' ^ " E<H\n";            # prints "Perl\n";

If you are intending to manipulate bitstrings, you should be certain that you're supplying bitstrings: If an operand is a number, that will imply a numeric bitwise operation. You may explicitly show which type of operation you intend by using "" or 0+, as in the examples below. 

    $foo =  150  |  105 ;       # yields 255  (0x96 | 0x69 is 0xFF)

    $foo = '150' |  105 ;       # yields 255

    $foo =  150  | '105';       # yields 255

    $foo = '150' | '105';       # yields string '155' (under ASCII)

    $baz = 0+$foo & 0+$bar;     # both ops explicitly numeric

    $biz = "$foo" ^ "$bar";     # both ops explicitly stringy



Integer Arithmetic

By default Perl assumes that it must do most of its arithmetic in floating point. But by saying 

    use integer;

you may tell the compiler that it's okay to use integer operations from here to the end of the enclosing BLOCK. An inner BLOCK may countermand this by saying 

    no integer;

which lasts until the end of that BLOCK. 

The bitwise operators (``&'', ``|'', ``^'', ``~'', ``<<``, and ''>>``) always produce integral results. (But see also Bitwise String Operators.) However, use integer still has meaning for them. By default, their results are interpreted as unsigned integers. However, if use integer is in effect, their results are interpreted as signed integers. For example, ~0 usually evaluates to a large integral value. However, use integer; ~0 is -1 on twos-complement machines. 



Floating-point Arithmetic

While use integer provides integer-only arithmetic, there is no similar ways to provide rounding or truncation at a certain number of decimal places. For rounding to a certain number of digits, sprintf() or printf() is usually the easiest route. 

Floating-point numbers are only approximations to what a mathematician would call real numbers. There are infinitely more reals than floats, so some corners must be cut. For example: 

    printf "%.20g\n", 123456789123456789;

    #        produces 123456789123456784

Testing for exact equality of floating-point equality or inequality is not a good idea. Here's a (relatively expensive) work-around to compare whether two floating-point numbers are equal to a particular number of decimal places. See Knuth, volume II, for a more robust treatment of this topic. 

    sub fp_equal {

        my ($X, $Y, $POINTS) = @_;

        my ($tX, $tY);

        $tX = sprintf("%.${POINTS}g", $X);

        $tY = sprintf("%.${POINTS}g", $Y);

        return $tX eq $tY;

    }

The POSIX module (part of the standard perl distribution) implements ceil(), floor(), and a number of other mathematical and trigonometric functions. The Math::Complex module (part of the standard perl distribution) defines a number of mathematical functions that can also work on real numbers. Math::Complex not as efficient as POSIX, but POSIX can't work with complex numbers. 

Rounding in financial applications can have serious implications, and the rounding method used should be specified precisely. In these cases, it probably pays not to trust whichever system rounding is being used by Perl, but to instead implement the rounding function you need yourself. 



Bigger Numbers

The standard Math::BigInt and Math::BigFloat modules provide variable precision arithmetic and overloaded operators. At the cost of some space and considerable speed, they avoid the normal pitfalls associated with limited-precision representations. 

    use Math::BigInt;

    $x = Math::BigInt->new('123456789123456789');

    print $x * $x;

    # prints +15241578780673678515622620750190521



DISCLAIMER

We are painfully aware that these documents may contain incorrect links and misformatted HTML. Such bugs lie in the automatic translation process that automatically created the hundreds and hundreds of separate documents that you find here. Please do not report link or formatting bugs, because we cannot fix per-document problems. The only bug reports that will help us are those that supply working patches to the installhtml or pod2html programs, or to the Pod::HTML module itself, for which I and the entire Perl community will shower you with thanks and praises. 

If rather than formatting bugs, you encounter substantive content errors in these documents, such as mistakes in the explanations or code, please use the perlbug utility included with the Perl distribution. 

--Tom Christiansen, Perl Documentation Compiler and Editor

NAME

perlre - Perl regular expressions 



DESCRIPTION

This page describes the syntax of regular expressions in Perl. For a description of how to use regular expressions in matching operations, plus various examples of the same, see discussion of m//, s///, qr// and ?? in Regexp Quote-Like Operators. 

The matching operations can have various modifiers. The modifiers that relate to the interpretation of the regular expression inside are listed below. For the modifiers that alter the way a regular expression is used by Perl, see Regexp Quote-Like Operators and Gory details of parsing quoted constructs. 

i 

Do case-insensitive pattern matching. 

If use locale is in effect, the case map is taken from the current locale. See the perllocale manpage. 

m 

Treat string as multiple lines. That is, change ``^'' and ``$'' from matching at only the very start or end of the string to the start or end of any line anywhere within the string, 

s 

Treat string as single line. That is, change ``.'' to match any character whatsoever, even a newline, which it normally would not match. 

The /s and /m modifiers both override the $* setting. That is, no matter what $* contains, /s without /m will force ``^'' to match only at the beginning of the string and ``$'' to match only at the end (or just before a newline at the end) of the string. Together, as /ms, they let the ``.'' match any character whatsoever, while yet allowing ``^'' and ``$'' to match, respectively, just after and just before newlines within the string. 

x 

Extend your pattern's legibility by permitting whitespace and comments. 

These are usually written as ``the /x modifier'', even though the delimiter in question might not actually be a slash. In fact, any of these modifiers may also be embedded within the regular expression itself using the new (?...) construct. See below. 

The /x modifier itself needs a little more explanation. It tells the regular expression parser to ignore whitespace that is neither backslashed nor within a character class. You can use this to break up your regular expression into (slightly) more readable parts. The # character is also treated as a metacharacter introducing a comment, just as in ordinary Perl code. This also means that if you want real whitespace or # characters in the pattern (outside of a character class, where they are unaffected by /x), that you'll either have to escape them or encode them using octal or hex escapes. Taken together, these features go a long way towards making Perl's regular expressions more readable. Note that you have to be careful not to include the pattern delimiter in the comment--perl has no way of knowing you did not intend to close the pattern early. See the C-comment deletion code in the perlop manpage. 



Regular Expressions

The patterns used in pattern matching are regular expressions such as those supplied in the Version 8 regex routines. (In fact, the routines are derived (distantly) from Henry Spencer's freely redistributable reimplementation of the V8 routines.) See Version 8 Regular Expressions for details. 

In particular the following metacharacters have their standard egrep-ish meanings: 

    \   Quote the next metacharacter

    ^   Match the beginning of the line

    .   Match any character (except newline)

    $   Match the end of the line (or before newline at the end)

    |   Alternation

    ()  Grouping

    []  Character class

By default, the ``^'' character is guaranteed to match at only the beginning of the string, the ``$'' character at only the end (or before the newline at the end) and Perl does certain optimizations with the assumption that the string contains only one line. Embedded newlines will not be matched by ``^'' or ``$''. You may, however, wish to treat a string as a multi-line buffer, such that the ``^'' will match after any newline within the string, and ``$'' will match before any newline. At the cost of a little more overhead, you can do this by using the /m modifier on the pattern match operator. (Older programs did this by setting $*, but this practice is now deprecated.) 

To facilitate multi-line substitutions, the ``.'' character never matches a newline unless you use the /s modifier, which in effect tells Perl to pretend the string is a single line--even if it isn't. The /s modifier also overrides the setting of $*, in case you have some (badly behaved) older code that sets it in another module. 

The following standard quantifiers are recognized: 

    *      Match 0 or more times

    +      Match 1 or more times

    ?      Match 1 or 0 times

    {n}    Match exactly n times

    {n,}   Match at least n times

    {n,m}  Match at least n but not more than m times

(If a curly bracket occurs in any other context, it is treated as a regular character.) The ``*'' modifier is equivalent to {0,}, the ``+'' modifier to {1,}, and the ``?'' modifier to {0,1}. n and m are limited to integral values less than 65536. 

By default, a quantified subpattern is ``greedy'', that is, it will match as many times as possible (given a particular starting location) while still allowing the rest of the pattern to match. If you want it to match the minimum number of times possible, follow the quantifier with a ``?''. Note that the meanings don't change, just the ``greediness'': 

    *?     Match 0 or more times

    +?     Match 1 or more times

    ??     Match 0 or 1 time

    {n}?   Match exactly n times

    {n,}?  Match at least n times

    {n,m}? Match at least n but not more than m times

Because patterns are processed as double quoted strings, the following also work: 

    \t          tab                   (HT, TAB)

    \n          newline               (LF, NL)

    \r          return                (CR)

    \f          form feed             (FF)

    \a          alarm (bell)          (BEL)

    \e          escape (think troff)  (ESC)

    \033        octal char (think of a PDP-11)

    \x1B        hex char

    \c[         control char

    \l          lowercase next char (think vi)

    \u          uppercase next char (think vi)

    \L          lowercase till \E (think vi)

    \U          uppercase till \E (think vi)

    \E          end case modification (think vi)

    \Q          quote (disable) pattern metacharacters till \E

If use locale is in effect, the case map used by \l, \L, \u and \U is taken from the current locale. See the perllocale manpage. 

You cannot include a literal $ or @ within a \Q sequence. An unescaped $ or @ interpolates the corresponding variable, while escaping will cause the literal string \$ to be matched. You'll need to write something like m/\Quser\E\@\Qhost/. 

In addition, Perl defines the following: 

    \w  Match a "word" character (alphanumeric plus "_")

    \W  Match a non-word character

    \s  Match a whitespace character

    \S  Match a non-whitespace character

    \d  Match a digit character

    \D  Match a non-digit character

A \w matches a single alphanumeric character, not a whole word. To match a word you'd need to say \w+. If use locale is in effect, the list of alphabetic characters generated by \w is taken from the current locale. See the perllocale manpage. You may use \w, \W, \s, \S, \d, and \D within character classes (though not as either end of a range). 

Perl defines the following zero-width assertions: 

    \b  Match a word boundary

    \B  Match a non-(word boundary)

    \A  Match only at beginning of string

    \Z  Match only at end of string, or before newline at the end

    \z  Match only at end of string

    \G  Match only where previous m//g left off (works only with /g)

A word boundary (\b) is defined as a spot between two characters that has a \w on one side of it and a \W on the other side of it (in either order), counting the imaginary characters off the beginning and end of the string as matching a \W. (Within character classes \b represents backspace rather than a word boundary.) The \A and \Z are just like ``^'' and ``$'', except that they won't match multiple times when the /m modifier is used, while ``^'' and ``$'' will match at every internal line boundary. To match the actual end of the string, not ignoring newline, you can use \z. The \G assertion can be used to chain global matches (using m//g), as described in Regexp Quote-Like Operators. 

It is also useful when writing lex-like scanners, when you have several patterns that you want to match against consequent substrings of your string, see the previous reference. The actual location where \G will match can also be influenced by using pos() as an lvalue. See pos. 

When the bracketing construct ( ... ) is used, \<digit> matches the digit'th substring. Outside of the pattern, always use ``$'' instead of ``\'' in front of the digit. (While the \<digit> notation can on rare occasion work outside the current pattern, this should not be relied upon. See the WARNING below.) The scope of $<digit> (and $`, $&, and $') extends to the end of the enclosing BLOCK or eval string, or to the next successful pattern match, whichever comes first. If you want to use parentheses to delimit a subpattern (e.g., a set of alternatives) without saving it as a subpattern, follow the ( with a ?:. 

You may have as many parentheses as you wish. If you have more than 9 substrings, the variables $10, $11, ... refer to the corresponding substring. Within the pattern, \10, \11, etc. refer back to substrings if there have been at least that many left parentheses before the backreference. Otherwise (for backward compatibility) \10 is the same as \010, a backspace, and \11 the same as \011, a tab. And so on. (\1 through \9 are always backreferences.) 

$+ returns whatever the last bracket match matched. $& returns the entire matched string. ($0 used to return the same thing, but not any more.) $` returns everything before the matched string. $' returns everything after the matched string. Examples: 

    s/^([^ ]*) *([^ ]*)/$2 $1/;     # swap first two words

    if (/Time: (..):(..):(..)/) {

        $hours = $1;

        $minutes = $2;

        $seconds = $3;

    }

Once perl sees that you need one of $&, $` or $' anywhere in the program, it has to provide them on each and every pattern match. This can slow your program down. The same mechanism that handles these provides for the use of $1, $2, etc., so you pay the same price for each pattern that contains capturing parentheses. But if you never use $&, etc., in your script, then patterns without capturing parentheses won't be penalized. So avoid $&, $', and $` if you can, but if you can't (and some algorithms really appreciate them), once you've used them once, use them at will, because you've already paid the price. As of 5.005, $& is not so costly as the other two. 

Backslashed metacharacters in Perl are alphanumeric, such as \b, \w, \n. Unlike some other regular expression languages, there are no backslashed symbols that aren't alphanumeric. So anything that looks like \\, \(, \), \<, \>, \{, or \} is always interpreted as a literal character, not a metacharacter. This was once used in a common idiom to disable or quote the special meanings of regular expression metacharacters in a string that you want to use for a pattern. Simply quote all non-alphanumeric characters: 

    $pattern =~ s/(\W)/\\$1/g;

Now it is much more common to see either the quotemeta() function or the \Q escape sequence used to disable all metacharacters' special meanings like this: 

    /$unquoted\Q$quoted\E$unquoted/

Perl defines a consistent extension syntax for regular expressions. The syntax is a pair of parentheses with a question mark as the first thing within the parentheses (this was a syntax error in older versions of Perl). The character after the question mark gives the function of the extension. Several extensions are already supported: 

(?#text) 

A comment. The text is ignored. If the /x switch is used to enable whitespace formatting, a simple # will suffice. Note that perl closes the comment as soon as it sees a ), so there is no way to put a literal ) in the comment. 

(?:pattern) 

(?imsx-imsx:pattern) 

This is for clustering, not capturing; it groups subexpressions like ``()'', but doesn't make backreferences as ``()'' does. So 

    @fields = split(/\b(?:a|b|c)\b/)

is like 

    @fields = split(/\b(a|b|c)\b/)

but doesn't spit out extra fields. 

The letters between ? and : act as flags modifiers, see (?imsx-imsx). In particular, 

    /(?s-i:more.*than).*million/i

is equivalent to more verbose 

    /(?:(?s-i)more.*than).*million/i

(?=pattern) 

A zero-width positive lookahead assertion. For example, /\w+(?=\t)/ matches a word followed by a tab, without including the tab in $&. 

(?!pattern) 

A zero-width negative lookahead assertion. For example /foo(?!bar)/ matches any occurrence of ``foo'' that isn't followed by ``bar''. Note however that lookahead and lookbehind are NOT the same thing. You cannot use this for lookbehind. 

If you are looking for a ``bar'' that isn't preceded by a ``foo'', /(?!foo)bar/ will not do what you want. That's because the (?!foo) is just saying that the next thing cannot be ``foo''--and it's not, it's a ``bar'', so ``foobar'' will match. You would have to do something like /(?!foo)...bar/ for that. We say ``like'' because there's the case of your ``bar'' not having three characters before it. You could cover that this way: /(?:(?!foo)...|^.{0,2})bar/. Sometimes it's still easier just to say: 

    if (/bar/ && $` !~ /foo$/)

For lookbehind see below. 

(?<=pattern) 

A zero-width positive lookbehind assertion. For example, /(?<=\t)\w+/ matches a word following a tab, without including the tab in $&. Works only for fixed-width lookbehind. 

(?<!pattern) 

A zero-width negative lookbehind assertion. For example /(?<!bar)foo/ matches any occurrence of ``foo'' that isn't following ``bar''. Works only for fixed-width lookbehind. 

(?{ code }) 

Experimental ``evaluate any Perl code'' zero-width assertion. Always succeeds. code is not interpolated. Currently the rules to determine where the code ends are somewhat convoluted. 

The code is properly scoped in the following sense: if the assertion is backtracked (compare Backtracking), all the changes introduced after localisation are undone, so 

  $_ = 'a' x 8;

  m< 

     (?{ $cnt = 0 })                    # Initialize $cnt.

     (

       a 

       (?{

           local $cnt = $cnt + 1;       # Update $cnt, backtracking-safe.

       })

     )*  

     aaaa

     (?{ $res = $cnt })                 # On success copy to non-localized

                                        # location.

   >x;

will set $res = 4. Note that after the match $cnt returns to the globally introduced value 0, since the scopes which restrict local statements are unwound. 

This assertion may be used as (?(condition)yes-pattern) switch. If not used in this way, the result of evaluation of code is put into variable $^R. This happens immediately, so $^R can be used from other (?{ code }) assertions inside the same regular expression. 

The above assignment to $^R is properly localized, thus the old value of $^R is restored if the assertion is backtracked (compare Backtracking). 

Due to security concerns, this construction is not allowed if the regular expression involves run-time interpolation of variables, unless use re 'eval' pragma is used (see the re manpage), or the variables contain results of qr() operator (see qr/STRING/imosx). 

This restriction is due to the wide-spread (questionable) practice of using the construct 

    $re = <>;

    chomp $re;

    $string =~ /$re/;

without tainting. While this code is frowned upon from security point of view, when (?{}) was introduced, it was considered bad to add new security holes to existing scripts. 

NOTE: Use of the above insecure snippet without also enabling taint mode is to be severely frowned upon. use re 'eval' does not disable tainting checks, thus to allow $re in the above snippet to contain (?{}) with tainting enabled, one needs both use re 'eval' and untaint the $re. 

(?>pattern) 

An ``independent'' subexpression. Matches the substring that a standalone pattern would match if anchored at the given position, and only this substring. 

Say, ^(?>a*)ab will never match, since (?>a*) (anchored at the beginning of string, as above) will match all characters a at the beginning of string, leaving no a for ab to match. In contrast, a*ab will match the same as a+b, since the match of the subgroup a* is influenced by the following group ab (see Backtracking). In particular, a* inside a*ab will match fewer characters than a standalone a*, since this makes the tail match. 

An effect similar to (?>pattern) may be achieved by 

   (?=(pattern))\1

since the lookahead is in "logical" context, thus matches the same substring as a standalone a+. The following \1 eats the matched string, thus making a zero-length assertion into an analogue of (?>...). (The difference between these two constructs is that the second one uses a catching group, thus shifting ordinals of backreferences in the rest of a regular expression.) 

This construct is useful for optimizations of ``eternal'' matches, because it will not backtrack (see Backtracking). 

    m{ \(

          ( 

            [^()]+ 

          | 

            \( [^()]* \)

          )+

       \) 

     }x

That will efficiently match a nonempty group with matching two-or-less-level-deep parentheses. However, if there is no such group, it will take virtually forever on a long string. That's because there are so many different ways to split a long string into several substrings. This is what (.+)+ is doing, and (.+)+ is similar to a subpattern of the above pattern. Consider that the above pattern detects no-match on ((()aaaaaaaaaaaaaaaaaa in several seconds, but that each extra letter doubles this time. This exponential performance will make it appear that your program has hung. 

However, a tiny modification of this pattern 

    m{ \( 

          ( 

            (?> [^()]+ )

          | 

            \( [^()]* \)

          )+

       \) 

     }x

which uses (?>...) matches exactly when the one above does (verifying this yourself would be a productive exercise), but finishes in a fourth the time when used on a similar string with 1000000 as. Be aware, however, that this pattern currently triggers a warning message under -w saying it "matches the null string many times"): 

On simple groups, such as the pattern (? [^()]+ )>, a comparable effect may be achieved by negative lookahead, as in [^()]+ (?! [^()] ). This was only 4 times slower on a string with 1000000 as. 

(?(condition)yes-pattern|no-pattern) 

(?(condition)yes-pattern) 

Conditional expression. (condition) should be either an integer in parentheses (which is valid if the corresponding pair of parentheses matched), or lookahead/lookbehind/evaluate zero-width assertion. 

Say, 

    m{ ( \( )? 

       [^()]+ 

       (?(1) \) ) 

     }x

matches a chunk of non-parentheses, possibly included in parentheses themselves. 

(?imsx-imsx) 

One or more embedded pattern-match modifiers. This is particularly useful for patterns that are specified in a table somewhere, some of which want to be case sensitive, and some of which don't. The case insensitive ones need to include merely (?i) at the front of the pattern. For example: 

    $pattern = "foobar";

    if ( /$pattern/i ) { } 

    # more flexible:

    $pattern = "(?i)foobar";

    if ( /$pattern/ ) { } 

Letters after - switch modifiers off. 

These modifiers are localized inside an enclosing group (if any). Say, 

    ( (?i) blah ) \s+ \1

(assuming x modifier, and no i modifier outside of this group) will match a repeated (including the case!) word blah in any case. 

A question mark was chosen for this and for the new minimal-matching construct because 1) question mark is pretty rare in older regular expressions, and 2) whenever you see one, you should stop and ``question'' exactly what is going on. That's psychology... 



Backtracking

A fundamental feature of regular expression matching involves the notion called backtracking, which is currently used (when needed) by all regular expression quantifiers, namely *, *?, +, +?, {n,m}, and {n,m}?. 

For a regular expression to match, the entire regular expression must match, not just part of it. So if the beginning of a pattern containing a quantifier succeeds in a way that causes later parts in the pattern to fail, the matching engine backs up and recalculates the beginning part--that's why it's called backtracking. 

Here is an example of backtracking: Let's say you want to find the word following ``foo'' in the string ``Food is on the foo table.'': 

    $_ = "Food is on the foo table.";

    if ( /\b(foo)\s+(\w+)/i ) {

        print "$2 follows $1.\n";

    }

When the match runs, the first part of the regular expression (\b(foo)) finds a possible match right at the beginning of the string, and loads up $1 with ``Foo''. However, as soon as the matching engine sees that there's no whitespace following the ``Foo'' that it had saved in $1, it realizes its mistake and starts over again one character after where it had the tentative match. This time it goes all the way until the next occurrence of ``foo''. The complete regular expression matches this time, and you get the expected output of ``table follows foo.'' 

Sometimes minimal matching can help a lot. Imagine you'd like to match everything between ``foo'' and ``bar''. Initially, you write something like this: 

    $_ =  "The food is under the bar in the barn.";

    if ( /foo(.*)bar/ ) {

        print "got <$1>\n";

    }

Which perhaps unexpectedly yields: 

  got <d is under the bar in the >

That's because .* was greedy, so you get everything between the first ``foo'' and the last ``bar''. In this case, it's more effective to use minimal matching to make sure you get the text between a ``foo'' and the first ``bar'' thereafter. 

    if ( /foo(.*?)bar/ ) { print "got <$1>\n" }

  got <d is under the >

Here's another example: let's say you'd like to match a number at the end of a string, and you also want to keep the preceding part the match. So you write this: 

    $_ = "I have 2 numbers: 53147";

    if ( /(.*)(\d*)/ ) {                                # Wrong!

        print "Beginning is <$1>, number is <$2>.\n";

    }

That won't work at all, because .* was greedy and gobbled up the whole string. As \d* can match on an empty string the complete regular expression matched successfully. 

    Beginning is <I have 2 numbers: 53147>, number is <>.

Here are some variants, most of which don't work: 

    $_ = "I have 2 numbers: 53147";

    @pats = qw{

        (.*)(\d*)

        (.*)(\d+)

        (.*?)(\d*)

        (.*?)(\d+)

        (.*)(\d+)$

        (.*?)(\d+)$

        (.*)\b(\d+)$

        (.*\D)(\d+)$

    };

    for $pat (@pats) {

        printf "%-12s ", $pat;

        if ( /$pat/ ) {

            print "<$1> <$2>\n";

        } else {

            print "FAIL\n";

        }

    }

That will print out: 

    (.*)(\d*)    <I have 2 numbers: 53147> <>

    (.*)(\d+)    <I have 2 numbers: 5314> <7>

    (.*?)(\d*)   <> <>

    (.*?)(\d+)   <I have > <2>

    (.*)(\d+)$   <I have 2 numbers: 5314> <7>

    (.*?)(\d+)$  <I have 2 numbers: > <53147>

    (.*)\b(\d+)$ <I have 2 numbers: > <53147>

    (.*\D)(\d+)$ <I have 2 numbers: > <53147>

As you see, this can be a bit tricky. It's important to realize that a regular expression is merely a set of assertions that gives a definition of success. There may be 0, 1, or several different ways that the definition might succeed against a particular string. And if there are multiple ways it might succeed, you need to understand backtracking to know which variety of success you will achieve. 

When using lookahead assertions and negations, this can all get even tricker. Imagine you'd like to find a sequence of non-digits not followed by ``123''. You might try to write that as 

    $_ = "ABC123";

    if ( /^\D*(?!123)/ ) {              # Wrong!

        print "Yup, no 123 in $_\n";

    }

But that isn't going to match; at least, not the way you're hoping. It claims that there is no 123 in the string. Here's a clearer picture of why it that pattern matches, contrary to popular expectations: 

    $x = 'ABC123' ;

    $y = 'ABC445' ;

    print "1: got $1\n" if $x =~ /^(ABC)(?!123)/ ;

    print "2: got $1\n" if $y =~ /^(ABC)(?!123)/ ;

    print "3: got $1\n" if $x =~ /^(\D*)(?!123)/ ;

    print "4: got $1\n" if $y =~ /^(\D*)(?!123)/ ;

This prints 

    2: got ABC

    3: got AB

    4: got ABC

You might have expected test 3 to fail because it seems to a more general purpose version of test 1. The important difference between them is that test 3 contains a quantifier (\D*) and so can use backtracking, whereas test 1 will not. What's happening is that you've asked ``Is it true that at the start of $x, following 0 or more non-digits, you have something that's not 123?'' If the pattern matcher had let \D* expand to ``ABC'', this would have caused the whole pattern to fail. The search engine will initially match \D* with ``ABC''. Then it will try to match (?!123 with ``123'', which of course fails. But because a quantifier (\D*) has been used in the regular expression, the search engine can backtrack and retry the match differently in the hope of matching the complete regular expression. 

The pattern really, really wants to succeed, so it uses the standard pattern back-off-and-retry and lets \D* expand to just ``AB'' this time. Now there's indeed something following ``AB'' that is not ``123''. It's in fact ``C123'', which suffices. 

We can deal with this by using both an assertion and a negation. We'll say that the first part in $1 must be followed by a digit, and in fact, it must also be followed by something that's not ``123''. Remember that the lookaheads are zero-width expressions--they only look, but don't consume any of the string in their match. So rewriting this way produces what you'd expect; that is, case 5 will fail, but case 6 succeeds: 

    print "5: got $1\n" if $x =~ /^(\D*)(?=\d)(?!123)/ ;

    print "6: got $1\n" if $y =~ /^(\D*)(?=\d)(?!123)/ ;

    6: got ABC

In other words, the two zero-width assertions next to each other work as though they're ANDed together, just as you'd use any builtin assertions: /^$/ matches only if you're at the beginning of the line AND the end of the line simultaneously. The deeper underlying truth is that juxtaposition in regular expressions always means AND, except when you write an explicit OR using the vertical bar. /ab/ means match ``a'' AND (then) match ``b'', although the attempted matches are made at different positions because ``a'' is not a zero-width assertion, but a one-width assertion. 

One warning: particularly complicated regular expressions can take exponential time to solve due to the immense number of possible ways they can use backtracking to try match. For example this will take a very long time to run 

    /((a{0,5}){0,5}){0,5}/

And if you used *'s instead of limiting it to 0 through 5 matches, then it would take literally forever--or until you ran out of stack space. 

A powerful tool for optimizing such beasts is ``independent'' groups, which do not backtrace (see (?>pattern)). Note also that zero-length lookahead/lookbehind assertions will not backtrace to make the tail match, since they are in ``logical'' context: only the fact whether they match or not is considered relevant. For an example where side-effects of a lookahead might have influenced the following match, see (?>pattern). 



Version 8 Regular Expressions

In case you're not familiar with the ``regular'' Version 8 regex routines, here are the pattern-matching rules not described above. 

Any single character matches itself, unless it is a metacharacter with a special meaning described here or above. You can cause characters that normally function as metacharacters to be interpreted literally by prefixing them with a ``\'' (e.g., ``\.'' matches a ``.'', not any character; ``\\'' matches a ``\''). A series of characters matches that series of characters in the target string, so the pattern blurfl would match ``blurfl'' in the target string. 

You can specify a character class, by enclosing a list of characters in [], which will match any one character from the list. If the first character after the ``['' is ``^'', the class matches any character not in the list. Within a list, the ``-'' character is used to specify a range, so that a-z represents all characters between ``a'' and ``z'', inclusive. If you want ``-'' itself to be a member of a class, put it at the start or end of the list, or escape it with a backslash. (The following all specify the same class of three characters: [-az], [az-], and [a\-z]. All are different from [a-z], which specifies a class containing twenty-six characters.) 

Characters may be specified using a metacharacter syntax much like that used in C: ``\n'' matches a newline, ``\t'' a tab, ``\r'' a carriage return, ``\f'' a form feed, etc. More generally, \ nnn, where nnn is a string of octal digits, matches the character whose ASCII value is nnn. Similarly, \xnn, where nn are hexadecimal digits, matches the character whose ASCII value is nn. The expression \cx matches the ASCII character control-x. Finally, the ``.'' metacharacter matches any character except ``\n'' (unless you use /s). 

You can specify a series of alternatives for a pattern using ``|'' to separate them, so that fee|fie|foe will match any of ``fee'', ``fie'', or ``foe'' in the target string (as would f(e|i|o)e). The first alternative includes everything from the last pattern delimiter (``('', ``['', or the beginning of the pattern) up to the first ``|'', and the last alternative contains everything from the last ``|'' to the next pattern delimiter. For this reason, it's common practice to include alternatives in parentheses, to minimize confusion about where they start and end. 

Alternatives are tried from left to right, so the first alternative found for which the entire expression matches, is the one that is chosen. This means that alternatives are not necessarily greedy. For example: when mathing foo|foot against ``barefoot'', only the ``foo'' part will match, as that is the first alternative tried, and it successfully matches the target string. (This might not seem important, but it is important when you are capturing matched text using parentheses.) 

Also remember that ``|'' is interpreted as a literal within square brackets, so if you write [fee|fie|foe] you're really only matching [feio|]. 

Within a pattern, you may designate subpatterns for later reference by enclosing them in parentheses, and you may refer back to the nth subpattern later in the pattern using the metacharacter \n. Subpatterns are numbered based on the left to right order of their opening parenthesis. A backreference matches whatever actually matched the subpattern in the string being examined, not the rules for that subpattern. Therefore, (0|0x)\d*\s\1\d* will match ``0x1234 0x4321'', but not ``0x1234 01234'', because subpattern 1 actually matched ``0x'', even though the rule 0|0x could potentially match the leading 0 in the second number. 



WARNING on \1 vs $1

Some people get too used to writing things like: 

    $pattern =~ s/(\W)/\\\1/g;

This is grandfathered for the RHS of a substitute to avoid shocking the sed addicts, but it's a dirty habit to get into. That's because in PerlThink, the righthand side of a s/// is a double-quoted string. \1 in the usual double-quoted string means a control-A. The customary Unix meaning of \1 is kludged in for s///. However, if you get into the habit of doing that, you get yourself into trouble if you then add an /e modifier. 

    s/(\d+)/ \1 + 1 /eg;        # causes warning under -w

Or if you try to do 

    s/(\d+)/\1000/;

You can't disambiguate that by saying \{1}000, whereas you can fix it with ${1}000. Basically, the operation of interpolation should not be confused with the operation of matching a backreference. Certainly they mean two different things on the left side of the s///. 



Repeated patterns matching zero-length substring

WARNING: Difficult material (and prose) ahead. This section needs a rewrite. 

Regular expressions provide a terse and powerful programming language. As with most other power tools, power comes together with the ability to wreak havoc. 

A common abuse of this power stems from the ability to make infinite loops using regular expressions, with something as innocous as: 

    'foo' =~ m{ ( o? )* }x;

The o? can match at the beginning of 'foo', and since the position in the string is not moved by the match, o? would match again and again due to the * modifier. Another common way to create a similar cycle is with the looping modifier //g: 

    @matches = ( 'foo' =~ m{ o? }xg );

or 

    print "match: <$&>\n" while 'foo' =~ m{ o? }xg;

or the loop implied by split(). 

However, long experience has shown that many programming tasks may be significantly simplified by using repeated subexpressions which may match zero-length substrings, with a simple example being: 

    @chars = split //, $string;           # // is not magic in split

    ($whitewashed = $string) =~ s/()/ /g; # parens avoid magic s// /

Thus Perl allows the /()/ construct, which forcefully breaks the infinite loop. The rules for this are different for lower-level loops given by the greedy modifiers *+{}, and for higher-level ones like the /g modifier or split() operator. 

The lower-level loops are interrupted when it is detected that a repeated expression did match a zero-length substring, thus 

   m{ (?: NON_ZERO_LENGTH | ZERO_LENGTH )* }x;

is made equivalent to 

   m{   (?: NON_ZERO_LENGTH )* 

      | 

        (?: ZERO_LENGTH )? 

    }x;

The higher level-loops preserve an additional state between iterations: whether the last match was zero-length. To break the loop, the following match after a zero-length match is prohibited to have a length of zero. This prohibition interacts with backtracking (see Backtracking), and so the second best match is chosen if the best match is of zero length. 

Say, 

    $_ = 'bar';

    s/\w??/<$&>/g;

results in "<<b><><a><><r><>``>. At each position of the string the best match given by non-greedy ?? is the zero-length match, and the second best match is what is matched by \w. Thus zero-length matches alternate with one-character-long matches. 

Similarly, for repeated m/()/g the second-best match is the match at the position one notch further in the string. 

The additional state of being matched with zero-length is associated to the matched string, and is reset by each assignment to pos(). 



Creating custom RE engines

Overloaded constants (see the overload manpage) provide a simple way to extend the functionality of the RE engine. 

Suppose that we want to enable a new RE escape-sequence \Y| which matches at boundary between white-space characters and non-whitespace characters. Note that (?=\S)(?<!\S)|(?!\S)(?<=\S) matches exactly at these positions, so we want to have each \Y| in the place of the more complicated version. We can create a module customre to do this: 

    package customre;

    use overload;

    sub import {

      shift;

      die "No argument to customre::import allowed" if @_;

      overload::constant 'qr' => \&convert;

    }

    sub invalid { die "/$_[0]/: invalid escape '\\$_[1]'"}

    my %rules = ( '\\' => '\\', 

                  'Y|' => qr/(?=\S)(?<!\S)|(?!\S)(?<=\S)/ );

    sub convert {

      my $re = shift;

      $re =~ s{ 

                \\ ( \\ | Y . )

              }

              { $rules{$1} or invalid($re,$1) }sgex; 

      return $re;

    }

Now use customre enables the new escape in constant regular expressions, i.e., those without any runtime variable interpolations. As documented in the overload manpage, this conversion will work only over literal parts of regular expressions. For \Y|$re\Y| the variable part of this regular expression needs to be converted explicitly (but only if the special meaning of \Y| should be enabled inside $re): 

    use customre;

    $re = <>;

    chomp $re;

    $re = customre::convert $re;

    /\Y|$re\Y|/;



SEE ALSO

Regexp Quote-Like Operators. 

Gory details of parsing quoted constructs. 

pos. 

the perllocale manpage. 

Mastering Regular Expressions (see the perlbook manpage) by Jeffrey Friedl. 



DISCLAIMER

We are painfully aware that these documents may contain incorrect links and misformatted HTML. Such bugs lie in the automatic translation process that automatically created the hundreds and hundreds of separate documents that you find here. Please do not report link or formatting bugs, because we cannot fix per-document problems. The only bug reports that will help us are those that supply working patches to the installhtml or pod2html programs, or to the Pod::HTML module itself, for which I and the entire Perl community will shower you with thanks and praises. 

If rather than formatting bugs, you encounter substantive content errors in these documents, such as mistakes in the explanations or code, please use the perlbug utility included with the Perl distribution. 

--Tom Christiansen, Perl Documentation Compiler and Editor

perlport - Writing portable Perl 



DESCRIPTION

Perl runs on a variety of operating systems. While most of them share a lot in common, they also have their own very particular and unique features. 

This document is meant to help you to find out what constitutes portable Perl code, so that once you have made your decision to write portably, you know where the lines are drawn, and you can stay within them. 

There is a tradeoff between taking full advantage of a particular type of computer, and taking advantage of a full range of them. Naturally, as you make your range bigger (and thus more diverse), the common denominators drop, and you are left with fewer areas of common ground in which you can operate to accomplish a particular task. Thus, when you begin attacking a problem, it is important to consider which part of the tradeoff curve you want to operate under. Specifically, whether it is important to you that the task that you are coding needs the full generality of being portable, or if it is sufficient to just get the job done. This is the hardest choice to be made. The rest is easy, because Perl provides lots of choices, whichever way you want to approach your problem. 

Looking at it another way, writing portable code is usually about willfully limiting your available choices. Naturally, it takes discipline to do that. 

Be aware of two important points: 

Not all Perl programs have to be portable 

There is no reason why you should not use Perl as a language to glue Unix tools together, or to prototype a Macintosh application, or to manage the Windows registry. If it makes no sense to aim for portability for one reason or another in a given program, then don't bother. 

The vast majority of Perl is portable 

Don't be fooled into thinking that it is hard to create portable Perl code. It isn't. Perl tries its level-best to bridge the gaps between what's available on different platforms, and all the means available to use those features. Thus almost all Perl code runs on any machine without modification. But there are some significant issues in writing portable code, and this document is entirely about those issues. 

Here's the general rule: When you approach a task that is commonly done using a whole range of platforms, think in terms of writing portable code. That way, you don't sacrifice much by way of the implementation choices you can avail yourself of, and at the same time you can give your users lots of platform choices. On the other hand, when you have to take advantage of some unique feature of a particular platform, as is often the case with systems programming (whether for Unix, Windows, Mac OS, VMS, etc.), consider writing platform-specific code. 

When the code will run on only two or three operating systems, then you may only need to consider the differences of those particular systems. The important thing is to decide where the code will run, and to be deliberate in your decision. 

The material below is separated into three main sections: main issues of portability (ISSUES, platform-specific issues (PLATFORMS, and builtin perl functions that behave differently on various ports (FUNCTION IMPLEMENTATIONS. 

This information should not be considered complete; it includes possibly transient information about idiosyncrasies of some of the ports, almost all of which are in a state of constant evolution. Thus this material should be considered a perpetual work in progress (<IMG SRC=``yellow_sign.gif'' ALT=``Under Construction''>). 



ISSUES



Newlines

In most operating systems, lines in files are terminated with newlines. Just what is used as a newline may vary from OS to OS. Unix traditionally uses \012, one kind of Windows I/O uses \015\012, and Mac OS uses \015. 

Perl uses \n to represent the ``logical'' newline, where what is logical may depend on the platform in use. In MacPerl, \n always means \015. In DOSish perls, \n usually means \012, but when accessing a file in ``text'' mode, STDIO translates it to (or from) \015\012. 

Due to the ``text'' mode translation, DOSish perls have limitations of using seek and tell when a file is being accessed in ``text'' mode. Specifically, if you stick to seek-ing to locations you got from tell (and no others), you are usually free to use seek and tell even in ``text'' mode. In general, using seek or tell or other file operations that count bytes instead of characters, without considering the length of \n, may be non-portable. If you use binmode on a file, however, you can usually use seek and tell with arbitrary values quite safely. 

A common misconception in socket programming is that \n eq \012 everywhere. When using protocols such as common Internet protocols, \012 and \015 are called for specifically, and the values of the logical \n and \r (carriage return) are not reliable. 

    print SOCKET "Hi there, client!\r\n";      # WRONG

    print SOCKET "Hi there, client!\015\012";  # RIGHT

[NOTE: this does not necessarily apply to communications that are filtered by another program or module before sending to the socket; the the most popular EBCDIC webserver, for instance, accepts \r\n, which translates those characters, along with all other characters in text streams, from EBCDIC to ASCII.] 

However, using \015\012 (or \cM\cJ, or \x0D\x0A) can be tedious and unsightly, as well as confusing to those maintaining the code. As such, the Socket module supplies the Right Thing for those who want it. 

    use Socket qw(:DEFAULT :crlf);

    print SOCKET "Hi there, client!$CRLF"      # RIGHT

When reading from a socket, remember that the default input record separator ($/) is \n, but code like this should recognize $/ as \012 or \015\012: 

    while (<SOCKET>) {

        # ...

    }

Better: 

    use Socket qw(:DEFAULT :crlf);

    local($/) = LF;      # not needed if $/ is already \012

    while (<SOCKET>) {

        s/$CR?$LF/\n/;   # not sure if socket uses LF or CRLF, OK

    #   s/\015?\012/\n/; # same thing

    }

And this example is actually better than the previous one even for Unix platforms, because now any \015's (\cM's) are stripped out (and there was much rejoicing). 



Numbers endianness and Width

Different CPUs store integers and floating point numbers in different orders (called endianness) and widths (32-bit and 64-bit being the most common). This affects your programs if they attempt to transfer numbers in binary format from a CPU architecture to another over some channel: either 'live' via network connections or storing the numbers to secondary storage such as a disk file. 

Conflicting storage orders make utter mess out of the numbers: if a little-endian host (Intel, Alpha) stores 0x12345678 (305419896 in decimal), a big-endian host (Motorola, MIPS, Sparc, PA) reads it as 0x78563412 (2018915346 in decimal). To avoid this problem in network (socket) connections use the pack() and unpack() formats "n" and "N", the ``network'' orders, they are guaranteed to be portable. 

Different widths can cause truncation even between platforms of equal endianness: the platform of shorter width loses the upper parts of the number. There is no good solution for this problem except to avoid transferring or storing raw binary numbers. 

One can circumnavigate both these problems in two ways: either transfer and store numbers always in text format, instead of raw binary, or consider using modules like Data::Dumper (included in the standard distribution as of Perl 5.005) and Storable. 



Files

Most platforms these days structure files in a hierarchical fashion. So, it is reasonably safe to assume that any platform supports the notion of a ``path'' to uniquely identify a file on the system. Just how that path is actually written, differs. 

While they are similar, file path specifications differ between Unix, Windows, Mac OS, OS/2, VMS, RISC OS and probably others. Unix, for example, is one of the few OSes that has the idea of a single root directory. 

VMS, Windows, and OS/2 can work similarly to Unix with / as path separator, or in their own idiosyncratic ways (such as having several root directories and various ``unrooted'' device files such NIL: and LPT:). 

Mac OS uses : as a path separator instead of /. 

RISC OS perl can emulate Unix filenames with / as path separator, or go native and use . for path separator and : to signal filing systems and disc names. 

As with the newline problem above, there are modules that can help. The File::Spec modules provide methods to do the Right Thing on whatever platform happens to be running the program. 

    use File::Spec;

    chdir(File::Spec->;updir());        # go up one directory

    $file = File::Spec->;catfile(

        File::Spec->;curdir(), 'temp', 'file.txt'

    );

    # on Unix and Win32, './temp/file.txt'

    # on Mac OS, ':temp:file.txt'

File::Spec is available in the standard distribution, as of version 5.004_05. 

In general, production code should not have file paths hardcoded; making them user supplied or from a configuration file is better, keeping in mind that file path syntax varies on different machines. 

This is especially noticeable in scripts like Makefiles and test suites, which often assume / as a path separator for subdirectories. 

Also of use is File::Basename, from the standard distribution, which splits a pathname into pieces (base filename, full path to directory, and file suffix). 

Even when on a single platform (if you can call UNIX a single platform), remember not to count on the existence or the contents of system-specific files, like /etc/passwd, /etc/sendmail.conf, or /etc/resolv.conf. For example the /etc/passwd may exist but it may not contain the encrypted passwords because the system is using some form of enhanced security-- or it may not contain all the accounts because the system is using NIS. If code does need to rely on such a file, include a description of the file and its format in the code's documentation, and make it easy for the user to override the default location of the file. 

Do not have two files of the same name with different case, like test.pl and <Test.pl>, as many platforms have case-insensitive filenames. Also, try not to have non-word characters (except for .) in the names, and keep them to the 8.3 convention, for maximum portability. 

Likewise, if using AutoSplit, try to keep the split functions to 8.3 naming and case-insensitive conventions; or, at the very least, make it so the resulting files have a unique (case-insensitively) first 8 characters. 

Don't assume < won't be the first character of a filename. Always use > explicitly to open a file for reading: 

    open(FILE, "<$existing_file") or die $!;



System Interaction

Not all platforms provide for the notion of a command line, necessarily. These are usually platforms that rely on a Graphical User Interface (GUI) for user interaction. So a program requiring command lines might not work everywhere. But this is probably for the user of the program to deal with. 

Some platforms can't delete or rename files that are being held open by the system. Remember to close files when you are done with them. Don't unlink or rename an open file. Don't tie to or open a file that is already tied to or opened; untie or close first. 

Don't open the same file more than once at a time for writing, as some operating systems put mandatory locks on such files. 

Don't count on a specific environment variable existing in %ENV. Don't count on %ENV entries being case-sensitive, or even case-preserving. 

Don't count on signals. 

Don't count on filename globbing. Use opendir, readdir, and closedir instead. 

Don't count on per-program environment variables, or per-program current directories. 



Interprocess Communication (IPC)

In general, don't directly access the system in code that is meant to be portable. That means, no system, exec, fork, pipe, ``, qx//, open with a |, nor any of the other things that makes being a Unix perl hacker worth being. 

Commands that launch external processes are generally supported on most platforms (though many of them do not support any type of forking), but the problem with using them arises from what you invoke with them. External tools are often named differently on different platforms, often not available in the same location, often accept different arguments, often behave differently, and often represent their results in a platform-dependent way. Thus you should seldom depend on them to produce consistent results. 

One especially common bit of Perl code is opening a pipe to sendmail: 

    open(MAIL, '|/usr/lib/sendmail -t') or die $!;

This is fine for systems programming when sendmail is known to be available. But it is not fine for many non-Unix systems, and even some Unix systems that may not have sendmail installed. If a portable solution is needed, see the Mail::Send and Mail::Mailer modules in the MailTools distribution. Mail::Mailer provides several mailing methods, including mail, sendmail, and direct SMTP (via Net::SMTP) if a mail transfer agent is not available. 

The rule of thumb for portable code is: Do it all in portable Perl, or use a module (that may internally implement it with platform-specific code, but expose a common interface). 

The UNIX System V IPC ( msg*(), sem*(), shm*()) is not available even in all UNIX platforms. 



External Subroutines (XS)

XS code, in general, can be made to work with any platform; but dependent libraries, header files, etc., might not be readily available or portable, or the XS code itself might be platform-specific, just as Perl code might be. If the libraries and headers are portable, then it is normally reasonable to make sure the XS code is portable, too. 

There is a different kind of portability issue with writing XS code: availability of a C compiler on the end-user's system. C brings with it its own portability issues, and writing XS code will expose you to some of those. Writing purely in perl is a comparatively easier way to achieve portability. 



Standard Modules

In general, the standard modules work across platforms. Notable exceptions are CPAN.pm (which currently makes connections to external programs that may not be available), platform-specific modules (like ExtUtils::MM_VMS), and DBM modules. 

There is no one DBM module that is available on all platforms. SDBM_File and the others are generally available on all Unix and DOSish ports, but not in MacPerl, where only NBDM_File and DB_File are available. 

The good news is that at least some DBM module should be available, and AnyDBM_File will use whichever module it can find. Of course, then the code needs to be fairly strict, dropping to the lowest common denominator (e.g., not exceeding 1K for each record). 



Time and Date

The system's notion of time of day and calendar date is controlled in widely different ways. Don't assume the timezone is stored in $ENV{TZ}, and even if it is, don't assume that you can control the timezone through that variable. 

Don't assume that the epoch starts at 00:00:00, January 1, 1970, because that is OS-specific. Better to store a date in an unambiguous representation. The ISO 8601 standard defines YYYY-MM-DD as the date format. A text representation (like 1 Jan 1970) can be easily converted into an OS-specific value using a module like Date::Parse. An array of values, such as those returned by localtime, can be converted to an OS-specific representation using Time::Local. 



Character sets and character encoding

Assume very little about character sets. Do not assume anything about the numerical values (ord(), chr()) of characters. Do not assume that the alphabetic characters are encoded contiguously (in numerical sense). Do no assume anything about the ordering of the characters. The lowercase letters may come before or after the uppercase letters, the lowercase and uppercase may be interlaced so that both 'a' and 'A' come before the 'b', the accented and other international characters may be interlaced so that ä comes before the 'b'. 



Internationalisation

If you may assume POSIX (a rather large assumption, that: in practise that means UNIX) you may read more about the POSIX locale system from the perllocale manpage. The locale system at least attempts to make things a little bit more portable or at least more convenient and native-friendly for non-English users. The system affects character sets and encoding, and date and time formatting, among other things. 



System Resources

If your code is destined for systems with severely constrained (or missing!) virtual memory systems then you want to be especially mindful of avoiding wasteful constructs such as: 

    # NOTE: this is no longer "bad" in perl5.005

    for (0..10000000) {}                       # bad

    for (my $x = 0; $x <= 10000000; ++$x) {}   # good

    @lines = <VERY_LARGE_FILE>;                # bad

    while (<FILE>) {$file .= $_}               # sometimes bad

    $file = join('', <FILE>);                  # better

The last two may appear unintuitive to most people. The first of those two constructs repeatedly grows a string, while the second allocates a large chunk of memory in one go. On some systems, the latter is more efficient that the former. 



Security

Most multi-user platforms provide basic levels of security that is usually felt at the file-system level. Other platforms usually don't (unfortunately). Thus the notion of user id, or ``home'' directory, or even the state of being logged-in, may be unrecognizable on many platforms. If you write programs that are security conscious, it is usually best to know what type of system you will be operating under, and write code explicitly for that platform (or class of platforms). 



Style

For those times when it is necessary to have platform-specific code, consider keeping the platform-specific code in one place, making porting to other platforms easier. Use the Config module and the special variable $^O to differentiate platforms, as described in PLATFORMS. 



CPAN Testers

Modules uploaded to CPAN are tested by a variety of volunteers on different platforms. These CPAN testers are notified by mail of each new upload, and reply to the list with PASS, FAIL, NA (not applicable to this platform), or UNKNOWN (unknown), along with any relevant notations. 

The purpose of the testing is twofold: one, to help developers fix any problems in their code that crop up because of lack of testing on other platforms; two, to provide users with information about whether or not a given module works on a given platform. 

Mailing list: cpan-testers@perl.org 

Testing results: http://www.connect.net/gbarr/cpan-test/ 



PLATFORMS

As of version 5.002, Perl is built with a $^O variable that indicates the operating system it was built on. This was implemented to help speed up code that would otherwise have to use Config; and use the value of $Config{'osname'}. Of course, to get detailed information about the system, looking into %Config is certainly recommended. 



Unix

Perl works on a bewildering variety of Unix and Unix-like platforms (see e.g. most of the files in the hints/ directory in the source code kit). On most of these systems, the value of $^O (hence $Config{'osname'}, too) is determined by lowercasing and stripping punctuation from the first field of the string returned by typing uname -a (or a similar command) at the shell prompt. Here, for example, are a few of the more popular Unix flavors: 

    uname        $^O        $Config{'archname'}

    -------------------------------------------

    AIX          aix        aix

    FreeBSD      freebsd    freebsd-i386    

    Linux        linux      i386-linux

    HP-UX        hpux       PA-RISC1.1

    IRIX         irix       irix

    OSF1         dec_osf    alpha-dec_osf

    SunOS        solaris    sun4-solaris

    SunOS        solaris    i86pc-solaris

    SunOS4       sunos      sun4-sunos

Note that because the $Config{'archname'} may depend on the hardware architecture it may vary quite a lot, much more than the $^O. 



DOS and Derivatives

Perl has long been ported to PC style microcomputers running under systems like PC-DOS, MS-DOS, OS/2, and most Windows platforms you can bring yourself to mention (except for Windows CE, if you count that). Users familiar with COMMAND.COM and/or CMD.EXE style shells should be aware that each of these file specifications may have subtle differences: 

    $filespec0 = "c:/foo/bar/file.txt";

    $filespec1 = "c:\\foo\\bar\\file.txt";

    $filespec2 = 'c:\foo\bar\file.txt';

    $filespec3 = 'c:\\foo\\bar\\file.txt';

System calls accept either / or \ as the path separator. However, many command-line utilities of DOS vintage treat / as the option prefix, so they may get confused by filenames containing /. Aside from calling any external programs, / will work just fine, and probably better, as it is more consistent with popular usage, and avoids the problem of remembering what to backwhack and what not to. 

The DOS FAT filesystem can only accommodate ``8.3'' style filenames. Under the ``case insensitive, but case preserving'' HPFS (OS/2) and NTFS (NT) filesystems you may have to be careful about case returned with functions like readdir or used with functions like open or opendir. 

DOS also treats several filenames as special, such as AUX, PRN, NUL, CON, COM1, LPT1, LPT2 etc. Unfortunately these filenames won't even work if you include an explicit directory prefix, in some cases. It is best to avoid such filenames, if you want your code to be portable to DOS and its derivatives. 

Users of these operating systems may also wish to make use of scripts such as pl2bat.bat or pl2cmd as appropriate to put wrappers around your scripts. 

Newline (\n) is translated as \015\012 by STDIO when reading from and writing to files. binmode(FILEHANDLE) will keep \n translated as \012 for that filehandle. Since it is a noop on other systems, binmode should be used for cross-platform code that deals with binary data. 

The $^O variable and the $Config{'archname'} values for various DOSish perls are as follows: 

    OS            $^O        $Config{'archname'}

    --------------------------------------------

    MS-DOS        dos

    PC-DOS        dos

    OS/2          os2

    Windows 95    MSWin32    MSWin32-x86

    Windows NT    MSWin32    MSWin32-x86

    Windows NT    MSWin32    MSWin32-alpha

    Windows NT    MSWin32    MSWin32-ppc

Also see: 

The djgpp environment for DOS, http://www.delorie.com/djgpp/ 

The EMX environment for DOS, OS/2, etc. emx@iaehv.nl, http://www.juge.com/bbs/Hobb.19.html 

Build instructions for Win32, perlwin32. 

The ActiveState Pages, http://www.activestate.com/ 



Mac OS

Any module requiring XS compilation is right out for most people, because MacPerl is built using non-free (and non-cheap!) compilers. Some XS modules that can work with MacPerl are built and distributed in binary form on CPAN. See MacPerl: Power and Ease and CPAN Testers for more details. 

Directories are specified as: 

    volume:folder:file              for absolute pathnames

    volume:folder:                  for absolute pathnames

    :folder:file                    for relative pathnames

    :folder:                        for relative pathnames

    :file                           for relative pathnames

    file                            for relative pathnames

Files in a directory are stored in alphabetical order. Filenames are limited to 31 characters, and may include any character except :, which is reserved as a path separator. 

Instead of flock, see FSpSetFLock and FSpRstFLock in the Mac::Files module. 

In the MacPerl application, you can't run a program from the command line; programs that expect @ARGV to be populated can be edited with something like the following, which brings up a dialog box asking for the command line arguments. 

    if (!@ARGV) {

        @ARGV = split /\s+/, MacPerl::Ask('Arguments?');

    }

A MacPerl script saved as a droplet will populate @ARGV with the full pathnames of the files dropped onto the script. 

Mac users can use programs on a kind of command line under MPW (Macintosh Programmer's Workshop, a free development environment from Apple). MacPerl was first introduced as an MPW tool, and MPW can be used like a shell: 

    perl myscript.plx some arguments

ToolServer is another app from Apple that provides access to MPW tools from MPW and the MacPerl app, which allows MacPerl programs to use system, backticks, and piped open. 

``Mac OS'' is the proper name for the operating system, but the value in $^O is ``MacOS''. To determine architecture, version, or whether the application or MPW tool version is running, check: 

    $is_app    = $MacPerl::Version =~ /App/;

    $is_tool   = $MacPerl::Version =~ /MPW/;

    ($version) = $MacPerl::Version =~ /^(\S+)/;

    $is_ppc    = $MacPerl::Architecture eq 'MacPPC';

    $is_68k    = $MacPerl::Architecture eq 'Mac68K';

Mac OS X, to be based on NeXT's OpenStep OS, will be able to run MacPerl natively (in the Blue Box, and even in the Yellow Box, once some changes to the toolbox calls are made), but Unix perl will also run natively. 

Also see: 

The MacPerl Pages, http://www.ptf.com/macperl/. 

The MacPerl mailing list, mac-perl-request@iis.ee.ethz.ch. 



VMS

Perl on VMS is discussed in vms/perlvms.pod in the perl distribution. Note that perl on VMS can accept either VMS- or Unix-style file specifications as in either of the following: 

    $ perl -ne "print if /perl_setup/i" SYS$LOGIN:LOGIN.COM

    $ perl -ne "print if /perl_setup/i" /sys$login/login.com

but not a mixture of both as in: 

    $ perl -ne "print if /perl_setup/i" sys$login:/login.com

    Can't open sys$login:/login.com: file specification syntax error

Interacting with Perl from the Digital Command Language (DCL) shell often requires a different set of quotation marks than Unix shells do. For example: 

    $ perl -e "print ""Hello, world.\n"""

    Hello, world.

There are a number of ways to wrap your perl scripts in DCL .COM files if you are so inclined. For example: 

    $ write sys$output "Hello from DCL!"

    $ if p1 .eqs. ""

    $ then perl -x 'f$environment("PROCEDURE")

    $ else perl -x - 'p1 'p2 'p3 'p4 'p5 'p6 'p7 'p8

    $ deck/dollars="__END__"

    #!/usr/bin/perl

    print "Hello from Perl!\n";

    __END__

    $ endif

Do take care with $ ASSIGN/nolog/user SYS$COMMAND: SYS$INPUT if your perl-in-DCL script expects to do things like $read = <STDIN>;. 

Filenames are in the format ``name.extension;version''. The maximum length for filenames is 39 characters, and the maximum length for extensions is also 39 characters. Version is a number from 1 to 32767. Valid characters are /[A-Z0-9$_-]/. 

VMS' RMS filesystem is case insensitive and does not preserve case. readdir returns lowercased filenames, but specifying a file for opening remains case insensitive. Files without extensions have a trailing period on them, so doing a readdir with a file named A.;5 will return a. (though that file could be opened with open(FH, 'A')). 

RMS had an eight level limit on directory depths from any rooted logical (allowing 16 levels overall) prior to VMS 7.2. Hence PERL_ROOT:[LIB.2.3.4.5.6.7.8] is a valid directory specification but PERL_ROOT:[LIB.2.3.4.5.6.7.8.9] is not. Makefile.PL authors might have to take this into account, but at least they can refer to the former as /PERL_ROOT/lib/2/3/4/5/6/7/8/. 

The VMS::Filespec module, which gets installed as part of the build process on VMS, is a pure Perl module that can easily be installed on non-VMS platforms and can be helpful for conversions to and from RMS native formats. 

What \n represents depends on the type of file that is open. It could be \015, \012, \015\012, or nothing. Reading from a file translates newlines to \012, unless binmode was executed on that handle, just like DOSish perls. 

TCP/IP stacks are optional on VMS, so socket routines might not be implemented. UDP sockets may not be supported. 

The value of $^O on OpenVMS is ``VMS''. To determine the architecture that you are running on without resorting to loading all of %Config you can examine the content of the @INC array like so: 

    if (grep(/VMS_AXP/, @INC)) {

        print "I'm on Alpha!\n";

    } elsif (grep(/VMS_VAX/, @INC)) {

        print "I'm on VAX!\n";

    } else {

        print "I'm not so sure about where $^O is...\n";

    }

Also see: 

perlvms.pod 

vmsperl list, vmsperl-request@newman.upenn.edu 

Put words SUBSCRIBE VMSPERL in message body. 

vmsperl on the web, http://www.sidhe.org/vmsperl/index.html 



EBCDIC Platforms

Recent versions of Perl have been ported to platforms such as OS/400 on AS/400 minicomputers as well as OS/390 for IBM Mainframes. Such computers use EBCDIC character sets internally (usually Character Code Set ID 00819 for OS/400 and IBM-1047 for OS/390). Note that on the mainframe perl currently works under the ``Unix system services for OS/390'' (formerly known as OpenEdition). 

As of R2.5 of USS for OS/390 that Unix sub-system did not support the #! shebang trick for script invocation. Hence, on OS/390 perl scripts can executed with a header similar to the following simple script: 

    : # use perl

        eval 'exec /usr/local/bin/perl -S $0 ${1+"$@"}'

            if 0;

    #!/usr/local/bin/perl     # just a comment really

    print "Hello from perl!\n";

On these platforms, bear in mind that the EBCDIC character set may have an effect on what happens with some perl functions (such as chr, pack, print, printf, ord, sort, sprintf, unpack), as well as bit-fiddling with ASCII constants using operators like ^, & and |, not to mention dealing with socket interfaces to ASCII computers (see NEWLINES). 

Fortunately, most web servers for the mainframe will correctly translate the \n in the following statement to its ASCII equivalent (note that \r is the same under both Unix and OS/390): 

    print "Content-type: text/html\r\n\r\n";

The value of $^O on OS/390 is ``os390''. 

Some simple tricks for determining if you are running on an EBCDIC platform could include any of the following (perhaps all): 

    if ("\t" eq "\05")   { print "EBCDIC may be spoken here!\n"; }

    if (ord('A') == 193) { print "EBCDIC may be spoken here!\n"; }

    if (chr(169) eq 'z') { print "EBCDIC may be spoken here!\n"; }

Note that one thing you may not want to rely on is the EBCDIC encoding of punctuation characters since these may differ from code page to code page (and once your module or script is rumoured to work with EBCDIC, folks will want it to work with all EBCDIC character sets). 

Also see: 

perl-mvs list 

The perl-mvs@perl.org list is for discussion of porting issues as well as general usage issues for all EBCDIC Perls. Send a message body of ``subscribe perl-mvs'' to majordomo@perl.org. 

AS/400 Perl information at http://as400.rochester.ibm.com/ 



Acorn RISC OS

As Acorns use ASCII with newlines (\n) in text files as \012 like Unix and Unix filename emulation is turned on by default, it is quite likely that most simple scripts will work ``out of the box''. The native filing system is modular, and individual filing systems are free to be case-sensitive or insensitive, and are usually case-preserving. Some native filing systems have name length limits which file and directory names are silently truncated to fit - scripts should be aware that the standard disc filing system currently has a name length limit of 10 characters, with up to 77 items in a directory, but other filing systems may not impose such limitations. 

Native filenames are of the form 

    Filesystem#Special_Field::DiscName.$.Directory.Directory.File

where 

    Special_Field is not usually present, but may contain . and $ .

    Filesystem =~ m|[A-Za-z0-9_]|

    DsicName   =~ m|[A-Za-z0-9_/]|

    $ represents the root directory

    . is the path separator

    @ is the current directory (per filesystem but machine global)

    ^ is the parent directory

    Directory and File =~ m|[^\0- "\.\$\%\&:\@\\^\|\177]+|

The default filename translation is roughly tr|/.|./|; 

Note that "ADFS::HardDisc.$.File" ne 'ADFS::HardDisc.$.File' and that the second stage of $ interpolation in regular expressions will fall foul of the $. if scripts are not careful. 

Logical paths specified by system variables containing comma-separated search lists are also allowed, hence System:Modules is a valid filename, and the filesystem will prefix Modules with each section of System$Path until a name is made that points to an object on disc. Writing to a new file System:Modules would only be allowed if System$Path contains a single item list. The filesystem will also expand system variables in filenames if enclosed in angle brackets, so <System$Dir>.Modules would look for the file $ENV{'System$Dir'} . 'Modules'. The obvious implication of this is that B<> and should be protected when open is used for input. 

Because . was in use as a directory separator and filenames could not be assumed to be unique after 10 characters, Acorn implemented the C compiler to strip the trailing .c .h .s and .o suffix from filenames specified in source code and store the respective files in subdirectories named after the suffix. Hence files are translated: 

    foo.h           h.foo

    C:foo.h         C:h.foo        (logical path variable)

    sys/os.h        sys.h.os       (C compiler groks Unix-speak)

    10charname.c    c.10charname

    10charname.o    o.10charname

    11charname_.c   c.11charname   (assuming filesystem truncates at 10)

The Unix emulation library's translation of filenames to native assumes that this sort of translation is required, and allows a user defined list of known suffixes which it will transpose in this fashion. This may appear transparent, but consider that with these rules foo/bar/baz.h and foo/bar/h/baz both map to foo.bar.h.baz, and that readdir and glob cannot and do not attempt to emulate the reverse mapping. Other .s in filenames are translated to /. 

As implied above the environment accessed through %ENV is global, and the convention is that program specific environment variables are of the form Program$Name. Each filing system maintains a current directory, and the current filing system's current directory is the global current directory. Consequently, sociable scripts don't change the current directory but rely on full pathnames, and scripts (and Makefiles) cannot assume that they can spawn a child process which can change the current directory without affecting its parent (and everyone else for that matter). 

As native operating system filehandles are global and currently are allocated down from 255, with 0 being a reserved value the Unix emulation library emulates Unix filehandles. Consequently, you can't rely on passing STDIN, STDOUT, or STDERR to your children. 

The desire of users to express filenames of the form <Foo$Dir>.Bar on the command line unquoted causes problems, too: `` command output capture has to perform a guessing game. It assumes that a string <[^<>]+\$[^<>]> is a reference to an environment variable, whereas anything else involving < or > is redirection, and generally manages to be 99% right. Of course, the problem remains that scripts cannot rely on any Unix tools being available, or that any tools found have Unix-like command line arguments. 

Extensions and XS are, in theory, buildable by anyone using free tools. In practice, many don't, as users of the Acorn platform are used to binary distribution. MakeMaker does run, but no available make currently copes with MakeMaker's makefiles; even if/when this is fixed, the lack of a Unix-like shell can cause problems with makefile rules, especially lines of the form cd sdbm && make all, and anything using quoting. 

``RISC OS'' is the proper name for the operating system, but the value in $^O is ``riscos'' (because we don't like shouting). 

Also see: 

perl list 



Other perls

Perl has been ported to a variety of platforms that do not fit into any of the above categories. Some, such as AmigaOS, BeOS, QNX, and Plan 9, have been well-integrated into the standard Perl source code kit. You may need to see the ports/ directory on CPAN for information, and possibly binaries, for the likes of: aos, atari, lynxos, riscos, Tandem Guardian, vos, etc. (yes we know that some of these OSes may fall under the Unix category, but we are not a standards body.) 

See also: 

Atari, Guido Flohr's page http://stud.uni-sb.de/~gufl0000/ 

HP 300 MPE/iX http://www.cccd.edu/~markb/perlix.html 

Novell Netware 

A free perl5-based PERL.NLM for Novell Netware is available from http://www.novell.com/ 



FUNCTION IMPLEMENTATIONS

Listed below are functions unimplemented or implemented differently on various platforms. Following each description will be, in parentheses, a list of platforms that the description applies to. 

The list may very well be incomplete, or wrong in some places. When in doubt, consult the platform-specific README files in the Perl source distribution, and other documentation resources for a given port. 

Be aware, moreover, that even among Unix-ish systems there are variations. 

For many functions, you can also query %Config, exported by default from Config.pm. For example, to check if the platform has the lstat call, check $Config{'d_lstat'}. See Config.pm for a full description of available variables. 



Alphabetical Listing of Perl Functions

-X FILEHANDLE 

-X EXPR 

-X 

-r, -w, and -x have only a very limited meaning; directories and applications are executable, and there are no uid/gid considerations. -o is not supported. (Mac OS) 

-r, -w, -x, and -o tell whether or not file is accessible, which may not reflect UIC-based file protections. (VMS) 

-s returns the size of the data fork, not the total size of data fork plus resource fork. (Mac OS). 

-s by name on an open file will return the space reserved on disk, rather than the current extent. -s on an open filehandle returns the current size. (RISC OS) 

-R, -W, -X, -O are indistinguishable from -r, -w, -x, -o. (Mac OS, Win32, VMS, RISC OS) 

-b, -c, -k, -g, -p, -u, -A are not implemented. (Mac OS) 

-g, -k, -l, -p, -u, -A are not particularly meaningful. (Win32, VMS, RISC OS) 

-d is true if passed a device spec without an explicit directory. (VMS) 

-T and -B are implemented, but might misclassify Mac text files with foreign characters; this is the case will all platforms, but may affect Mac OS often. (Mac OS) 

-x (or -X) determine if a file ends in one of the executable suffixes. -S is meaningless. (Win32) 

-x (or -X) determine if a file has an executable file type. (RISC OS) 

binmode FILEHANDLE 

Meaningless. (Mac OS, RISC OS) 

Reopens file and restores pointer; if function fails, underlying filehandle may be closed, or pointer may be in a different position. (VMS) 

The value returned by tell may be affected after the call, and the filehandle may be flushed. (Win32) 

chmod LIST 

Only limited meaning. Disabling/enabling write permission is mapped to locking/unlocking the file. (Mac OS) 

Only good for changing ``owner'' read-write access, ``group'', and ``other'' bits are meaningless. (Win32) 

Only good for changing ``owner'' and ``other'' read-write access. (RISC OS) 

chown LIST 

Not implemented. (Mac OS, Win32, Plan9, RISC OS) 

Does nothing, but won't fail. (Win32) 

chroot FILENAME 

chroot 

Not implemented. (Mac OS, Win32, VMS, Plan9, RISC OS) 

crypt PLAINTEXT,SALT 

May not be available if library or source was not provided when building perl. (Win32) 

dbmclose HASH 

Not implemented. (VMS, Plan9) 

dbmopen HASH,DBNAME,MODE 

Not implemented. (VMS, Plan9) 

dump LABEL 

Not useful. (Mac OS, RISC OS) 

Not implemented. (Win32) 

Invokes VMS debugger. (VMS) 

exec LIST 

Not implemented. (Mac OS) 

fcntl FILEHANDLE,FUNCTION,SCALAR 

Not implemented. (Win32, VMS) 

flock FILEHANDLE,OPERATION 

Not implemented (Mac OS, VMS, RISC OS). 

Available only on Windows NT (not on Windows 95). (Win32) 

fork 

Not implemented. (Mac OS, Win32, AmigaOS, RISC OS) 

getlogin 

Not implemented. (Mac OS, RISC OS) 

getpgrp PID 

Not implemented. (Mac OS, Win32, VMS, RISC OS) 

getppid 

Not implemented. (Mac OS, Win32, VMS, RISC OS) 

getpriority WHICH,WHO 

Not implemented. (Mac OS, Win32, VMS, RISC OS) 

getpwnam NAME 

Not implemented. (Mac OS, Win32) 

Not useful. (RISC OS) 

getgrnam NAME 

Not implemented. (Mac OS, Win32, VMS, RISC OS) 

getnetbyname NAME 

Not implemented. (Mac OS, Win32, Plan9) 

getpwuid UID 

Not implemented. (Mac OS, Win32) 

Not useful. (RISC OS) 

getgrgid GID 

Not implemented. (Mac OS, Win32, VMS, RISC OS) 

getnetbyaddr ADDR,ADDRTYPE 

Not implemented. (Mac OS, Win32, Plan9) 

getprotobynumber NUMBER 

Not implemented. (Mac OS) 

getservbyport PORT,PROTO 

Not implemented. (Mac OS) 

getpwent 

Not implemented. (Mac OS, Win32) 

getgrent 

Not implemented. (Mac OS, Win32, VMS) 

gethostent 

Not implemented. (Mac OS, Win32) 

getnetent 

Not implemented. (Mac OS, Win32, Plan9) 

getprotoent 

Not implemented. (Mac OS, Win32, Plan9) 

getservent 

Not implemented. (Win32, Plan9) 

setpwent 

Not implemented. (Mac OS, Win32, RISC OS) 

setgrent 

Not implemented. (Mac OS, Win32, VMS, RISC OS) 

sethostent STAYOPEN 

Not implemented. (Mac OS, Win32, Plan9, RISC OS) 

setnetent STAYOPEN 

Not implemented. (Mac OS, Win32, Plan9, RISC OS) 

setprotoent STAYOPEN 

Not implemented. (Mac OS, Win32, Plan9, RISC OS) 

setservent STAYOPEN 

Not implemented. (Plan9, Win32, RISC OS) 

endpwent 

Not implemented. (Mac OS, Win32) 

endgrent 

Not implemented. (Mac OS, Win32, VMS, RISC OS) 

endhostent 

Not implemented. (Mac OS, Win32) 

endnetent 

Not implemented. (Mac OS, Win32, Plan9) 

endprotoent 

Not implemented. (Mac OS, Win32, Plan9) 

endservent 

Not implemented. (Plan9, Win32) 

getsockopt SOCKET,LEVEL,OPTNAME 

Not implemented. (Mac OS, Plan9) 

glob EXPR 

glob 

Globbing built-in, but only * and ? metacharacters are supported. (Mac OS) 

Features depend on external perlglob.exe or perlglob.bat. May be overridden with something like File::DosGlob, which is recommended. (Win32) 

Globbing built-in, but only * and ? metacharacters are supported. Globbing relies on operating system calls, which may return filenames in any order. As most filesystems are case-insensitive, even ``sorted'' filenames will not be in case-sensitive order. (RISC OS) 

ioctl FILEHANDLE,FUNCTION,SCALAR 

Not implemented. (VMS) 

Available only for socket handles, and it does what the ioctlsocket() call in the Winsock API does. (Win32) 

Available only for socket handles. (RISC OS) 

kill LIST 

Not implemented, hence not useful for taint checking. (Mac OS, RISC OS) 

Available only for process handles returned by the system(1, ...) method of spawning a process. (Win32) 

link OLDFILE,NEWFILE 

Not implemented. (Mac OS, Win32, VMS, RISC OS) 

lstat FILEHANDLE 

lstat EXPR 

lstat 

Not implemented. (VMS, RISC OS) 

Return values may be bogus. (Win32) 

msgctl ID,CMD,ARG 

msgget KEY,FLAGS 

msgsnd ID,MSG,FLAGS 

msgrcv ID,VAR,SIZE,TYPE,FLAGS 

Not implemented. (Mac OS, Win32, VMS, Plan9, RISC OS) 

open FILEHANDLE,EXPR 

open FILEHANDLE 

The | variants are only supported if ToolServer is installed. (Mac OS) 

open to |- and -| are unsupported. (Mac OS, Win32, RISC OS) 

pipe READHANDLE,WRITEHANDLE 

Not implemented. (Mac OS) 

readlink EXPR 

readlink 

Not implemented. (Win32, VMS, RISC OS) 

select RBITS,WBITS,EBITS,TIMEOUT 

Only implemented on sockets. (Win32) 

Only reliable on sockets. (RISC OS) 

semctl ID,SEMNUM,CMD,ARG 

semget KEY,NSEMS,FLAGS 

semop KEY,OPSTRING 

Not implemented. (Mac OS, Win32, VMS, RISC OS) 

setpgrp PID,PGRP 

Not implemented. (Mac OS, Win32, VMS, RISC OS) 

setpriority WHICH,WHO,PRIORITY 

Not implemented. (Mac OS, Win32, VMS, RISC OS) 

setsockopt SOCKET,LEVEL,OPTNAME,OPTVAL 

Not implemented. (Mac OS, Plan9) 

shmctl ID,CMD,ARG 

shmget KEY,SIZE,FLAGS 

shmread ID,VAR,POS,SIZE 

shmwrite ID,STRING,POS,SIZE 

Not implemented. (Mac OS, Win32, VMS, RISC OS) 

socketpair SOCKET1,SOCKET2,DOMAIN,TYPE,PROTOCOL 

Not implemented. (Mac OS, Win32, VMS, RISC OS) 

stat FILEHANDLE 

stat EXPR 

stat 

mtime and atime are the same thing, and ctime is creation time instead of inode change time. (Mac OS) 

device and inode are not meaningful. (Win32) 

device and inode are not necessarily reliable. (VMS) 

mtime, atime and ctime all return the last modification time. Device and inode are not necessarily reliable. (RISC OS) 

symlink OLDFILE,NEWFILE 

Not implemented. (Win32, VMS, RISC OS) 

syscall LIST 

Not implemented. (Mac OS, Win32, VMS, RISC OS) 

sysopen FILEHANDLE,FILENAME,MODE,PERMS 

The traditional ``0'', ``1'', and ``2'' MODEs are implemented with different numeric values on some systems. The flags exported by Fcntl (O_RDONLY, O_WRONLY, O_RDWR) should work everywhere though. ( Mac OS, OS/390) 

system LIST 

Only implemented if ToolServer is installed. (Mac OS) 

As an optimization, may not call the command shell specified in $ENV{PERL5SHELL}. http://www.perl.com/pub/doc/manual/html/pod/perlfunc/system.html spawns an external process and immediately returns its process designator, without waiting for it to terminate. Return value may be used subsequently in wait or waitpid. (Win32) 

There is no shell to process metacharacters, and the native standard is to pass a command line terminated by ``\n'' ``\r'' or ``\0'' to the spawned program. Redirection such as > foo is performed (if at all) by the run time library of the spawned program. system list will call the Unix emulation library's exec emulation, which attempts to provide emulation of the stdin, stdout, stderr in force in the parent, providing the child program uses a compatible version of the emulation library. scalar will call the native command line direct and no such emulation of a child Unix program will exists. Mileage will vary. (RISC OS) 

times 

Only the first entry returned is nonzero. (Mac OS) 

``cumulative'' times will be bogus. On anything other than Windows NT, ``system'' time will be bogus, and ``user'' time is actually the time returned by the clock() function in the C runtime library. (Win32) 

Not useful. (RISC OS) 

truncate FILEHANDLE,LENGTH 

truncate EXPR,LENGTH 

Not implemented. (VMS) 

umask EXPR 

umask 

Returns undef where unavailable, as of version 5.005. 

utime LIST 

Only the modification time is updated. (Mac OS, VMS, RISC OS) 

May not behave as expected. Behavior depends on the C runtime library's implementation of utime(), and the filesystem being used. The FAT filesystem typically does not support an ``access time'' field, and it may limit timestamps to a granularity of two seconds. (Win32) 

wait 

waitpid PID,FLAGS 

Not implemented. (Mac OS) 

Can only be applied to process handles returned for processes spawned using system(1, ...). (Win32) 

Not useful. (RISC OS) 



CHANGES

1. 33, 06 August 1998 

Integrate more minor changes. 

2. 32, 05 August 1998 Integrate more minor changes. 

3. 30, 03 August 1998 Major update for RISC OS, other minor changes. 

4. 23, 10 July 1998 First public release with perl5.005. 

DISCLAIMER

We are painfully aware that these documents may contain incorrect links and misformatted HTML. Such bugs lie in the automatic translation process that automatically created the hundreds and hundreds of separate documents that you find here. Please do not report link or formatting bugs, because we cannot fix per-document problems. The only bug reports that will help us are those that supply working patches to the installhtml or pod2html programs, or to the Pod::HTML module itself, for which I and the entire Perl community will shower you with thanks and praises. 

If rather than formatting bugs, you encounter substantive content errors in these documents, such as mistakes in the explanations or code, please use the perlbug utility included with the Perl distribution. 

--Tom Christiansen, Perl Documentation Compiler and Editor

NAME

perlstyle - Perl style guide 



DESCRIPTION

Each programmer will, of course, have his or her own preferences in regards to formatting, but there are some general guidelines that will make your programs easier to read, understand, and maintain. 

The most important thing is to run your programs under the -w flag at all times. You may turn it off explicitly for particular portions of code via the $^W variable if you must. You should also always run under use strict or know the reason why not. The use sigtrap and even use diagnostics pragmas may also prove useful. 

Regarding aesthetics of code lay out, about the only thing Larry cares strongly about is that the closing curly brace of a multi-line BLOCK should line up with the keyword that started the construct. Beyond that, he has other preferences that aren't so strong: 

· 4-column indent. 

· Opening curly on same line as keyword, if possible, otherwise line up. 

· Space before the opening curly of a multi-line BLOCK. 

· One-line BLOCK may be put on one line, including curlies. 

· No space before the semicolon. 

· Semicolon omitted in ``short'' one-line BLOCK. 

· Space around most operators. 

· Space around a ``complex'' subscript (inside brackets). 

· Blank lines between chunks that do different things. 

· Uncuddled elses. 

· No space between function name and its opening parenthesis. 

· Space after each comma. 

· Long lines broken after an operator (except ``and'' and ``or''). 

· Space after last parenthesis matching on current line. 

· Line up corresponding items vertically. 

· Omit redundant punctuation as long as clarity doesn't suffer. 

Larry has his reasons for each of these things, but he doesn't claim that everyone else's mind works the same as his does. 

Here are some other more substantive style issues to think about: 

· Just because you CAN do something a particular way doesn't mean that you SHOULD do it that way. Perl is designed to give you several ways to do anything, so consider picking the most readable one. For instance 

·     open(FOO,$foo) || die "Can't open $foo: $!";

is better than 

    die "Can't open $foo: $!" unless open(FOO,$foo);

because the second way hides the main point of the statement in a modifier. On the other hand 

    print "Starting analysis\n" if $verbose;

is better than 

    $verbose && print "Starting analysis\n";

because the main point isn't whether the user typed -v or not. 

Similarly, just because an operator lets you assume default arguments doesn't mean that you have to make use of the defaults. The defaults are there for lazy systems programmers writing one-shot programs. If you want your program to be readable, consider supplying the argument. 

Along the same lines, just because you CAN omit parentheses in many places doesn't mean that you ought to: 

    return print reverse sort num values %array;

    return print(reverse(sort num (values(%array))));

When in doubt, parenthesize. At the very least it will let some poor schmuck bounce on the % key in vi. 

Even if you aren't in doubt, consider the mental welfare of the person who has to maintain the code after you, and who will probably put parentheses in the wrong place. 

· Don't go through silly contortions to exit a loop at the top or the bottom, when Perl provides the last operator so you can exit in the middle. Just ``outdent'' it a little to make it more visible: 

·     LINE:

·         for (;;) {

·             statements;

·           last LINE if $foo;

·             next LINE if /^#/;

·             statements;

·         }

· Don't be afraid to use loop labels--they're there to enhance readability as well as to allow multilevel loop breaks. See the previous example. 

· Avoid using grep() (or map()) or `backticks` in a void context, that is, when you just throw away their return values. Those functions all have return values, so use them. Otherwise use a foreach() loop or the system() function instead. 

· For portability, when using features that may not be implemented on every machine, test the construct in an eval to see if it fails. If you know what version or patchlevel a particular feature was implemented, you can test $] ($PERL_VERSION in English) to see if it will be there. The Config module will also let you interrogate values determined by the Configure program when Perl was installed. 

· Choose mnemonic identifiers. If you can't remember what mnemonic means, you've got a problem. 

· While short identifiers like $gotit are probably ok, use underscores to separate words. It is generally easier to read $var_names_like_this than $VarNamesLikeThis, especially for non-native speakers of English. It's also a simple rule that works consistently with VAR_NAMES_LIKE_THIS. 

Package names are sometimes an exception to this rule. Perl informally reserves lowercase module names for ``pragma'' modules like integer and strict. Other modules should begin with a capital letter and use mixed case, but probably without underscores due to limitations in primitive file systems' representations of module names as files that must fit into a few sparse bytes. 

· You may find it helpful to use letter case to indicate the scope or nature of a variable. For example: 

·     $ALL_CAPS_HERE   constants only (beware clashes with perl vars!)

·     $Some_Caps_Here  package-wide global/static

·     $no_caps_here    function scope my() or local() variables

Function and method names seem to work best as all lowercase. E.g., $obj->as_string(). 

You can use a leading underscore to indicate that a variable or function should not be used outside the package that defined it. 

· If you have a really hairy regular expression, use the /x modifier and put in some whitespace to make it look a little less like line noise. Don't use slash as a delimiter when your regexp has slashes or backslashes. 

· Use the new ``and'' and ``or'' operators to avoid having to parenthesize list operators so much, and to reduce the incidence of punctuation operators like && and ||. Call your subroutines as if they were functions or list operators to avoid excessive ampersands and parentheses. 

· Use here documents instead of repeated print() statements. 

· Line up corresponding things vertically, especially if it'd be too long to fit on one line anyway. 

·     $IDX = $ST_MTIME;

·     $IDX = $ST_ATIME       if $opt_u;

·     $IDX = $ST_CTIME       if $opt_c;

·     $IDX = $ST_SIZE        if $opt_s;

·     mkdir $tmpdir, 0700 or die "can't mkdir $tmpdir: $!";

·     chdir($tmpdir)      or die "can't chdir $tmpdir: $!";

·     mkdir 'tmp',   0777 or die "can't mkdir $tmpdir/tmp: $!";

· Always check the return codes of system calls. Good error messages should go to STDERR, include which program caused the problem, what the failed system call and arguments were, and (VERY IMPORTANT) should contain the standard system error message for what went wrong. Here's a simple but sufficient example: 

·     opendir(D, $dir)     or die "can't opendir $dir: $!";

· Line up your transliterations when it makes sense: 

·     tr [abc]

·        [xyz];

· Think about reusability. Why waste brainpower on a one-shot when you might want to do something like it again? Consider generalizing your code. Consider writing a module or object class. Consider making your code run cleanly with use strict and -w in effect. Consider giving away your code. Consider changing your whole world view. Consider... oh, never mind. 

· Be consistent. 

· Be nice. 



DISCLAIMER

We are painfully aware that these documents may contain incorrect links and misformatted HTML. Such bugs lie in the automatic translation process that automatically created the hundreds and hundreds of separate documents that you find here. Please do not report link or formatting bugs, because we cannot fix per-document problems. The only bug reports that will help us are those that supply working patches to the installhtml or pod2html programs, or to the Pod::HTML module itself, for which I and the entire Perl community will shower you with thanks and praises. 

If rather than formatting bugs, you encounter substantive content errors in these documents, such as mistakes in the explanations or code, please use the perlbug utility included with the Perl distribution. 

--Tom Christiansen, Perl Documentation Compiler and Editor

NAME

perlrun - how to execute the Perl interpreter 



SYNOPSIS

perl [ -sTuU ] [ -hv ] [ -V[:configvar] ] [ -cw ] [ -d[:debugger] ] [ -D[number/list] ] [ -pna ] [ -Fpattern ] [ -l[octal] ] [ -0[octal] ] [ -Idir ] [ -m[-]module ] [ -M[-]'module...' ] [ -P ] [ -S ] [ -x[dir] ] [ -i[extension] ] [ -e 'command' ] [ -- ] [ programfile ] [ argument ]... 



DESCRIPTION

Upon startup, Perl looks for your script in one of the following places: 

1. . 

Specified line by line via -e switches on the command line. 

2. . Contained in the file specified by the first filename on the command line. (Note that systems supporting the #! notation invoke interpreters this way. See Location of Perl.) 

3. . Passed in implicitly via standard input. This works only if there are no filename arguments--to pass arguments to a STDIN script you must explicitly specify a ``-'' for the script name. 

With methods 2 and 3, Perl starts parsing the input file from the beginning, unless you've specified a -x switch, in which case it scans for the first line starting with #! and containing the word ``perl'', and starts there instead. This is useful for running a script embedded in a larger message. (In this case you would indicate the end of the script using the __END__ token.) 

The #! line is always examined for switches as the line is being parsed. Thus, if you're on a machine that allows only one argument with the #! line, or worse, doesn't even recognize the #! line, you still can get consistent switch behavior regardless of how Perl was invoked, even if -x was used to find the beginning of the script. 

Because many operating systems silently chop off kernel interpretation of the #! line after 32 characters, some switches may be passed in on the command line, and some may not; you could even get a ``-'' without its letter, if you're not careful. You probably want to make sure that all your switches fall either before or after that 32 character boundary. Most switches don't actually care if they're processed redundantly, but getting a - instead of a complete switch could cause Perl to try to execute standard input instead of your script. And a partial -I switch could also cause odd results. 

Some switches do care if they are processed twice, for instance combinations of -l and -0. Either put all the switches after the 32 character boundary (if applicable), or replace the use of -0digits by BEGIN{ $/ = "\0digits"; }. 

Parsing of the #! switches starts wherever ``perl'' is mentioned in the line. The sequences ``-*'' and ``- '' are specifically ignored so that you could, if you were so inclined, say 

    #!/bin/sh -- # -*- perl -*- -p

    eval 'exec /usr/bin/perl -wS $0 ${1+"$@"}'

        if $running_under_some_shell;

to let Perl see the -p switch. 

If the #! line does not contain the word ``perl'', the program named after the #! is executed instead of the Perl interpreter. This is slightly bizarre, but it helps people on machines that don't do #!, because they can tell a program that their SHELL is /usr/bin/perl, and Perl will then dispatch the program to the correct interpreter for them. 

After locating your script, Perl compiles the entire script to an internal form. If there are any compilation errors, execution of the script is not attempted. (This is unlike the typical shell script, which might run part-way through before finding a syntax error.) 

If the script is syntactically correct, it is executed. If the script runs off the end without hitting an exit() or die() operator, an implicit exit(0) is provided to indicate successful completion. 



#! and quoting on non-Unix systems

Unix's #! technique can be simulated on other systems: 

OS/2 

Put 

    extproc perl -S -your_switches

as the first line in *.cmd file (-S due to a bug in cmd.exe's `extproc' handling). 

MS-DOS 

Create a batch file to run your script, and codify it in ALTERNATIVE_SHEBANG (see the dosish.h file in the source distribution for more information). 

Win95/NT 

The Win95/NT installation, when using the Activeware port of Perl, will modify the Registry to associate the .pl extension with the perl interpreter. If you install another port of Perl, including the one in the Win32 directory of the Perl distribution, then you'll have to modify the Registry yourself. Note that this means you can no longer tell the difference between an executable Perl program and a Perl library file. 

Macintosh 

Macintosh perl scripts will have the appropriate Creator and Type, so that double-clicking them will invoke the perl application. 

Command-interpreters on non-Unix systems have rather different ideas on quoting than Unix shells. You'll need to learn the special characters in your command-interpreter (*, \ and " are common) and how to protect whitespace and these characters to run one-liners (see -e below). 

On some systems, you may have to change single-quotes to double ones, which you must NOT do on Unix or Plan9 systems. You might also have to change a single % to a %%. 

For example: 

    # Unix

    perl -e 'print "Hello world\n"'

    # MS-DOS, etc.

    perl -e "print \"Hello world\n\""

    # Macintosh

    print "Hello world\n"

     (then Run "Myscript" or Shift-Command-R)

    # VMS

    perl -e "print ""Hello world\n"""

The problem is that none of this is reliable: it depends on the command and it is entirely possible neither works. If 4DOS was the command shell, this would probably work better: 

    perl -e "print <Ctrl-x>"Hello world\n<Ctrl-x>""

CMD.EXE in Windows NT slipped a lot of standard Unix functionality in when nobody was looking, but just try to find documentation for its quoting rules. 

Under the Macintosh, it depends which environment you are using. The MacPerl shell, or MPW, is much like Unix shells in its support for several quoting variants, except that it makes free use of the Macintosh's non-ASCII characters as control characters. 

There is no general solution to all of this. It's just a mess. 



Location of Perl

It may seem obvious to say, but Perl is useful only when users can easily find it. When possible, it's good for both /usr/bin/perl and /usr/local/bin/perl to be symlinks to the actual binary. If that can't be done, system administrators are strongly encouraged to put (symlinks to) perl and its accompanying utilities, such as perldoc, into a directory typically found along a user's PATH, or in another obvious and convenient place. 

In this documentation, #!/usr/bin/perl on the first line of the script will stand in for whatever method works on your system. 



Switches

A single-character switch may be combined with the following switch, if any. 

    #!/usr/bin/perl -spi.bak    # same as -s -p -i.bak

Switches include: 

-0[digits] 

specifies the input record separator ($/) as an octal number. If there are no digits, the null character is the separator. Other switches may precede or follow the digits. For example, if you have a version of find which can print filenames terminated by the null character, you can say this: 

    find . -name '*.bak' -print0 | perl -n0e unlink

The special value 00 will cause Perl to slurp files in paragraph mode. The value 0777 will cause Perl to slurp files whole because there is no legal character with that value. 

-a 

turns on autosplit mode when used with a -n or -p. An implicit split command to the @F array is done as the first thing inside the implicit while loop produced by the -n or -p. 

    perl -ane 'print pop(@F), "\n";'

is equivalent to 

    while (<>) {

        @F = split(' ');

        print pop(@F), "\n";

    }

An alternate delimiter may be specified using -F. 

-c 

causes Perl to check the syntax of the script and then exit without executing it. Actually, it will execute BEGIN, END, and use blocks, because these are considered as occurring outside the execution of your program. 

-d 

runs the script under the Perl debugger. See the perldebug manpage. 

-d:foo 

runs the script under the control of a debugging or tracing module installed as Devel::foo. E.g., -d:DProf executes the script using the Devel::DProf profiler. See the perldebug manpage. 

-Dletters 

-Dnumber 

sets debugging flags. To watch how it executes your script, use -Dtls. (This works only if debugging is compiled into your Perl.) Another nice value is -Dx, which lists your compiled syntax tree. And -Dr displays compiled regular expressions. As an alternative, specify a number instead of list of letters (e.g., -D14 is equivalent to -Dtls): 

        1  p  Tokenizing and parsing

        2  s  Stack snapshots

        4  l  Context (loop) stack processing

        8  t  Trace execution

       16  o  Method and overloading resolution

       32  c  String/numeric conversions

       64  P  Print preprocessor command for -P

      128  m  Memory allocation

      256  f  Format processing

      512  r  Regular expression parsing and execution

     1024  x  Syntax tree dump

     2048  u  Tainting checks

     4096  L  Memory leaks (needs C<-DLEAKTEST> when compiling Perl)

     8192  H  Hash dump -- usurps values()

    16384  X  Scratchpad allocation

    32768  D  Cleaning up

    65536  S  Thread synchronization

All these flags require -DDEBUGGING when you compile the Perl executable. This flag is automatically set if you include -g option when Configure asks you about optimizer/debugger flags. 

-e commandline 

may be used to enter one line of script. If -e is given, Perl will not look for a script filename in the argument list. Multiple -e commands may be given to build up a multi-line script. Make sure to use semicolons where you would in a normal program. 

-Fpattern 

specifies the pattern to split on if -a is also in effect. The pattern may be surrounded by //, "", or '', otherwise it will be put in single quotes. 

-h 

prints a summary of the options. 

-i[extension] 

specifies that files processed by the <> construct are to be edited in-place. It does this by renaming the input file, opening the output file by the original name, and selecting that output file as the default for print() statements. The extension, if supplied, is used to modify the name of the old file to make a backup copy, following these rules: 

If no extension is supplied, no backup is made and the current file is overwritten. 

If the extension doesn't contain a * then it is appended to the end of the current filename as a suffix. 

If the extension does contain one or more * characters, then each * is replaced with the current filename. In perl terms you could think of this as: 

    ($backup = $extension) =~ s/\*/$file_name/g;

This allows you to add a prefix to the backup file, instead of (or in addition to) a suffix: 

    $ perl -pi'bak_*' -e 's/bar/baz/' fileA     # backup to 'bak_fileA'

Or even to place backup copies of the original files into another directory (provided the directory already exists): 

    $ perl -pi'old/*.bak' -e 's/bar/baz/' fileA # backup to 'old/fileA.bak'

These sets of one-liners are equivalent: 

    $ perl -pi -e 's/bar/baz/' fileA            # overwrite current file

    $ perl -pi'*' -e 's/bar/baz/' fileA         # overwrite current file

    $ perl -pi'.bak' -e 's/bar/baz/' fileA      # backup to 'fileA.bak'

    $ perl -pi'*.bak' -e 's/bar/baz/' fileA     # backup to 'fileA.bak'

From the shell, saying 

    $ perl -p -i.bak -e "s/foo/bar/; ... "

is the same as using the script: 

    #!/usr/bin/perl -pi.bak

    s/foo/bar/;

which is equivalent to 

    #!/usr/bin/perl

    $extension = '.bak';

    while (<>) {

        if ($ARGV ne $oldargv) {

            if ($extension !~ /\*/) {

                $backup = $ARGV . $extension;

            }

            else {

                ($backup = $extension) =~ s/\*/$ARGV/g;

            }

            rename($ARGV, $backup);

            open(ARGVOUT, ">$ARGV");

            select(ARGVOUT);

            $oldargv = $ARGV;

        }

        s/foo/bar/;

    }

    continue {

        print;  # this prints to original filename

    }

    select(STDOUT);

except that the -i form doesn't need to compare $ARGV to $oldargv to know when the filename has changed. It does, however, use ARGVOUT for the selected filehandle. Note that STDOUT is restored as the default output filehandle after the loop. 

As shown above, Perl creates the backup file whether or not any output is actually changed. So this is just a fancy way to copy files: 

    $ perl -p -i'/some/file/path/*' -e 1 file1 file2 file3...

  or

    $ perl -p -i'.bak' -e 1 file1 file2 file3...

You can use eof without parentheses to locate the end of each input file, in case you want to append to each file, or reset line numbering (see example in eof). 

If, for a given file, Perl is unable to create the backup file as specified in the extension then it will skip that file and continue on with the next one (if it exists). 

For a discussion of issues surrounding file permissions and -i, see Why does Perl let me delete read-only files? Why does -i clobber protected files? Isn't this a bug in Perl?. 

You cannot use -i to create directories or to strip extensions from files. 

Perl does not expand ~, so don't do that. 

Finally, note that the -i switch does not impede execution when no files are given on the command line. In this case, no backup is made (the original file cannot, of course, be determined) and processing proceeds from STDIN to STDOUT as might be expected. 

-Idirectory 

Directories specified by -I are prepended to the search path for modules (@INC), and also tells the C preprocessor where to search for include files. The C preprocessor is invoked with -P; by default it searches /usr/include and /usr/lib/perl. 

-l[octnum] 

enables automatic line-ending processing. It has two effects: first, it automatically chomps ``$/'' (the input record separator) when used with -n or -p, and second, it assigns ``$\'' (the output record separator) to have the value of octnum so that any print statements will have that separator added back on. If octnum is omitted, sets ``$\'' to the current value of ``$/''. For instance, to trim lines to 80 columns: 

    perl -lpe 'substr($_, 80) = ""'

Note that the assignment $\ = $/ is done when the switch is processed, so the input record separator can be different than the output record separator if the -l switch is followed by a -0 switch: 

    gnufind / -print0 | perl -ln0e 'print "found $_" if -p'

This sets $\ to newline and then sets $/ to the null character. 

-m[-]module 

-M[-]module 

-M[-]'module ...' 

-[mM][-]module=arg[,arg]... 

-mmodule executes use module (); before executing your script. 

-Mmodule executes use module ; before executing your script. You can use quotes to add extra code after the module name, e.g., -M'module qw(foo bar)'. 

If the first character after the -M or -m is a dash (-) then the 'use' is replaced with 'no'. 

A little builtin syntactic sugar means you can also say -mmodule=foo,bar or -Mmodule=foo,bar as a shortcut for -M'module qw(foo bar)'. This avoids the need to use quotes when importing symbols. The actual code generated by -Mmodule=foo,bar is use module split(/,/,q{foo,bar}). Note that the = form removes the distinction between -m and -M. 

-n 

causes Perl to assume the following loop around your script, which makes it iterate over filename arguments somewhat like sed -n or awk: 

    while (<>) {

        ...             # your script goes here

    }

Note that the lines are not printed by default. See -p to have lines printed. If a file named by an argument cannot be opened for some reason, Perl warns you about it, and moves on to the next file. 

Here is an efficient way to delete all files older than a week: 

    find . -mtime +7 -print | perl -nle 'unlink;'

This is faster than using the -exec switch of find because you don't have to start a process on every filename found. 

BEGIN and END blocks may be used to capture control before or after the implicit loop, just as in awk. 

-p 

causes Perl to assume the following loop around your script, which makes it iterate over filename arguments somewhat like sed: 

    while (<>) {

        ...             # your script goes here

    } continue {

        print or die "-p destination: $!\n";

    }

If a file named by an argument cannot be opened for some reason, Perl warns you about it, and moves on to the next file. Note that the lines are printed automatically. An error occuring during printing is treated as fatal. To suppress printing use the -n switch. A -p overrides a -n switch. 

BEGIN and END blocks may be used to capture control before or after the implicit loop, just as in awk. 

-P 

causes your script to be run through the C preprocessor before compilation by Perl. (Because both comments and cpp directives begin with the # character, you should avoid starting comments with any words recognized by the C preprocessor such as ``if'', ``else'', or ``define''.) 

-s 

enables some rudimentary switch parsing for switches on the command line after the script name but before any filename arguments (or before a --). Any switch found there is removed from @ARGV and sets the corresponding variable in the Perl script. The following script prints ``true'' if and only if the script is invoked with a -xyz switch. 

    #!/usr/bin/perl -s

    if ($xyz) { print "true\n"; }

-S 

makes Perl use the PATH environment variable to search for the script (unless the name of the script contains directory separators). On some platforms, this also makes Perl append suffixes to the filename while searching for it. For example, on Win32 platforms, the ``.bat'' and ``.cmd'' suffixes are appended if a lookup for the original name fails, and if the name does not already end in one of those suffixes. If your Perl was compiled with DEBUGGING turned on, using the -Dp switch to Perl shows how the search progresses. 

If the filename supplied contains directory separators (i.e. it is an absolute or relative pathname), and if the file is not found, platforms that append file extensions will do so and try to look for the file with those extensions added, one by one. 

On DOS-like platforms, if the script does not contain directory separators, it will first be searched for in the current directory before being searched for on the PATH. On Unix platforms, the script will be searched for strictly on the PATH. 

Typically this is used to emulate #! startup on platforms that don't support #!. This example works on many platforms that have a shell compatible with Bourne shell: 

    #!/usr/bin/perl

    eval 'exec /usr/bin/perl -wS $0 ${1+"$@"}'

            if $running_under_some_shell;

The system ignores the first line and feeds the script to /bin/sh, which proceeds to try to execute the Perl script as a shell script. The shell executes the second line as a normal shell command, and thus starts up the Perl interpreter. On some systems $0 doesn't always contain the full pathname, so the -S tells Perl to search for the script if necessary. After Perl locates the script, it parses the lines and ignores them because the variable $running_under_some_shell is never true. If the script will be interpreted by csh, you will need to replace ${1+"$@"} with $*, even though that doesn't understand embedded spaces (and such) in the argument list. To start up sh rather than csh, some systems may have to replace the #! line with a line containing just a colon, which will be politely ignored by Perl. Other systems can't control that, and need a totally devious construct that will work under any of csh, sh, or Perl, such as the following: 

        eval '(exit $?0)' && eval 'exec /usr/bin/perl -wS $0 ${1+"$@"}'

        & eval 'exec /usr/bin/perl -wS $0 $argv:q'

                if $running_under_some_shell;

-T 

forces ``taint'' checks to be turned on so you can test them. Ordinarily these checks are done only when running setuid or setgid. It's a good idea to turn them on explicitly for programs run on another's behalf, such as CGI programs. See the perlsec manpage. Note that (for security reasons) this option must be seen by Perl quite early; usually this means it must appear early on the command line or in the #! line (for systems which support that). 

-u 

causes Perl to dump core after compiling your script. You can then in theory take this core dump and turn it into an executable file by using the undump program (not supplied). This speeds startup at the expense of some disk space (which you can minimize by stripping the executable). (Still, a ``hello world'' executable comes out to about 200K on my machine.) If you want to execute a portion of your script before dumping, use the dump() operator instead. Note: availability of undump is platform specific and may not be available for a specific port of Perl. It has been superseded by the new perl-to-C compiler, which is more portable, even though it's still only considered beta. 

-U 

allows Perl to do unsafe operations. Currently the only ``unsafe'' operations are the unlinking of directories while running as superuser, and running setuid programs with fatal taint checks turned into warnings. Note that the -w switch (or the $^W variable) must be used along with this option to actually generate the taint-check warnings. 

-v 

prints the version and patchlevel of your Perl executable. 

-V 

prints summary of the major perl configuration values and the current value of @INC. 

-V:name 

Prints to STDOUT the value of the named configuration variable. 

-w 

prints warnings about variable names that are mentioned only once, and scalar variables that are used before being set. Also warns about redefined subroutines, and references to undefined filehandles or filehandles opened read-only that you are attempting to write on. Also warns you if you use values as a number that doesn't look like numbers, using an array as though it were a scalar, if your subroutines recurse more than 100 deep, and innumerable other things. 

You can disable specific warnings using __WARN__ hooks, as described in the perlvar manpage and warn. See also the perldiag manpage and the perltrap manpage. 

-x directory 

tells Perl that the script is embedded in a message. Leading garbage will be discarded until the first line that starts with #! and contains the string ``perl''. Any meaningful switches on that line will be applied. If a directory name is specified, Perl will switch to that directory before running the script. The -x switch controls only the disposal of leading garbage. The script must be terminated with __END__ if there is trailing garbage to be ignored (the script can process any or all of the trailing garbage via the DATA filehandle if desired). 



ENVIRONMENT

HOME 

Used if chdir has no argument. 

LOGDIR 

Used if chdir has no argument and HOME is not set. 

PATH 

Used in executing subprocesses, and in finding the script if -S is used. 

PERL5LIB 

A colon-separated list of directories in which to look for Perl library files before looking in the standard library and the current directory. If PERL5LIB is not defined, PERLLIB is used. When running taint checks (because the script was running setuid or setgid, or the -T switch was used), neither variable is used. The script should instead say 

    use lib "/my/directory";

PERL5OPT 

Command-line options (switches). Switches in this variable are taken as if they were on every Perl command line. Only the -[DIMUdmw] switches are allowed. When running taint checks (because the script was running setuid or setgid, or the -T switch was used), this variable is ignored. 

PERLLIB 

A colon-separated list of directories in which to look for Perl library files before looking in the standard library and the current directory. If PERL5LIB is defined, PERLLIB is not used. 

PERL5DB 

The command used to load the debugger code. The default is: 

        BEGIN { require 'perl5db.pl' }

PERL5SHELL (specific to WIN32 port) 

May be set to an alternative shell that perl must use internally for executing ``backtick'' commands or system(). Default is cmd.exe /x/c on WindowsNT and command.com /c on Windows95. The value is considered to be space delimited. Precede any character that needs to be protected (like a space or backslash) with a backslash. 

Note that Perl doesn't use COMSPEC for this purpose because COMSPEC has a high degree of variability among users, leading to portability concerns. Besides, perl can use a shell that may not be fit for interactive use, and setting COMSPEC to such a shell may interfere with the proper functioning of other programs (which usually look in COMSPEC to find a shell fit for interactive use). 

PERL_DEBUG_MSTATS 

Relevant only if perl is compiled with the malloc included with the perl distribution (that is, if perl -V:d_mymalloc is 'define'). If set, this causes memory statistics to be dumped after execution. If set to an integer greater than one, also causes memory statistics to be dumped after compilation. 

PERL_DESTRUCT_LEVEL 

Relevant only if your perl executable was built with -DDEBUGGING, this controls the behavior of global destruction of objects and other references. 

Perl also has environment variables that control how Perl handles data specific to particular natural languages. See the perllocale manpage. 

Apart from these, Perl uses no other environment variables, except to make them available to the script being executed, and to child processes. However, scripts running setuid would do well to execute the following lines before doing anything else, just to keep people honest: 

    $ENV{PATH} = '/bin:/usr/bin';    # or whatever you need

    $ENV{SHELL} = '/bin/sh' if exists $ENV{SHELL};

    delete @ENV{qw(IFS CDPATH ENV BASH_ENV)};



DISCLAIMER

We are painfully aware that these documents may contain incorrect links and misformatted HTML. Such bugs lie in the automatic translation process that automatically created the hundreds and hundreds of separate documents that you find here. Please do not report link or formatting bugs, because we cannot fix per-document problems. The only bug reports that will help us are those that supply working patches to the installhtml or pod2html programs, or to the Pod::HTML module itself, for which I and the entire Perl community will shower you with thanks and praises. 

If rather than formatting bugs, you encounter substantive content errors in these documents, such as mistakes in the explanations or code, please use the perlbug utility included with the Perl distribution. 

--Tom Christiansen, Perl Documentation Compiler and Editor

NAME

perlform - Perl formats 



DESCRIPTION

Perl has a mechanism to help you generate simple reports and charts. To facilitate this, Perl helps you code up your output page close to how it will look when it's printed. It can keep track of things like how many lines are on a page, what page you're on, when to print page headers, etc. Keywords are borrowed from FORTRAN: format() to declare and write() to execute; see their entries in the perlfunc manpage. Fortunately, the layout is much more legible, more like BASIC's PRINT USING statement. Think of it as a poor man's nroff(1). 

Formats, like packages and subroutines, are declared rather than executed, so they may occur at any point in your program. (Usually it's best to keep them all together though.) They have their own namespace apart from all the other ``types'' in Perl. This means that if you have a function named ``Foo'', it is not the same thing as having a format named ``Foo''. However, the default name for the format associated with a given filehandle is the same as the name of the filehandle. Thus, the default format for STDOUT is named ``STDOUT'', and the default format for filehandle TEMP is named ``TEMP''. They just look the same. They aren't. 

Output record formats are declared as follows: 

    format NAME =

    FORMLIST

    .

If name is omitted, format ``STDOUT'' is defined. FORMLIST consists of a sequence of lines, each of which may be one of three types: 

1. . 

A comment, indicated by putting a '#' in the first column. 

2. . A ``picture'' line giving the format for one output line. 

3. . An argument line supplying values to plug into the previous picture line. 

Picture lines are printed exactly as they look, except for certain fields that substitute values into the line. Each field in a picture line starts with either ``@'' (at) or ``^'' (caret). These lines do not undergo any kind of variable interpolation. The at field (not to be confused with the array marker @) is the normal kind of field; the other kind, caret fields, are used to do rudimentary multi-line text block filling. The length of the field is supplied by padding out the field with multiple ``<'', ``>'', or ``|'' characters to specify, respectively, left justification, right justification, or centering. If the variable would exceed the width specified, it is truncated. 

As an alternate form of right justification, you may also use ``#'' characters (with an optional ``.'') to specify a numeric field. This way you can line up the decimal points. If any value supplied for these fields contains a newline, only the text up to the newline is printed. Finally, the special field ``@*'' can be used for printing multi-line, nontruncated values; it should appear by itself on a line. 

The values are specified on the following line in the same order as the picture fields. The expressions providing the values should be separated by commas. The expressions are all evaluated in a list context before the line is processed, so a single list expression could produce multiple list elements. The expressions may be spread out to more than one line if enclosed in braces. If so, the opening brace must be the first token on the first line. If an expression evaluates to a number with a decimal part, and if the corresponding picture specifies that the decimal part should appear in the output (that is, any picture except multiple ``#'' characters without an embedded ``.''), the character used for the decimal point is always determined by the current LC_NUMERIC locale. This means that, if, for example, the run-time environment happens to specify a German locale, ``,'' will be used instead of the default ``.''. See the perllocale manpage and WARNINGS for more information. 

Picture fields that begin with ^ rather than @ are treated specially. With a # field, the field is blanked out if the value is undefined. For other field types, the caret enables a kind of fill mode. Instead of an arbitrary expression, the value supplied must be a scalar variable name that contains a text string. Perl puts as much text as it can into the field, and then chops off the front of the string so that the next time the variable is referenced, more of the text can be printed. (Yes, this means that the variable itself is altered during execution of the write() call, and is not returned.) Normally you would use a sequence of fields in a vertical stack to print out a block of text. You might wish to end the final field with the text ``...'', which will appear in the output if the text was too long to appear in its entirety. You can change which characters are legal to break on by changing the variable $: (that's $FORMAT_LINE_BREAK_CHARACTERS if you're using the English module) to a list of the desired characters. 

Using caret fields can produce variable length records. If the text to be formatted is short, you can suppress blank lines by putting a ``~'' (tilde) character anywhere in the line. The tilde will be translated to a space upon output. If you put a second tilde contiguous to the first, the line will be repeated until all the fields on the line are exhausted. (If you use a field of the at variety, the expression you supply had better not give the same value every time forever!) 

Top-of-form processing is by default handled by a format with the same name as the current filehandle with ``_TOP'' concatenated to it. It's triggered at the top of each page. See write. 

Examples: 

 # a report on the /etc/passwd file

 format STDOUT_TOP =

                         Passwd File

 Name                Login    Office   Uid   Gid Home

 ------------------------------------------------------------------

 .

 format STDOUT =

 @<<<<<<<<<<<<<<<<<< @||||||| @<<<<<<@>>>> @>>>> @<<<<<<<<<<<<<<<<<

 $name,              $login,  $office,$uid,$gid, $home

 .

 # a report from a bug report form

 format STDOUT_TOP =

                         Bug Reports

 @<<<<<<<<<<<<<<<<<<<<<<<     @|||         @>>>>>>>>>>>>>>>>>>>>>>>

 $system,                      $%,         $date

 ------------------------------------------------------------------

 .

 format STDOUT =

 Subject: @<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<

          $subject

 Index: @<<<<<<<<<<<<<<<<<<<<<<<<<<<< ^<<<<<<<<<<<<<<<<<<<<<<<<<<<<

        $index,                       $description

 Priority: @<<<<<<<<<< Date: @<<<<<<< ^<<<<<<<<<<<<<<<<<<<<<<<<<<<<

           $priority,        $date,   $description

 From: @<<<<<<<<<<<<<<<<<<<<<<<<<<<<< ^<<<<<<<<<<<<<<<<<<<<<<<<<<<<

       $from,                         $description

 Assigned to: @<<<<<<<<<<<<<<<<<<<<<< ^<<<<<<<<<<<<<<<<<<<<<<<<<<<<

              $programmer,            $description

 ~                                    ^<<<<<<<<<<<<<<<<<<<<<<<<<<<<

                                      $description

 ~                                    ^<<<<<<<<<<<<<<<<<<<<<<<<<<<<

                                      $description

 ~                                    ^<<<<<<<<<<<<<<<<<<<<<<<<<<<<

                                      $description

 ~                                    ^<<<<<<<<<<<<<<<<<<<<<<<<<<<<

                                      $description

 ~                                    ^<<<<<<<<<<<<<<<<<<<<<<<...

                                      $description

 .

It is possible to intermix print()s with write()s on the same output channel, but you'll have to handle $- ($FORMAT_LINES_LEFT) yourself. 



Format Variables

The current format name is stored in the variable $~ ($FORMAT_NAME), and the current top of form format name is in $^ ($FORMAT_TOP_NAME). The current output page number is stored in $% ($FORMAT_PAGE_NUMBER), and the number of lines on the page is in $= ($FORMAT_LINES_PER_PAGE). Whether to autoflush output on this handle is stored in $| ($OUTPUT_AUTOFLUSH). The string output before each top of page (except the first) is stored in $^L ($FORMAT_FORMFEED). These variables are set on a per-filehandle basis, so you'll need to select() into a different one to affect them: 

    select((select(OUTF),

            $~ = "My_Other_Format",

            $^ = "My_Top_Format"

           )[0]);

Pretty ugly, eh? It's a common idiom though, so don't be too surprised when you see it. You can at least use a temporary variable to hold the previous filehandle: (this is a much better approach in general, because not only does legibility improve, you now have intermediary stage in the expression to single-step the debugger through): 

    $ofh = select(OUTF);

    $~ = "My_Other_Format";

    $^ = "My_Top_Format";

    select($ofh);

If you use the English module, you can even read the variable names: 

    use English;

    $ofh = select(OUTF);

    $FORMAT_NAME     = "My_Other_Format";

    $FORMAT_TOP_NAME = "My_Top_Format";

    select($ofh);

But you still have those funny select()s. So just use the FileHandle module. Now, you can access these special variables using lowercase method names instead: 

    use FileHandle;

    format_name     OUTF "My_Other_Format";

    format_top_name OUTF "My_Top_Format";

Much better! 



NOTES

Because the values line may contain arbitrary expressions (for at fields, not caret fields), you can farm out more sophisticated processing to other functions, like sprintf() or one of your own. For example: 

    format Ident =

        @<<<<<<<<<<<<<<<

        &commify($n)

    .

To get a real at or caret into the field, do this: 

    format Ident =

    I have an @ here.

            "@"

    .

To center a whole line of text, do something like this: 

    format Ident =

    @|||||||||||||||||||||||||||||||||||||||||||||||

            "Some text line"

    .

There is no builtin way to say ``float this to the right hand side of the page, however wide it is.'' You have to specify where it goes. The truly desperate can generate their own format on the fly, based on the current number of columns, and then eval() it: 

    $format  = "format STDOUT = \n"

             . '^' . '<' x $cols . "\n"

             . '$entry' . "\n"

             . "\t^" . "<" x ($cols-8) . "~~\n"

             . '$entry' . "\n"

             . ".\n";

    print $format if $Debugging;

    eval $format;

    die $@ if $@;

Which would generate a format looking something like this: 

 format STDOUT =

 ^<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<

 $entry

         ^<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<~~

 $entry

 .

Here's a little program that's somewhat like fmt(1): 

 format =

 ^<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<<< ~~

 $_

 .

 $/ = '';

 while (<>) {

     s/\s*\n\s*/ /g;

     write;

 }



Footers

While $FORMAT_TOP_NAME contains the name of the current header format, there is no corresponding mechanism to automatically do the same thing for a footer. Not knowing how big a format is going to be until you evaluate it is one of the major problems. It's on the TODO list. 

Here's one strategy: If you have a fixed-size footer, you can get footers by checking $FORMAT_LINES_LEFT before each write() and print the footer yourself if necessary. 

Here's another strategy: Open a pipe to yourself, using open(MYSELF, "|-") (see open()) and always write() to MYSELF instead of STDOUT. Have your child process massage its STDIN to rearrange headers and footers however you like. Not very convenient, but doable. 



Accessing Formatting Internals

For low-level access to the formatting mechanism. you may use formline() and access $^A (the $ACCUMULATOR variable) directly. 

For example: 

    $str = formline <<'END', 1,2,3;

    @<<<  @|||  @>>>

    END

    print "Wow, I just stored `$^A' in the accumulator!\n";

Or to make an swrite() subroutine, which is to write() what sprintf() is to printf(), do this: 

    use Carp;

    sub swrite {

        croak "usage: swrite PICTURE ARGS" unless @_;

        my $format = shift;

        $^A = "";

        formline($format,@_);

        return $^A;

    }

    $string = swrite(<<'END', 1, 2, 3);

 Check me out

 @<<<  @|||  @>>>

 END

    print $string;



WARNINGS

The lone dot that ends a format can also prematurely end a mail message passing through a misconfigured Internet mailer (and based on experience, such misconfiguration is the rule, not the exception). So when sending format code through mail, you should indent it so that the format-ending dot is not on the left margin; this will prevent SMTP cutoff. 

Lexical variables (declared with ``my'') are not visible within a format unless the format is declared within the scope of the lexical variable. (They weren't visible at all before version 5.001.) 

Formats are the only part of Perl that unconditionally use information from a program's locale; if a program's environment specifies an LC_NUMERIC locale, it is always used to specify the decimal point character in formatted output. Perl ignores all other aspects of locale handling unless the use locale pragma is in effect. Formatted output cannot be controlled by use locale because the pragma is tied to the block structure of the program, and, for historical reasons, formats exist outside that block structure. See the perllocale manpage for further discussion of locale handling. 



DISCLAIMER

We are painfully aware that these documents may contain incorrect links and misformatted HTML. Such bugs lie in the automatic translation process that automatically created the hundreds and hundreds of separate documents that you find here. Please do not report link or formatting bugs, because we cannot fix per-document problems. The only bug reports that will help us are those that supply working patches to the installhtml or pod2html programs, or to the Pod::HTML module itself, for which I and the entire Perl community will shower you with thanks and praises. 

If rather than formatting bugs, you encounter substantive content errors in these documents, such as mistakes in the explanations or code, please use the perlbug utility included with the Perl distribution. 

--Tom Christiansen, Perl Documentation Compiler and Editor

NAME

perlvar - Perl predefined variables 



DESCRIPTION



Predefined Names

The following names have special meaning to Perl. Most punctuation names have reasonable mnemonics, or analogues in one of the shells. Nevertheless, if you wish to use long variable names, you just need to say 

    use English;

at the top of your program. This will alias all the short names to the long names in the current package. Some even have medium names, generally borrowed from awk. 

To go a step further, those variables that depend on the currently selected filehandle may instead (and preferably) be set by calling an object method on the FileHandle object. (Summary lines below for this contain the word HANDLE.) First you must say 

    use FileHandle;

after which you may use either 

    method HANDLE EXPR

or more safely, 

    HANDLE->method(EXPR)

Each of the methods returns the old value of the FileHandle attribute. The methods each take an optional EXPR, which if supplied specifies the new value for the FileHandle attribute in question. If not supplied, most of the methods do nothing to the current value, except for autoflush(), which will assume a 1 for you, just to be different. 

A few of these variables are considered ``read-only''. This means that if you try to assign to this variable, either directly or indirectly through a reference, you'll raise a run-time exception. 

The following list is ordered by scalar variables first, then the arrays, then the hashes (except $^M was added in the wrong place). This is somewhat obscured by the fact that %ENV and %SIG are listed as $ENV{expr} and $SIG{expr}. 

$ARG 

$_ 

The default input and pattern-searching space. The following pairs are equivalent: 

    while (<>) {...}    # equivalent in only while!

    while (defined($_ = <>)) {...}

    /^Subject:/

    $_ =~ /^Subject:/

    tr/a-z/A-Z/

    $_ =~ tr/a-z/A-Z/

    chop

    chop($_)

Here are the places where Perl will assume $_ even if you don't use it: 

· Various unary functions, including functions like ord() and int(), as well as the all file tests ( -f, -d) except for -t, which defaults to STDIN. 

· Various list functions like print() and unlink(). 

· The pattern matching operations m//, s///, and tr/// when used without an =~ operator. 

· The default iterator variable in a foreach loop if no other variable is supplied. 

· The implicit iterator variable in the grep() and map() functions. 

· The default place to put an input record when a <FH> operation's result is tested by itself as the sole criterion of a while test. Note that outside of a while test, this will not happen. 

(Mnemonic: underline is understood in certain operations.) 

$ltdigitsgt 

Contains the subpattern from the corresponding set of parentheses in the last pattern matched, not counting patterns matched in nested blocks that have been exited already. (Mnemonic: like \digits.) These variables are all read-only. 

$MATCH 

$& 

The string matched by the last successful pattern match (not counting any matches hidden within a BLOCK or eval() enclosed by the current BLOCK). (Mnemonic: like & in some editors.) This variable is read-only. 

$PREMATCH 

$` 

The string preceding whatever was matched by the last successful pattern match (not counting any matches hidden within a BLOCK or eval enclosed by the current BLOCK). (Mnemonic: ` often precedes a quoted string.) This variable is read-only. 

$POSTMATCH 

$' 

The string following whatever was matched by the last successful pattern match (not counting any matches hidden within a BLOCK or eval() enclosed by the current BLOCK). (Mnemonic: ' often follows a quoted string.) Example: 

    $_ = 'abcdefghi';

    /def/;

    print "$`:$&:$'\n";         # prints abc:def:ghi

This variable is read-only. 

$LAST_PAREN_MATCH 

$+ 

The last bracket matched by the last search pattern. This is useful if you don't know which of a set of alternative patterns matched. For example: 

    /Version: (.*)|Revision: (.*)/ && ($rev = $+);

(Mnemonic: be positive and forward looking.) This variable is read-only. 

$MULTILINE_MATCHING 

$* 

Set to 1 to do multi-line matching within a string, 0 to tell Perl that it can assume that strings contain a single line, for the purpose of optimizing pattern matches. Pattern matches on strings containing multiple newlines can produce confusing results when ``$*'' is 0. Default is 0. (Mnemonic: * matches multiple things.) Note that this variable influences the interpretation of only ``^'' and ``$''. A literal newline can be searched for even when $* == 0. 

Use of ``$*'' is deprecated in modern Perls, supplanted by the /s and /m modifiers on pattern matching. 

input_line_number HANDLE EXPR 

$INPUT_LINE_NUMBER 

$NR 

$. 

The current input line number for the last file handle from which you read (or performed a seek or tell on). An explicit close on a filehandle resets the line number. Because ``<>'' never does an explicit close, line numbers increase across ARGV files (but see examples under eof()). Localizing $. has the effect of also localizing Perl's notion of ``the last read filehandle''. (Mnemonic: many programs use ``.'' to mean the current line number.) 

input_record_separator HANDLE EXPR 

$INPUT_RECORD_SEPARATOR 

$RS 

$/ 

The input record separator, newline by default. Works like awk's RS variable, including treating empty lines as delimiters if set to the null string. (Note: An empty line cannot contain any spaces or tabs.) You may set it to a multi-character string to match a multi-character delimiter, or to undef to read to end of file. Note that setting it to "\n\n" means something slightly different than setting it to "", if the file contains consecutive empty lines. Setting it to "" will treat two or more consecutive empty lines as a single empty line. Setting it to "\n\n" will blindly assume that the next input character belongs to the next paragraph, even if it's a newline. (Mnemonic: / is used to delimit line boundaries when quoting poetry.) 

    undef $/;

    $_ = <FH>;          # whole file now here

    s/\n[ \t]+/ /g;

Remember: the value of $/ is a string, not a regexp. AWK has to be better for something :-) 

Setting $/ to a reference to an integer, scalar containing an integer, or scalar that's convertable to an integer will attempt to read records instead of lines, with the maximum record size being the referenced integer. So this: 

    $/ = \32768; # or \"32768", or \$var_containing_32768

    open(FILE, $myfile);

    $_ = <FILE>;

will read a record of no more than 32768 bytes from FILE. If you're not reading from a record-oriented file (or your OS doesn't have record-oriented files), then you'll likely get a full chunk of data with every read. If a record is larger than the record size you've set, you'll get the record back in pieces. 

On VMS, record reads are done with the equivalent of sysread, so it's best not to mix record and non-record reads on the same file. (This is likely not a problem, as any file you'd want to read in record mode is proably usable in line mode) Non-VMS systems perform normal I/O, so it's safe to mix record and non-record reads of a file. 

autoflush HANDLE EXPR 

$OUTPUT_AUTOFLUSH 

$| 

If set to nonzero, forces a flush right away and after every write or print on the currently selected output channel. Default is 0 (regardless of whether the channel is actually buffered by the system or not; $| tells you only whether you've asked Perl explicitly to flush after each write). Note that STDOUT will typically be line buffered if output is to the terminal and block buffered otherwise. Setting this variable is useful primarily when you are outputting to a pipe, such as when you are running a Perl script under rsh and want to see the output as it's happening. This has no effect on input buffering. (Mnemonic: when you want your pipes to be piping hot.) 

output_field_separator HANDLE EXPR 

$OUTPUT_FIELD_SEPARATOR 

$OFS 

$, 

The output field separator for the print operator. Ordinarily the print operator simply prints out the comma-separated fields you specify. To get behavior more like awk, set this variable as you would set awk's OFS variable to specify what is printed between fields. (Mnemonic: what is printed when there is a , in your print statement.) 

output_record_separator HANDLE EXPR 

$OUTPUT_RECORD_SEPARATOR 

$ORS 

$\ 

The output record separator for the print operator. Ordinarily the print operator simply prints out the comma-separated fields you specify, with no trailing newline or record separator assumed. To get behavior more like awk, set this variable as you would set awk's ORS variable to specify what is printed at the end of the print. (Mnemonic: you set ``$\'' instead of adding \n at the end of the print. Also, it's just like $/, but it's what you get ``back'' from Perl.) 

$LIST_SEPARATOR 

$" 

This is like ``$,'' except that it applies to array values interpolated into a double-quoted string (or similar interpreted string). Default is a space. (Mnemonic: obvious, I think.) 

$SUBSCRIPT_SEPARATOR 

$SUBSEP 

$; 

The subscript separator for multidimensional array emulation. If you refer to a hash element as 

    $foo{$a,$b,$c}

it really means 

    $foo{join($;, $a, $b, $c)}

But don't put 

    @foo{$a,$b,$c}      # a slice--note the @

which means 

    ($foo{$a},$foo{$b},$foo{$c})

Default is ``\034'', the same as SUBSEP in awk. Note that if your keys contain binary data there might not be any safe value for ``$;''. (Mnemonic: comma (the syntactic subscript separator) is a semi-semicolon. Yeah, I know, it's pretty lame, but `` $,'' is already taken for something more important.) 

Consider using ``real'' multidimensional arrays. 

$OFMT 

$# 

The output format for printed numbers. This variable is a half-hearted attempt to emulate awk's OFMT variable. There are times, however, when awk and Perl have differing notions of what is in fact numeric. The initial value is %.ng, where n is the value of the macro DBL_DIG from your system's float.h. This is different from awk's default OFMT setting of %.6g, so you need to set ``$#'' explicitly to get awk's value. (Mnemonic: # is the number sign.) 

Use of ``$#'' is deprecated. 

format_page_number HANDLE EXPR 

$FORMAT_PAGE_NUMBER 

$% 

The current page number of the currently selected output channel. (Mnemonic: % is page number in nroff.) 

format_lines_per_page HANDLE EXPR 

$FORMAT_LINES_PER_PAGE 

$= 

The current page length (printable lines) of the currently selected output channel. Default is 60. (Mnemonic: = has horizontal lines.) 

format_lines_left HANDLE EXPR 

$FORMAT_LINES_LEFT 

$- 

The number of lines left on the page of the currently selected output channel. (Mnemonic: lines_on_page - lines_printed.) 

format_name HANDLE EXPR 

$FORMAT_NAME 

$~ 

The name of the current report format for the currently selected output channel. Default is name of the filehandle. (Mnemonic: brother to ``$^''.) 

format_top_name HANDLE EXPR 

$FORMAT_TOP_NAME 

$^ 

The name of the current top-of-page format for the currently selected output channel. Default is name of the filehandle with _TOP appended. (Mnemonic: points to top of page.) 

format_line_break_characters HANDLE EXPR 

$FORMAT_LINE_BREAK_CHARACTERS 

$: 

The current set of characters after which a string may be broken to fill continuation fields (starting with ^) in a format. Default is " \n-", to break on whitespace or hyphens. (Mnemonic: a ``colon'' in poetry is a part of a line.) 

format_formfeed HANDLE EXPR 

$FORMAT_FORMFEED 

$^L 

What formats output to perform a form feed. Default is \f. 

$ACCUMULATOR 

$^A 

The current value of the write() accumulator for format() lines. A format contains formline() commands that put their result into $^A. After calling its format, write() prints out the contents of $^A and empties. So you never actually see the contents of $^A unless you call formline() yourself and then look at it. See the perlform manpage and formline(). 

$CHILD_ERROR 

$? 

The status returned by the last pipe close, backtick (``) command, or system() operator. Note that this is the status word returned by the wait() system call (or else is made up to look like it). Thus, the exit value of the subprocess is actually ( $? >> 8), and $? & 127 gives which signal, if any, the process died from, and $? & 128 reports whether there was a core dump. (Mnemonic: similar to sh and ksh.) 

Additionally, if the h_errno variable is supported in C, its value is returned via $? if any of the gethost*() functions fail. 

Note that if you have installed a signal handler for SIGCHLD, the value of $? will usually be wrong outside that handler. 

Inside an END subroutine $? contains the value that is going to be given to exit(). You can modify $? in an END subroutine to change the exit status of the script. 

Under VMS, the pragma use vmsish 'status' makes $? reflect the actual VMS exit status, instead of the default emulation of POSIX status. 

Also see Error Indicators. 

$OS_ERROR 

$ERRNO 

$! 

If used in a numeric context, yields the current value of errno, with all the usual caveats. (This means that you shouldn't depend on the value of $! to be anything in particular unless you've gotten a specific error return indicating a system error.) If used in a string context, yields the corresponding system error string. You can assign to $! to set errno if, for instance, you want "$!" to return the string for error n, or you want to set the exit value for the die() operator. (Mnemonic: What just went bang?) 

Also see Error Indicators. 

$EXTENDED_OS_ERROR 

$^E 

Error information specific to the current operating system. At the moment, this differs from $! under only VMS, OS/2, and Win32 (and for MacPerl). On all other platforms, $^E is always just the same as $!. 

Under VMS, $^E provides the VMS status value from the last system error. This is more specific information about the last system error than that provided by $!. This is particularly important when $! is set to EVMSERR. 

Under OS/2, $^E is set to the error code of the last call to OS/2 API either via CRT, or directly from perl. 

Under Win32, $^E always returns the last error information reported by the Win32 call GetLastError() which describes the last error from within the Win32 API. Most Win32-specific code will report errors via $^E. ANSI C and UNIX-like calls set errno and so most portable Perl code will report errors via $!. 

Caveats mentioned in the description of $! generally apply to $^E, also. (Mnemonic: Extra error explanation.) 

Also see Error Indicators. 

$EVAL_ERROR 

$@ 

The Perl syntax error message from the last eval() command. If null, the last eval() parsed and executed correctly (although the operations you invoked may have failed in the normal fashion). (Mnemonic: Where was the syntax error ``at''?) 

Note that warning messages are not collected in this variable. You can, however, set up a routine to process warnings by setting $SIG{__WARN__} as described below. 

Also see Error Indicators. 

$PROCESS_ID 

$PID 

$$ 

The process number of the Perl running this script. (Mnemonic: same as shells.) 

$REAL_USER_ID 

$UID 

$< 

The real uid of this process. (Mnemonic: it's the uid you came FROM, if you're running setuid.) 

$EFFECTIVE_USER_ID 

$EUID 

$> 

The effective uid of this process. Example: 

    $< = $>;            # set real to effective uid

    ($<,$>) = ($>,$<);  # swap real and effective uid

(Mnemonic: it's the uid you went TO, if you're running setuid.) Note: ``$<'' and ``$>'' can be swapped only on machines supporting setreuid(). 

$REAL_GROUP_ID 

$GID 

$( 

The real gid of this process. If you are on a machine that supports membership in multiple groups simultaneously, gives a space separated list of groups you are in. The first number is the one returned by getgid(), and the subsequent ones by getgroups(), one of which may be the same as the first number. 

However, a value assigned to ``$('' must be a single number used to set the real gid. So the value given by ``$('' should not be assigned back to ``$('' without being forced numeric, such as by adding zero. 

(Mnemonic: parentheses are used to GROUP things. The real gid is the group you LEFT, if you're running setgid.) 

$EFFECTIVE_GROUP_ID 

$EGID 

$) 

The effective gid of this process. If you are on a machine that supports membership in multiple groups simultaneously, gives a space separated list of groups you are in. The first number is the one returned by getegid(), and the subsequent ones by getgroups(), one of which may be the same as the first number. 

Similarly, a value assigned to ``$)'' must also be a space-separated list of numbers. The first number is used to set the effective gid, and the rest (if any) are passed to setgroups(). To get the effect of an empty list for setgroups(), just repeat the new effective gid; that is, to force an effective gid of 5 and an effectively empty setgroups() list, say <PRE> $) = &quot;5 5&quot; </PRE> . 

(Mnemonic: parentheses are used to GROUP things. The effective gid is the group that's RIGHT for you, if you're running setgid.) 

Note: ``$<'', ``$>'', ``$('' and ``$)'' can be set only on machines that support the corresponding set[re][ug]id() routine. ``$('' and ``$)'' can be swapped only on machines supporting setregid(). 

$PROGRAM_NAME 

$0 

Contains the name of the file containing the Perl script being executed. On some operating systems assigning to ``$0'' modifies the argument area that the ps(1) program sees. This is more useful as a way of indicating the current program state than it is for hiding the program you're running. (Mnemonic: same as sh and ksh.) 

$[ 

The index of the first element in an array, and of the first character in a substring. Default is 0, but you could set it to 1 to make Perl behave more like awk (or Fortran) when subscripting and when evaluating the index() and substr() functions. (Mnemonic: [ begins subscripts.) 

As of Perl 5, assignment to ``$['' is treated as a compiler directive, and cannot influence the behavior of any other file. Its use is discouraged. 

$PERL_VERSION 

$] 

The version + patchlevel / 1000 of the Perl interpreter. This variable can be used to determine whether the Perl interpreter executing a script is in the right range of versions. (Mnemonic: Is this version of perl in the right bracket?) Example: 

    warn "No checksumming!\n" if $] < 3.019;

See also the documentation of use VERSION and require VERSION for a convenient way to fail if the Perl interpreter is too old. 

$DEBUGGING 

$^D 

The current value of the debugging flags. (Mnemonic: value of -D switch.) 

$SYSTEM_FD_MAX 

$^F 

The maximum system file descriptor, ordinarily 2. System file descriptors are passed to exec()ed processes, while higher file descriptors are not. Also, during an open(), system file descriptors are preserved even if the open() fails. (Ordinary file descriptors are closed before the open() is attempted.) Note that the close-on-exec status of a file descriptor will be decided according to the value of $^F at the time of the open, not the time of the exec. 

$^H 

The current set of syntax checks enabled by use strict and other block scoped compiler hints. See the documentation of strict for more details. 

$INPLACE_EDIT 

$^I 

The current value of the inplace-edit extension. Use undef to disable inplace editing. (Mnemonic: value of -i switch.) 

$^M 

By default, running out of memory it is not trappable. However, if compiled for this, Perl may use the contents of $^M as an emergency pool after die()ing with this message. Suppose that your Perl were compiled with -DPERL_EMERGENCY_SBRK and used Perl's malloc. Then 

    $^M = 'a' x (1<<16);

would allocate a 64K buffer for use when in emergency. See the INSTALL file for information on how to enable this option. As a disincentive to casual use of this advanced feature, there is no the English manpage long name for this variable. 

$OSNAME 

$^O 

The name of the operating system under which this copy of Perl was built, as determined during the configuration process. The value is identical to $Config{'osname'}. 

$PERLDB 

$^P 

The internal variable for debugging support. Different bits mean the following (subject to change): 

1. x01 

Debug subroutine enter/exit. 

2. x02 Line-by-line debugging. 

3. x04 Switch off optimizations. 

4. x08 Preserve more data for future interactive inspections. 

5. x10 Keep info about source lines on which a subroutine is defined. 

6. x20 Start with single-step on. 

Note that some bits may be relevent at compile-time only, some at run-time only. This is a new mechanism and the details may change. 

$^R 

The result of evaluation of the last successful (?{ code }) regular expression assertion. (Excluding those used as switches.) May be written to. 

$^S 

Current state of the interpreter. Undefined if parsing of the current module/eval is not finished (may happen in $SIG{__DIE__} and $SIG{__WARN__} handlers). True if inside an eval, otherwise false. 

$BASETIME 

$^T 

The time at which the script began running, in seconds since the epoch (beginning of 1970). The values returned by the -M, -A, and -C filetests are based on this value. 

$WARNING 

$^W 

The current value of the warning switch, either TRUE or FALSE. (Mnemonic: related to the -w switch.) 

$EXECUTABLE_NAME 

$^X 

The name that the Perl binary itself was executed as, from C's argv[0]. 

$ARGV 

contains the name of the current file when reading from <>. 

@ARGV 

The array @ARGV contains the command line arguments intended for the script. Note that $#ARGV is the generally number of arguments minus one, because $ARGV[0] is the first argument, NOT the command name. See ``$0'' for the command name. 

@INC 

The array @INC contains the list of places to look for Perl scripts to be evaluated by the do EXPR, require, or use constructs. It initially consists of the arguments to any -I command line switches, followed by the default Perl library, probably /usr/local/lib/perl, followed by ``.'', to represent the current directory. If you need to modify this at runtime, you should use the use lib pragma to get the machine-dependent library properly loaded also: 

    use lib '/mypath/libdir/';

    use SomeMod;

@_ 

Within a subroutine the array @_ contains the parameters passed to that subroutine. See the perlsub manpage. 

%INC 

The hash %INC contains entries for each filename that has been included via do or require. The key is the filename you specified, and the value is the location of the file actually found. The require command uses this array to determine whether a given file has already been included. 

%ENV $ENV{expr} 

The hash %ENV contains your current environment. Setting a value in ENV changes the environment for child processes. 

%SIG $SIG{expr} 

The hash %SIG is used to set signal handlers for various signals. Example: 

    sub handler {       # 1st argument is signal name

        my($sig) = @_;

        print "Caught a SIG$sig--shutting down\n";

        close(LOG);

        exit(0);

    }

    $SIG{'INT'}  = \&handler;

    $SIG{'QUIT'} = \&handler;

    ...

    $SIG{'INT'} = 'DEFAULT';    # restore default action

    $SIG{'QUIT'} = 'IGNORE';    # ignore SIGQUIT

The %SIG array contains values for only the signals actually set within the Perl script. Here are some other examples: 

    $SIG{"PIPE"} = Plumber;     # SCARY!!

    $SIG{"PIPE"} = "Plumber";   # assumes main::Plumber (not recommended)

    $SIG{"PIPE"} = \&Plumber;   # just fine; assume current Plumber

    $SIG{"PIPE"} = Plumber();   # oops, what did Plumber() return??

The one marked scary is problematic because it's a bareword, which means sometimes it's a string representing the function, and sometimes it's going to call the subroutine call right then and there! Best to be sure and quote it or take a reference to it. *Plumber works too. See the perlsub manpage. 

If your system has the sigaction() function then signal handlers are installed using it. This means you get reliable signal handling. If your system has the SA_RESTART flag it is used when signals handlers are installed. This means that system calls for which it is supported continue rather than returning when a signal arrives. If you want your system calls to be interrupted by signal delivery then do something like this: 

    use POSIX ':signal_h';

    my $alarm = 0;

    sigaction SIGALRM, new POSIX::SigAction sub { $alarm = 1 }

        or die "Error setting SIGALRM handler: $!\n";

See the POSIX manpage. 

Certain internal hooks can be also set using the %SIG hash. The routine indicated by $SIG{__WARN__} is called when a warning message is about to be printed. The warning message is passed as the first argument. The presence of a __WARN__ hook causes the ordinary printing of warnings to STDERR to be suppressed. You can use this to save warnings in a variable, or turn warnings into fatal errors, like this: 

    local $SIG{__WARN__} = sub { die $_[0] };

    eval $proggie;

The routine indicated by $SIG{__DIE__} is called when a fatal exception is about to be thrown. The error message is passed as the first argument. When a __DIE__ hook routine returns, the exception processing continues as it would have in the absence of the hook, unless the hook routine itself exits via a goto, a loop exit, or a die(). The __DIE__ handler is explicitly disabled during the call, so that you can die from a __DIE__ handler. Similarly for __WARN__. 

Note that the $SIG{__DIE__} hook is called even inside eval()ed blocks/strings. See die and $^S for how to circumvent this. 

Note that __DIE__/__WARN__ handlers are very special in one respect: they may be called to report (probable) errors found by the parser. In such a case the parser may be in inconsistent state, so any attempt to evaluate Perl code from such a handler will probably result in a segfault. This means that calls which result/may-result in parsing Perl should be used with extreme causion, like this: 

    require Carp if defined $^S;

    Carp::confess("Something wrong") if defined &Carp::confess;

    die "Something wrong, but could not load Carp to give backtrace...

         To see backtrace try starting Perl with -MCarp switch";

Here the first line will load Carp unless it is the parser who called the handler. The second line will print backtrace and die if Carp was available. The third line will be executed only if Carp was not available. 

See die, warn and eval for additional info. 



Error Indicators

The variables $@, $!, $^E, and $? contain information about different types of error conditions that may appear during execution of Perl script. The variables are shown ordered by the ``distance'' between the subsystem which reported the error and the Perl process, and correspond to errors detected by the Perl interpreter, C library, operating system, or an external program, respectively. 

To illustrate the differences between these variables, consider the following Perl expression: 

   eval '

         open PIPE, "/cdrom/install |";

         @res = <PIPE>;

         close PIPE or die "bad pipe: $?, $!";

        ';

After execution of this statement all 4 variables may have been set. 

$@ is set if the string to be eval-ed did not compile (this may happen if open or close were imported with bad prototypes), or if Perl code executed during evaluation die()d (either implicitly, say, if open was imported from module the Fatal manpage, or the die after close was triggered). In these cases the value of $@ is the compile error, or Fatal error (which will interpolate $!!), or the argument to die (which will interpolate $! and $?!). 

When the above expression is executed, open(), <PIPE>, and close are translated to C run-time library calls. $! is set if one of these calls fails. The value is a symbolic indicator chosen by the C run-time library, say No such file or directory. 

On some systems the above C library calls are further translated to calls to the kernel. The kernel may have set more verbose error indicator that one of the handful of standard C errors. In such cases $^E contains this verbose error indicator, which may be, say, CDROM tray not closed. On systems where C library calls are identical to system calls $^E is a duplicate of $!. 

Finally, $? may be set to non-0 value if the external program /cdrom/install fails. Upper bits of the particular value may reflect specific error conditions encountered by this program (this is program-dependent), lower-bits reflect mode of failure (segfault, completion, etc.). Note that in contrast to $@, $!, and $^E, which are set only if error condition is detected, the variable $? is set on each wait or pipe close, overwriting the old value. 

For more details, see the individual descriptions at $@, $!, $^E, and $?. 



DISCLAIMER

We are painfully aware that these documents may contain incorrect links and misformatted HTML. Such bugs lie in the automatic translation process that automatically created the hundreds and hundreds of separate documents that you find here. Please do not report link or formatting bugs, because we cannot fix per-document problems. The only bug reports that will help us are those that supply working patches to the installhtml or pod2html programs, or to the Pod::HTML module itself, for which I and the entire Perl community will shower you with thanks and praises. 

If rather than formatting bugs, you encounter substantive content errors in these documents, such as mistakes in the explanations or code, please use the perlbug utility included with the Perl distribution. 

--Tom Christiansen, Perl Documentation Compiler and Editor

NAME

perlref - Perl references and nested data structures 



DESCRIPTION

Before release 5 of Perl it was difficult to represent complex data structures, because all references had to be symbolic--and even then it was difficult to refer to a variable instead of a symbol table entry. Perl now not only makes it easier to use symbolic references to variables, but also lets you have ``hard'' references to any piece of data or code. Any scalar may hold a hard reference. Because arrays and hashes contain scalars, you can now easily build arrays of arrays, arrays of hashes, hashes of arrays, arrays of hashes of functions, and so on. 

Hard references are smart--they keep track of reference counts for you, automatically freeing the thing referred to when its reference count goes to zero. (Note: the reference counts for values in self-referential or cyclic data structures may not go to zero without a little help; see Two-Phased Garbage Collection for a detailed explanation.) If that thing happens to be an object, the object is destructed. See the perlobj manpage for more about objects. (In a sense, everything in Perl is an object, but we usually reserve the word for references to objects that have been officially ``blessed'' into a class package.) 

Symbolic references are names of variables or other objects, just as a symbolic link in a Unix filesystem contains merely the name of a file. The *glob notation is a kind of symbolic reference. (Symbolic references are sometimes called ``soft references'', but please don't call them that; references are confusing enough without useless synonyms.) 

In contrast, hard references are more like hard links in a Unix file system: They are used to access an underlying object without concern for what its (other) name is. When the word ``reference'' is used without an adjective, as in the following paragraph, it is usually talking about a hard reference. 

References are easy to use in Perl. There is just one overriding principle: Perl does no implicit referencing or dereferencing. When a scalar is holding a reference, it always behaves as a simple scalar. It doesn't magically start being an array or hash or subroutine; you have to tell it explicitly to do so, by dereferencing it. 



Making References

References can be created in several ways. 

1. . 

By using the backslash operator on a variable, subroutine, or value. (This works much like the & (address-of) operator in C.) Note that this typically creates ANOTHER reference to a variable, because there's already a reference to the variable in the symbol table. But the symbol table reference might go away, and you'll still have the reference that the backslash returned. Here are some examples: 

    $scalarref = \$foo;

    $arrayref  = \@ARGV;

    $hashref   = \%ENV;

    $coderef   = \&handler;

    $globref   = \*foo;

It isn't possible to create a true reference to an IO handle (filehandle or dirhandle) using the backslash operator. The most you can get is a reference to a typeglob, which is actually a complete symbol table entry. But see the explanation of the *foo{THING} syntax below. However, you can still use type globs and globrefs as though they were IO handles. 

2. . A reference to an anonymous array can be created using square brackets: 

3.     $arrayref = [1, 2, ['a', 'b', 'c']];

Here we've created a reference to an anonymous array of three elements whose final element is itself a reference to another anonymous array of three elements. (The multidimensional syntax described later can be used to access this. For example, after the above, $arrayref->[2][1] would have the value ``b''.) 

Note that taking a reference to an enumerated list is not the same as using square brackets--instead it's the same as creating a list of references! 

    @list = (\$a, \@b, \%c);

    @list = \($a, @b, %c);      # same thing!

As a special case, \(@foo) returns a list of references to the contents of @foo, not a reference to @foo itself. Likewise for %foo. 

4. . A reference to an anonymous hash can be created using curly brackets: 

5.     $hashref = {

6.         'Adam'  => 'Eve',

7.         'Clyde' => 'Bonnie',

8.     };

Anonymous hash and array composers like these can be intermixed freely to produce as complicated a structure as you want. The multidimensional syntax described below works for these too. The values above are literals, but variables and expressions would work just as well, because assignment operators in Perl (even within local() or my()) are executable statements, not compile-time declarations. 

Because curly brackets (braces) are used for several other things including BLOCKs, you may occasionally have to disambiguate braces at the beginning of a statement by putting a + or a return in front so that Perl realizes the opening brace isn't starting a BLOCK. The economy and mnemonic value of using curlies is deemed worth this occasional extra hassle. 

For example, if you wanted a function to make a new hash and return a reference to it, you have these options: 

    sub hashem {        { @_ } }   # silently wrong

    sub hashem {       +{ @_ } }   # ok

    sub hashem { return { @_ } }   # ok

On the other hand, if you want the other meaning, you can do this: 

    sub showem {        { @_ } }   # ambiguous (currently ok, but may change)

    sub showem {       {; @_ } }   # ok

    sub showem { { return @_ } }   # ok

Note how the leading +{ and {; always serve to disambiguate the expression to mean either the HASH reference, or the BLOCK. 

9. . A reference to an anonymous subroutine can be created by using sub without a subname: 

10.     $coderef = sub { print "Boink!\n" };

Note the presence of the semicolon. Except for the fact that the code inside isn't executed immediately, a sub {} is not so much a declaration as it is an operator, like do{} or eval{}. (However, no matter how many times you execute that particular line (unless you're in an eval("...")), $coderef will still have a reference to the SAME anonymous subroutine.) 

Anonymous subroutines act as closures with respect to my() variables, that is, variables visible lexically within the current scope. Closure is a notion out of the Lisp world that says if you define an anonymous function in a particular lexical context, it pretends to run in that context even when it's called outside of the context. 

In human terms, it's a funny way of passing arguments to a subroutine when you define it as well as when you call it. It's useful for setting up little bits of code to run later, such as callbacks. You can even do object-oriented stuff with it, though Perl already provides a different mechanism to do that--see the perlobj manpage. 

You can also think of closure as a way to write a subroutine template without using eval. (In fact, in version 5.000, eval was the only way to get closures. You may wish to use ``require 5.001'' if you use closures.) 

Here's a small example of how closures works: 

    sub newprint {

        my $x = shift;

        return sub { my $y = shift; print "$x, $y!\n"; };

    }

    $h = newprint("Howdy");

    $g = newprint("Greetings");

    # Time passes...

    &$h("world");

    &$g("earthlings");

This prints 

    Howdy, world!

    Greetings, earthlings!

Note particularly that $x continues to refer to the value passed into newprint() despite the fact that the ``my $x'' has seemingly gone out of scope by the time the anonymous subroutine runs. That's what closure is all about. 

This applies only to lexical variables, by the way. Dynamic variables continue to work as they have always worked. Closure is not something that most Perl programmers need trouble themselves about to begin with. 

11. . References are often returned by special subroutines called constructors. Perl objects are just references to a special kind of object that happens to know which package it's associated with. Constructors are just special subroutines that know how to create that association. They do so by starting with an ordinary reference, and it remains an ordinary reference even while it's also being an object. Constructors are often named new() and called indirectly: 

12.     $objref = new Doggie (Tail => 'short', Ears => 'long');

But don't have to be: 

    $objref   = Doggie->new(Tail => 'short', Ears => 'long');

    use Term::Cap;

    $terminal = Term::Cap->Tgetent( { OSPEED => 9600 });

    use Tk;

    $main    = MainWindow->new();

    $menubar = $main->Frame(-relief              => "raised",

                            -borderwidth         => 2)

13. . 

References of the appropriate type can spring into existence if you dereference them in a context that assumes they exist. Because we haven't talked about dereferencing yet, we can't show you any examples yet. 

14. . A reference can be created by using a special syntax, lovingly known as the *foo{THING} syntax. *foo{THING} returns a reference to the THING slot in *foo (which is the symbol table entry which holds everything known as foo). 

15.     $scalarref = *foo{SCALAR};

16.     $arrayref  = *ARGV{ARRAY};

17.     $hashref   = *ENV{HASH};

18.     $coderef   = *handler{CODE};

19.     $ioref     = *STDIN{IO};

20.     $globref   = *foo{GLOB};

All of these are self-explanatory except for *foo{IO}. It returns the IO handle, used for file handles (open), sockets (socket and socketpair), and directory handles (opendir). For compatibility with previous versions of Perl, *foo{FILEHANDLE} is a synonym for *foo{IO}. 

*foo{THING} returns undef if that particular THING hasn't been used yet, except in the case of scalars. *foo{SCALAR} returns a reference to an anonymous scalar if $foo hasn't been used yet. This might change in a future release. 

*foo{IO} is an alternative to the \*HANDLE mechanism given in Typeglobs and Filehandles for passing filehandles into or out of subroutines, or storing into larger data structures. Its disadvantage is that it won't create a new filehandle for you. Its advantage is that you have no risk of clobbering more than you want to with a typeglob assignment, although if you assign to a scalar instead of a typeglob, you're ok. 

    splutter(*STDOUT);

    splutter(*STDOUT{IO});

    sub splutter {

        my $fh = shift;

        print $fh "her um well a hmmm\n";

    }

    $rec = get_rec(*STDIN);

    $rec = get_rec(*STDIN{IO});

    sub get_rec {

        my $fh = shift;

        return scalar <$fh>;

    }



Using References

That's it for creating references. By now you're probably dying to know how to use references to get back to your long-lost data. There are several basic methods. 

1. . 

Anywhere you'd put an identifier (or chain of identifiers) as part of a variable or subroutine name, you can replace the identifier with a simple scalar variable containing a reference of the correct type: 

    $bar = $$scalarref;

    push(@$arrayref, $filename);

    $$arrayref[0] = "January";

    $$hashref{"KEY"} = "VALUE";

    &$coderef(1,2,3);

    print $globref "output\n";

It's important to understand that we are specifically NOT dereferencing $arrayref[0] or $hashref{"KEY"} there. The dereference of the scalar variable happens BEFORE it does any key lookups. Anything more complicated than a simple scalar variable must use methods 2 or 3 below. However, a ``simple scalar'' includes an identifier that itself uses method 1 recursively. Therefore, the following prints ``howdy''. 

    $refrefref = \\\"howdy";

    print $$$$refrefref;

2. . 

Anywhere you'd put an identifier (or chain of identifiers) as part of a variable or subroutine name, you can replace the identifier with a BLOCK returning a reference of the correct type. In other words, the previous examples could be written like this: 

    $bar = ${$scalarref};

    push(@{$arrayref}, $filename);

    ${$arrayref}[0] = "January";

    ${$hashref}{"KEY"} = "VALUE";

    &{$coderef}(1,2,3);

    $globref->print("output\n");  # iff IO::Handle is loaded

Admittedly, it's a little silly to use the curlies in this case, but the BLOCK can contain any arbitrary expression, in particular, subscripted expressions: 

    &{ $dispatch{$index} }(1,2,3);      # call correct routine

Because of being able to omit the curlies for the simple case of $$x, people often make the mistake of viewing the dereferencing symbols as proper operators, and wonder about their precedence. If they were, though, you could use parentheses instead of braces. That's not the case. Consider the difference below; case 0 is a short-hand version of case 1, NOT case 2: 

    $$hashref{"KEY"}   = "VALUE";       # CASE 0

    ${$hashref}{"KEY"} = "VALUE";       # CASE 1

    ${$hashref{"KEY"}} = "VALUE";       # CASE 2

    ${$hashref->{"KEY"}} = "VALUE";     # CASE 3

Case 2 is also deceptive in that you're accessing a variable called %hashref, not dereferencing through $hashref to the hash it's presumably referencing. That would be case 3. 

3. . Subroutine calls and lookups of individual array elements arise often enough that it gets cumbersome to use method 2. As a form of syntactic sugar, the examples for method 2 may be written: 

4.     $arrayref->[0] = "January";   # Array element

5.     $hashref->{"KEY"} = "VALUE";  # Hash element

6.     $coderef->(1,2,3);            # Subroutine call

The left side of the arrow can be any expression returning a reference, including a previous dereference. Note that $array[$x] is NOT the same thing as $array->[$x] here: 

    $array[$x]->{"foo"}->[0] = "January";

This is one of the cases we mentioned earlier in which references could spring into existence when in an lvalue context. Before this statement, $array[$x] may have been undefined. If so, it's automatically defined with a hash reference so that we can look up {"foo"} in it. Likewise $array[$x]->{"foo"} will automatically get defined with an array reference so that we can look up [0] in it. This process is called autovivification. 

One more thing here. The arrow is optional BETWEEN brackets subscripts, so you can shrink the above down to 

    $array[$x]{"foo"}[0] = "January";

Which, in the degenerate case of using only ordinary arrays, gives you multidimensional arrays just like C's: 

    $score[$x][$y][$z] += 42;

Well, okay, not entirely like C's arrays, actually. C doesn't know how to grow its arrays on demand. Perl does. 

7. . If a reference happens to be a reference to an object, then there are probably methods to access the things referred to, and you should probably stick to those methods unless you're in the class package that defines the object's methods. In other words, be nice, and don't violate the object's encapsulation without a very good reason. Perl does not enforce encapsulation. We are not totalitarians here. We do expect some basic civility though. 

The ref() operator may be used to determine what type of thing the reference is pointing to. See the perlfunc manpage. 

The bless() operator may be used to associate the object a reference points to with a package functioning as an object class. See the perlobj manpage. 

A typeglob may be dereferenced the same way a reference can, because the dereference syntax always indicates the kind of reference desired. So ${*foo} and ${\$foo} both indicate the same scalar variable. 

Here's a trick for interpolating a subroutine call into a string: 

    print "My sub returned @{[mysub(1,2,3)]} that time.\n";

The way it works is that when the @{...} is seen in the double-quoted string, it's evaluated as a block. The block creates a reference to an anonymous array containing the results of the call to mysub(1,2,3). So the whole block returns a reference to an array, which is then dereferenced by @{...} and stuck into the double-quoted string. This chicanery is also useful for arbitrary expressions: 

    print "That yields @{[$n + 5]} widgets\n";



Symbolic references

We said that references spring into existence as necessary if they are undefined, but we didn't say what happens if a value used as a reference is already defined, but ISN'T a hard reference. If you use it as a reference in this case, it'll be treated as a symbolic reference. That is, the value of the scalar is taken to be the NAME of a variable, rather than a direct link to a (possibly) anonymous value. 

People frequently expect it to work like this. So it does. 

    $name = "foo";

    $$name = 1;                 # Sets $foo

    ${$name} = 2;               # Sets $foo

    ${$name x 2} = 3;           # Sets $foofoo

    $name->[0] = 4;             # Sets $foo[0]

    @$name = ();                # Clears @foo

    &$name();                   # Calls &foo() (as in Perl 4)

    $pack = "THAT";

    ${"${pack}::$name"} = 5;    # Sets $THAT::foo without eval

This is very powerful, and slightly dangerous, in that it's possible to intend (with the utmost sincerity) to use a hard reference, and accidentally use a symbolic reference instead. To protect against that, you can say 

    use strict 'refs';

and then only hard references will be allowed for the rest of the enclosing block. An inner block may countermand that with 

    no strict 'refs';

Only package variables (globals, even if localized) are visible to symbolic references. Lexical variables (declared with my()) aren't in a symbol table, and thus are invisible to this mechanism. For example: 

    local $value = 10;

    $ref = \$value;

    {

        my $value = 20;

        print $$ref;

    }

This will still print 10, not 20. Remember that local() affects package variables, which are all ``global'' to the package. 



Not-so-symbolic references

A new feature contributing to readability in perl version 5.001 is that the brackets around a symbolic reference behave more like quotes, just as they always have within a string. That is, 

    $push = "pop on ";

    print "${push}over";

has always meant to print ``pop on over'', despite the fact that push is a reserved word. This has been generalized to work the same outside of quotes, so that 

    print ${push} . "over";

and even 

    print ${ push } . "over";

will have the same effect. (This would have been a syntax error in Perl 5.000, though Perl 4 allowed it in the spaceless form.) Note that this construct is not considered to be a symbolic reference when you're using strict refs: 

    use strict 'refs';

    ${ bareword };      # Okay, means $bareword.

    ${ "bareword" };    # Error, symbolic reference.

Similarly, because of all the subscripting that is done using single words, we've applied the same rule to any bareword that is used for subscripting a hash. So now, instead of writing 

    $array{ "aaa" }{ "bbb" }{ "ccc" }

you can write just 

    $array{ aaa }{ bbb }{ ccc }

and not worry about whether the subscripts are reserved words. In the rare event that you do wish to do something like 

    $array{ shift }

you can force interpretation as a reserved word by adding anything that makes it more than a bareword: 

    $array{ shift() }

    $array{ +shift }

    $array{ shift @_ }

The -w switch will warn you if it interprets a reserved word as a string. But it will no longer warn you about using lowercase words, because the string is effectively quoted. 



Pseudo-hashes: Using an array as a hash

WARNING: This section describes an experimental feature. Details may change without notice in future versions. 

Beginning with release 5.005 of Perl you can use an array reference in some contexts that would normally require a hash reference. This allows you to access array elements using symbolic names, as if they were fields in a structure. 

For this to work, the array must contain extra information. The first element of the array has to be a hash reference that maps field names to array indices. Here is an example: 

   $struct = [{foo => 1, bar => 2}, "FOO", "BAR"];

   $struct->{foo};  # same as $struct->[1], i.e. "FOO"

   $struct->{bar};  # same as $struct->[2], i.e. "BAR"

   keys %$struct;   # will return ("foo", "bar") in some order

   values %$struct; # will return ("FOO", "BAR") in same some order

   while (my($k,$v) = each %$struct) {

       print "$k => $v\n";

   }

Perl will raise an exception if you try to delete keys from a pseudo-hash or try to access nonexistent fields. For better performance, Perl can also do the translation from field names to array indices at compile time for typed object references. See the fields manpage. 



Function Templates

As explained above, a closure is an anonymous function with access to the lexical variables visible when that function was compiled. It retains access to those variables even though it doesn't get run until later, such as in a signal handler or a Tk callback. 

Using a closure as a function template allows us to generate many functions that act similarly. Suppopose you wanted functions named after the colors that generated HTML font changes for the various colors: 

    print "Be ", red("careful"), "with that ", green("light");

The red() and green() functions would be very similar. To create these, we'll assign a closure to a typeglob of the name of the function we're trying to build. 

    @colors = qw(red blue green yellow orange purple violet);

    for my $name (@colors) {

        no strict 'refs';       # allow symbol table manipulation

        *$name = *{uc $name} = sub { "<FONT COLOR='$name'>@_</FONT>" };

    } 

Now all those different functions appear to exist independently. You can call red(), RED(), blue(), BLUE(), green(), etc. This technique saves on both compile time and memory use, and is less error-prone as well, since syntax checks happen at compile time. It's critical that any variables in the anonymous subroutine be lexicals in order to create a proper closure. That's the reasons for the my on the loop iteration variable. 

This is one of the only places where giving a prototype to a closure makes much sense. If you wanted to impose scalar context on the arguments of these functions (probably not a wise idea for this particular example), you could have written it this way instead: 

    *$name = sub ($) { "<FONT COLOR='$name'>$_[0]</FONT>" };

However, since prototype checking happens at compile time, the assignment above happens too late to be of much use. You could address this by putting the whole loop of assignments within a BEGIN block, forcing it to occur during compilation. 

Access to lexicals that change over type--like those in the for loop above--only works with closures, not general subroutines. In the general case, then, named subroutines do not nest properly, although anonymous ones do. If you are accustomed to using nested subroutines in other programming languages with their own private variables, you'll have to work at it a bit in Perl. The intuitive coding of this kind of thing incurs mysterious warnings about ``will not stay shared''. For example, this won't work: 

    sub outer {

        my $x = $_[0] + 35;

        sub inner { return $x * 19 }   # WRONG

        return $x + inner();

    } 

A work-around is the following: 

    sub outer {

        my $x = $_[0] + 35;

        local *inner = sub { return $x * 19 };

        return $x + inner();

    } 

Now inner() can only be called from within outer(), because of the temporary assignments of the closure (anonymous subroutine). But when it does, it has normal access to the lexical variable $x from the scope of outer(). 

This has the interesting effect of creating a function local to another function, something not normally supported in Perl. 



WARNING

You may not (usefully) use a reference as the key to a hash. It will be converted into a string: 

    $x{ \$a } = $a;

If you try to dereference the key, it won't do a hard dereference, and you won't accomplish what you're attempting. You might want to do something more like 

    $r = \@a;

    $x{ $r } = $r;

And then at least you can use the values(), which will be real refs, instead of the keys(), which won't. 

The standard Tie::RefHash module provides a convenient workaround to this. 



SEE ALSO

Besides the obvious documents, source code can be instructive. Some rather pathological examples of the use of references can be found in the t/op/ref.t regression test in the Perl source directory. 

See also the perldsc manpage and the perllol manpage for how to use references to create complex data structures, and the perltoot manpage, the perlobj manpage, and the perlbot manpage for how to use them to create objects. 



DISCLAIMER

We are painfully aware that these documents may contain incorrect links and misformatted HTML. Such bugs lie in the automatic translation process that automatically created the hundreds and hundreds of separate documents that you find here. Please do not report link or formatting bugs, because we cannot fix per-document problems. The only bug reports that will help us are those that supply working patches to the installhtml or pod2html programs, or to the Pod::HTML module itself, for which I and the entire Perl community will shower you with thanks and praises. 

If rather than formatting bugs, you encounter substantive content errors in these documents, such as mistakes in the explanations or code, please use the perlbug utility included with the Perl distribution. 

--Tom Christiansen, Perl Documentation Compiler and Editor

NAME

perlfaq4 - Data Manipulation ($Revision: 1.26 $, $Date: 1998/08/05 12:04:00 $) 



DESCRIPTION

The section of the FAQ answers question related to the manipulation of data as numbers, dates, strings, arrays, hashes, and miscellaneous data issues. 



Data: Numbers



Why am I getting long decimals (eg, 19.9499999999999) instead of the numbers I should be getting (eg, 19.95)?

The infinite set that a mathematician thinks of as the real numbers can only be approximate on a computer, since the computer only has a finite number of bits to store an infinite number of, um, numbers. 

Internally, your computer represents floating-point numbers in binary. Floating-point numbers read in from a file or appearing as literals in your program are converted from their decimal floating-point representation (eg, 19.95) to the internal binary representation. 

However, 19.95 can't be precisely represented as a binary floating-point number, just like 1/3 can't be exactly represented as a decimal floating-point number. The computer's binary representation of 19.95, therefore, isn't exactly 19.95. 

When a floating-point number gets printed, the binary floating-point representation is converted back to decimal. These decimal numbers are displayed in either the format you specify with printf(), or the current output format for numbers (see $# if you use print. $# has a different default value in Perl5 than it did in Perl4. Changing $# yourself is deprecated. 

This affects all computer languages that represent decimal floating-point numbers in binary, not just Perl. Perl provides arbitrary-precision decimal numbers with the Math::BigFloat module (part of the standard Perl distribution), but mathematical operations are consequently slower. 

To get rid of the superfluous digits, just use a format (eg, printf("%.2f", 19.95)) to get the required precision. See Floating-point Arithmetic. 



Why isn't my octal data interpreted correctly?

Perl only understands octal and hex numbers as such when they occur as literals in your program. If they are read in from somewhere and assigned, no automatic conversion takes place. You must explicitly use oct() or hex() if you want the values converted. oct() interprets both hex (``0x350'') numbers and octal ones (``0350'' or even without the leading ``0'', like ``377''), while hex() only converts hexadecimal ones, with or without a leading ``0x'', like ``0x255'', ``3A'', ``ff'', or ``deadbeef''. 

This problem shows up most often when people try using chmod(), mkdir(), umask(), or sysopen(), which all want permissions in octal. 

    chmod(644,  $file); # WRONG -- perl -w catches this

    chmod(0644, $file); # right



Does perl have a round function? What about ceil() and floor()? Trig functions?

Remember that int() merely truncates toward 0. For rounding to a certain number of digits, sprintf() or printf() is usually the easiest route. 

    printf("%.3f", 3.1415926535);       # prints 3.142

The POSIX module (part of the standard perl distribution) implements ceil(), floor(), and a number of other mathematical and trigonometric functions. 

    use POSIX;

    $ceil   = ceil(3.5);                        # 4

    $floor  = floor(3.5);                       # 3

In 5.000 to 5.003 Perls, trigonometry was done in the Math::Complex module. With 5.004, the Math::Trig module (part of the standard perl distribution) implements the trigonometric functions. Internally it uses the Math::Complex module and some functions can break out from the real axis into the complex plane, for example the inverse sine of 2. 

Rounding in financial applications can have serious implications, and the rounding method used should be specified precisely. In these cases, it probably pays not to trust whichever system rounding is being used by Perl, but to instead implement the rounding function you need yourself. 



How do I convert bits into ints?

To turn a string of 1s and 0s like 10110110 into a scalar containing its binary value, use the pack() function (documented in pack): 

    $decimal = pack('B8', '10110110');

Here's an example of going the other way: 

    $binary_string = join('', unpack('B*', "\x29"));



How do I multiply matrices?

Use the Math::Matrix or Math::MatrixReal modules (available from CPAN) or the PDL extension (also available from CPAN). 



How do I perform an operation on a series of integers?

To call a function on each element in an array, and collect the results, use: 

    @results = map { my_func($_) } @array;

For example: 

    @triple = map { 3 * $_ } @single;

To call a function on each element of an array, but ignore the results: 

    foreach $iterator (@array) {

        &my_func($iterator);

    }

To call a function on each integer in a (small) range, you can use: 

    @results = map { &my_func($_) } (5 .. 25);

but you should be aware that the .. operator creates an array of all integers in the range. This can take a lot of memory for large ranges. Instead use: 

    @results = ();

    for ($i=5; $i < 500_005; $i++) {

        push(@results, &my_func($i));

    }



How can I output Roman numerals?

Get the http://www.perl.com/CPAN/modules/by-module/Roman module. 



Why aren't my random numbers random?

The short explanation is that you're getting pseudorandom numbers, not random ones, because computers are good at being predictable and bad at being random (despite appearances caused by bugs in your programs :-). A longer explanation is available on http://www.perl.com/CPAN/doc/FMTEYEWTK/random, courtesy of Tom Phoenix. John von Neumann said, ``Anyone who attempts to generate random numbers by deterministic means is, of course, living in a state of sin.'' 

You should also check out the Math::TrulyRandom module from CPAN. It uses the imperfections in your system's timer to generate random numbers, but this takes quite a while. If you want a better pseudorandom generator than comes with your operating system, look at ``Numerical Recipes in C'' at http://nr.harvard.edu/nr/bookc.html . 



Data: Dates



How do I find the week-of-the-year/day-of-the-year?

The day of the year is in the array returned by localtime() (see localtime): 

    $day_of_year = (localtime(time()))[7];

or more legibly (in 5.004 or higher): 

    use Time::localtime;

    $day_of_year = localtime(time())->yday;

You can find the week of the year by dividing this by 7: 

    $week_of_year = int($day_of_year / 7);

Of course, this believes that weeks start at zero. The Date::Calc module from CPAN has a lot of date calculation functions, including day of the year, week of the year, and so on. Note that not all business consider ``week 1'' to be the same; for example, American business often consider the first week with a Monday in it to be Work Week #1, despite ISO 8601, which consider WW1 to be the frist week with a Thursday in it. 



How can I compare two dates and find the difference?

If you're storing your dates as epoch seconds then simply subtract one from the other. If you've got a structured date (distinct year, day, month, hour, minute, seconds values) then use one of the Date::Manip and Date::Calc modules from CPAN. 



How can I take a string and turn it into epoch seconds?

If it's a regular enough string that it always has the same format, you can split it up and pass the parts to timelocal in the standard Time::Local module. Otherwise, you should look into the Date::Calc and Date::Manip modules from CPAN. 



How can I find the Julian Day?

Neither Date::Manip nor Date::Calc deal with Julian days. Instead, there is an example of Julian date calculation that should help you in http://www.perl.com/CPAN/authors/David_Muir_Sharnoff/modules/Time/JulianDay.pm.gz . 



Does Perl have a year 2000 problem? Is Perl Y2K compliant?

Short answer: No, Perl does not have a Year 2000 problem. Yes, Perl is Y2K compliant. The programmers you're hired to use it, however, probably are not. 

Long answer: Perl is just as Y2K compliant as your pencil--no more, and no less. The date and time functions supplied with perl (gmtime and localtime) supply adequate information to determine the year well beyond 2000 (2038 is when trouble strikes for 32-bit machines). The year returned by these functions when used in an array context is the year minus 1900. For years between 1910 and 1999 this happens to be a 2-digit decimal number. To avoid the year 2000 problem simply do not treat the year as a 2-digit number. It isn't. 

When gmtime() and localtime() are used in scalar context they return a timestamp string that contains a fully-expanded year. For example, $timestamp = gmtime(1005613200) sets $timestamp to ``Tue Nov 13 01:00:00 2001''. There's no year 2000 problem here. 

That doesn't mean that Perl can't be used to create non-Y2K compliant programs. It can. But so can your pencil. It's the fault of the user, not the language. At the risk of inflaming the NRA: ``Perl doesn't break Y2K, people do.'' See http://language.perl.com/news/y2k.html for a longer exposition. 



Data: Strings



How do I validate input?

The answer to this question is usually a regular expression, perhaps with auxiliary logic. See the more specific questions (numbers, mail addresses, etc.) for details. 



How do I unescape a string?

It depends just what you mean by ``escape''. URL escapes are dealt with in the perlfaq9 manpage. Shell escapes with the backslash (\) character are removed with: 

    s/\\(.)/$1/g;

This won't expand "\n" or "\t" or any other special escapes. 



How do I remove consecutive pairs of characters?

To turn "abbcccd" into "abccd": 

    s/(.)\1/$1/g;



How do I expand function calls in a string?

This is documented in the perlref manpage. In general, this is fraught with quoting and readability problems, but it is possible. To interpolate a subroutine call (in list context) into a string: 

    print "My sub returned @{[mysub(1,2,3)]} that time.\n";

If you prefer scalar context, similar chicanery is also useful for arbitrary expressions: 

    print "That yields ${\($n + 5)} widgets\n";

Version 5.004 of Perl had a bug that gave list context to the expression in ${...}, but this is fixed in version 5.005. 

See also ``How can I expand variables in text strings?'' in this section of the FAQ. 



How do I find matching/nesting anything?

This isn't something that can be done in one regular expression, no matter how complicated. To find something between two single characters, a pattern like /x([^x]*)x/ will get the intervening bits in $1. For multiple ones, then something more like /alpha(.*?)omega/ would be needed. But none of these deals with nested patterns, nor can they. For that you'll have to write a parser. 

If you are serious about writing a parser, there are a number of modules or oddities that will make your life a lot easier. There is the CPAN module Parse::RecDescent, the standard module Text::Balanced, the byacc program, and Mark-Jason Dominus's excellent py tool at http://www.plover.com/~mjd/perl/py/ . 

One simple destructive, inside-out approach that you might try is to pull out the smallest nesting parts one at a time: 

    while (s//BEGIN((?:(?!BEGIN)(?!END).)*)END/gs) {

        # do something with $1

    } 



How do I reverse a string?

Use reverse() in scalar context, as documented in reverse. 

    $reversed = reverse $string;



How do I expand tabs in a string?

You can do it yourself: 

    1 while $string =~ s/\t+/' ' x (length($&) * 8 - length($`) % 8)/e;

Or you can just use the Text::Tabs module (part of the standard perl distribution). 

    use Text::Tabs;

    @expanded_lines = expand(@lines_with_tabs);



How do I reformat a paragraph?

Use Text::Wrap (part of the standard perl distribution): 

    use Text::Wrap;

    print wrap("\t", '  ', @paragraphs);

The paragraphs you give to Text::Wrap should not contain embedded newlines. Text::Wrap doesn't justify the lines (flush-right). 



How can I access/change the first N letters of a string?

There are many ways. If you just want to grab a copy, use substr(): 

    $first_byte = substr($a, 0, 1);

If you want to modify part of a string, the simplest way is often to use substr() as an lvalue: 

    substr($a, 0, 3) = "Tom";

Although those with a pattern matching kind of thought process will likely prefer: 

    $a =~ s/^.../Tom/;



How do I change the Nth occurrence of something?

You have to keep track of N yourself. For example, let's say you want to change the fifth occurrence of "whoever" or "whomever" into "whosoever" or "whomsoever", case insensitively. 

    $count = 0;

    s{((whom?)ever)}{

        ++$count == 5           # is it the 5th?

            ? "${2}soever"      # yes, swap

            : $1                # renege and leave it there

    }igex;

In the more general case, you can use the /g modifier in a while loop, keeping count of matches. 

    $WANT = 3;

    $count = 0;

    while (/(\w+)\s+fish\b/gi) {

        if (++$count == $WANT) {

            print "The third fish is a $1 one.\n";

            # Warning: don't `last' out of this loop

        }

    }

That prints out: "The third fish is a red one." You can also use a repetition count and repeated pattern like this: 

    /(?:\w+\s+fish\s+){2}(\w+)\s+fish/i;



How can I count the number of occurrences of a substring within a string?

There are a number of ways, with varying efficiency: If you want a count of a certain single character (X) within a string, you can use the tr/// function like so: 

    $string = "ThisXlineXhasXsomeXx'sXinXit":

    $count = ($string =~ tr/X//);

    print "There are $count X charcters in the string";

This is fine if you are just looking for a single character. However, if you are trying to count multiple character substrings within a larger string, tr/// won't work. What you can do is wrap a while() loop around a global pattern match. For example, let's count negative integers: 

    $string = "-9 55 48 -2 23 -76 4 14 -44";

    while ($string =~ /-\d+/g) { $count++ }

    print "There are $count negative numbers in the string";



How do I capitalize all the words on one line?

To make the first letter of each word upper case: 

        $line =~ s/\b(\w)/\U$1/g;

This has the strange effect of turning ``don't do it'' into ``Don'T Do It''. Sometimes you might want this, instead (Suggested by Brian Foy): 

    $string =~ s/ (

                 (^\w)    #at the beginning of the line

                   |      # or

                 (\s\w)   #preceded by whitespace

                   )

                /\U$1/xg;

    $string =~ /([\w']+)/\u\L$1/g;

To make the whole line upper case: 

        $line = uc($line);

To force each word to be lower case, with the first letter upper case: 

        $line =~ s/(\w+)/\u\L$1/g;

You can (and probably should) enable locale awareness of those characters by placing a use locale pragma in your program. See the perllocale manpage for endless details on locales. 



How can I split a [character] delimited string except when inside [character]? (Comma-separated files)

Take the example case of trying to split a string that is comma-separated into its different fields. (We'll pretend you said comma-separated, not comma-delimited, which is different and almost never what you mean.) You can't use split(/,/) because you shouldn't split if the comma is inside quotes. For example, take a data line like this: 

    SAR001,"","Cimetrix, Inc","Bob Smith","CAM",N,8,1,0,7,"Error, Core Dumped"

Due to the restriction of the quotes, this is a fairly complex problem. Thankfully, we have Jeffrey Friedl, author of a highly recommended book on regular expressions, to handle these for us. He suggests (assuming your string is contained in $text): 

     @new = ();

     push(@new, $+) while $text =~ m{

         "([^\"\\]*(?:\\.[^\"\\]*)*)",?  # groups the phrase inside the quotes

       | ([^,]+),?

       | ,

     }gx;

     push(@new, undef) if substr($text,-1,1) eq ',';

If you want to represent quotation marks inside a quotation-mark-delimited field, escape them with backslashes (eg, "like \"this\"". Unescaping them is a task addressed earlier in this section. 

Alternatively, the Text::ParseWords module (part of the standard perl distribution) lets you say: 

    use Text::ParseWords;

    @new = quotewords(",", 0, $text);



How do I strip blank space from the beginning/end of a string?

Although the simplest approach would seem to be: 

    $string =~ s/^\s*(.*?)\s*$/$1/;

This is unneccesarily slow, destructive, and fails with embedded newlines. It is much better faster to do this in two steps: 

    $string =~ s/^\s+//;

    $string =~ s/\s+$//;

Or more nicely written as: 

    for ($string) {

        s/^\s+//;

        s/\s+$//;

    }

This idiom takes advantage of the foreach loop's aliasing behavior to factor out common code. You can do this on several strings at once, or arrays, or even the values of a hash if you use a slide: 

    # trim whitespace in the scalar, the array, 

    # and all the values in the hash

    foreach ($scalar, @array, @hash{keys %hash}) {

        s/^\s+//;

        s/\s+$//;

    }



How do I extract selected columns from a string?

Use substr() or unpack(), both documented in the perlfunc manpage. If you prefer thinking in terms of columns instead of widths, you can use this kind of thing: 

    # determine the unpack format needed to split Linux ps output

    # arguments are cut columns

    my $fmt = cut2fmt(8, 14, 20, 26, 30, 34, 41, 47, 59, 63, 67, 72);

    sub cut2fmt { 

        my(@positions) = @_;

        my $template  = '';

        my $lastpos   = 1;

        for my $place (@positions) {

            $template .= "A" . ($place - $lastpos) . " "; 

            $lastpos   = $place;

        }

        $template .= "A*";

        return $template;

    }



How do I find the soundex value of a string?

Use the standard Text::Soundex module distributed with perl. 



How can I expand variables in text strings?

Let's assume that you have a string like: 

    $text = 'this has a $foo in it and a $bar';

If those were both global variables, then this would suffice: 

    $text =~ s/\$(\w+)/${$1}/g;

But since they are probably lexicals, or at least, they could be, you'd have to do this: 

    $text =~ s/(\$\w+)/$1/eeg;

    die if $@;                  # needed on /ee, not /e

It's probably better in the general case to treat those variables as entries in some special hash. For example: 

    %user_defs = ( 

        foo  => 23,

        bar  => 19,

    );

    $text =~ s/\$(\w+)/$user_defs{$1}/g;

See also ``How do I expand function calls in a string?'' in this section of the FAQ. 



What's wrong with always quoting "$vars"?

The problem is that those double-quotes force stringification, coercing numbers and references into strings, even when you don't want them to be. 

If you get used to writing odd things like these: 

    print "$var";       # BAD

    $new = "$old";      # BAD

    somefunc("$var");   # BAD

You'll be in trouble. Those should (in 99.8% of the cases) be the simpler and more direct: 

    print $var;

    $new = $old;

    somefunc($var);

Otherwise, besides slowing you down, you're going to break code when the thing in the scalar is actually neither a string nor a number, but a reference: 

    func(\@array);

    sub func {

        my $aref = shift;

        my $oref = "$aref";  # WRONG

    }

You can also get into subtle problems on those few operations in Perl that actually do care about the difference between a string and a number, such as the magical ++ autoincrement operator or the syscall() function. 

Stringification also destroys arrays. 

    @lines = `command`;

    print "@lines";             # WRONG - extra blanks

    print @lines;               # right



Why don't my <<HERE documents work?

Check for these three things: 

1. There must be no space after the << part. 

2. There (probably) should be a semicolon at the end. 

3. You can't (easily) have any space in front of the tag. 

If you want to indent the text in the here document, you can do this: 

    # all in one

    ($VAR = <<HERE_TARGET) =~ s/^\s+//gm;

        your text

        goes here

    HERE_TARGET

But the HERE_TARGET must still be flush against the margin. If you want that indented also, you'll have to quote in the indentation. 

    ($quote = <<'    FINIS') =~ s/^\s+//gm;

            ...we will have peace, when you and all your works have

            perished--and the works of your dark master to whom you

            would deliver us. You are a liar, Saruman, and a corrupter

            of men's hearts.  --Theoden in /usr/src/perl/taint.c

        FINIS

    $quote =~ s/\s*--/\n--/;

A nice general-purpose fixer-upper function for indented here documents follows. It expects to be called with a here document as its argument. It looks to see whether each line begins with a common substring, and if so, strips that off. Otherwise, it takes the amount of leading white space found on the first line and removes that much off each subsequent line. 

    sub fix {

        local $_ = shift;

        my ($white, $leader);  # common white space and common leading string

        if (/^\s*(?:([^\w\s]+)(\s*).*\n)(?:\s*\1\2?.*\n)+$/) {

            ($white, $leader) = ($2, quotemeta($1));

        } else {

            ($white, $leader) = (/^(\s+)/, '');

        }

        s/^\s*?$leader(?:$white)?//gm;

        return $_;

    }

This works with leading special strings, dynamically determined: 

    $remember_the_main = fix<<'    MAIN_INTERPRETER_LOOP';

        @@@ int

        @@@ runops() {

        @@@     SAVEI32(runlevel);

        @@@     runlevel++;

        @@@     while ( op = (*op->op_ppaddr)() ) ;

        @@@     TAINT_NOT;

        @@@     return 0;

        @@@ }

    MAIN_INTERPRETER_LOOP

Or with a fixed amount of leading white space, with remaining indentation correctly preserved: 

    $poem = fix<<EVER_ON_AND_ON;

       Now far ahead the Road has gone,

          And I must follow, if I can,

       Pursuing it with eager feet,

          Until it joins some larger way

       Where many paths and errands meet.

          And whither then? I cannot say.

                --Bilbo in /usr/src/perl/pp_ctl.c

    EVER_ON_AND_ON



Data: Arrays



What is the difference between $array[1] and @array[1]?

The former is a scalar value, the latter an array slice, which makes it a list with one (scalar) value. You should use $ when you want a scalar value (most of the time) and @ when you want a list with one scalar value in it (very, very rarely; nearly never, in fact). 

Sometimes it doesn't make a difference, but sometimes it does. For example, compare: 

    $good[0] = `some program that outputs several lines`;

with 

    @bad[0]  = `same program that outputs several lines`;

The -w flag will warn you about these matters. 



How can I extract just the unique elements of an array?

There are several possible ways, depending on whether the array is ordered and whether you wish to preserve the ordering. 

a) If @in is sorted, and you want @out to be sorted: (this assumes all true values in the array) 

    $prev = 'nonesuch';

    @out = grep($_ ne $prev && ($prev = $_), @in);

This is nice in that it doesn't use much extra memory, simulating uniq(1)'s behavior of removing only adjacent duplicates. It's less nice in that it won't work with false values like undef, 0, or ``''; ``0 but true'' is ok, though. 

b) If you don't know whether @in is sorted: 

    undef %saw;

    @out = grep(!$saw{$_}++, @in);

c) Like (b), but @in contains only small integers: 

    @out = grep(!$saw[$_]++, @in);

d) A way to do (b) without any loops or greps: 

    undef %saw;

    @saw{@in} = ();

    @out = sort keys %saw;  # remove sort if undesired

e) Like (d), but @in contains only small positive integers: 

    undef @ary;

    @ary[@in] = @in;

    @out = @ary;



How can I tell whether a list or array contains a certain element?

Hearing the word ``in'' is an indication that you probably should have used a hash, not a list or array, to store your data. Hashes are designed to answer this question quickly and efficiently. Arrays aren't. 

That being said, there are several ways to approach this. If you are going to make this query many times over arbitrary string values, the fastest way is probably to invert the original array and keep an associative array lying about whose keys are the first array's values. 

    @blues = qw/azure cerulean teal turquoise lapis-lazuli/;

    undef %is_blue;

    for (@blues) { $is_blue{$_} = 1 }

Now you can check whether $is_blue{$some_color}. It might have been a good idea to keep the blues all in a hash in the first place. 

If the values are all small integers, you could use a simple indexed array. This kind of an array will take up less space: 

    @primes = (2, 3, 5, 7, 11, 13, 17, 19, 23, 29, 31);

    undef @is_tiny_prime;

    for (@primes) { $is_tiny_prime[$_] = 1; }

Now you check whether $is_tiny_prime[$some_number]. 

If the values in question are integers instead of strings, you can save quite a lot of space by using bit strings instead: 

    @articles = ( 1..10, 150..2000, 2017 );

    undef $read;

    for (@articles) { vec($read,$_,1) = 1 }

Now check whether vec($read,$n,1) is true for some $n. 

Please do not use 

    $is_there = grep $_ eq $whatever, @array;

or worse yet 

    $is_there = grep /$whatever/, @array;

These are slow (checks every element even if the first matches), inefficient (same reason), and potentially buggy (what if there are regexp characters in $whatever?). 



How do I compute the difference of two arrays? How do I compute the intersection of two arrays?

Use a hash. Here's code to do both and more. It assumes that each element is unique in a given array: 

    @union = @intersection = @difference = ();

    %count = ();

    foreach $element (@array1, @array2) { $count{$element}++ }

    foreach $element (keys %count) {

        push @union, $element;

        push @{ $count{$element} > 1 ? \@intersection : \@difference }, $element;

    }



How do I find the first array element for which a condition is true?

You can use this if you care about the index: 

    for ($i=0; $i < @array; $i++) {

        if ($array[$i] eq "Waldo") {

            $found_index = $i;

            last;

        }

    }

Now $found_index has what you want. 



How do I handle linked lists?

In general, you usually don't need a linked list in Perl, since with regular arrays, you can push and pop or shift and unshift at either end, or you can use splice to add and/or remove arbitrary number of elements at arbitrary points. Both pop and shift are both O(1) operations on perl's dynamic arrays. In the absence of shifts and pops, push in general needs to reallocate on the order every log(N) times, and unshift will need to copy pointers each time. 

If you really, really wanted, you could use structures as described in the perldsc manpage or the perltoot manpage and do just what the algorithm book tells you to do. 



How do I handle circular lists?

Circular lists could be handled in the traditional fashion with linked lists, or you could just do something like this with an array: 

    unshift(@array, pop(@array));  # the last shall be first

    push(@array, shift(@array));   # and vice versa



How do I shuffle an array randomly?

Use this: 

    # fisher_yates_shuffle( \@array ) : 

    # generate a random permutation of @array in place

    sub fisher_yates_shuffle {

        my $array = shift;

        my $i;

        for ($i = @$array; --$i; ) {

            my $j = int rand ($i+1);

            next if $i == $j;

            @$array[$i,$j] = @$array[$j,$i];

        }

    }

    fisher_yates_shuffle( \@array );    # permutes @array in place

You've probably seen shuffling algorithms that works using splice, randomly picking another element to swap the current element with: 

    srand;

    @new = ();

    @old = 1 .. 10;  # just a demo

    while (@old) {

        push(@new, splice(@old, rand @old, 1));

    }

This is bad because splice is already O(N), and since you do it N times, you just invented a quadratic algorithm; that is, O(N**2). This does not scale, although Perl is so efficient that you probably won't notice this until you have rather largish arrays. 



How do I process/modify each element of an array?

Use for/foreach: 

    for (@lines) {

        s/foo/bar/;     # change that word

        y/XZ/ZX/;       # swap those letters

    }

Here's another; let's compute spherical volumes: 

    for (@volumes = @radii) {   # @volumes has changed parts

        $_ **= 3;

        $_ *= (4/3) * 3.14159;  # this will be constant folded

    }

If you want to do the same thing to modify the values of the hash, you may not use the values function, oddly enough. You need a slice: 

    for $orbit ( @orbits{keys %orbits} ) {

        ($orbit **= 3) *= (4/3) * 3.14159; 

    }



How do I select a random element from an array?

Use the rand() function (see rand): 

    # at the top of the program:

    srand;                      # not needed for 5.004 and later

    # then later on

    $index   = rand @array;

    $element = $array[$index];

Make sure you only call srand once per program, if then. If you are calling it more than once (such as before each call to rand), you're almost certainly doing something wrong. 



How do I permute N elements of a list?

Here's a little program that generates all permutations of all the words on each line of input. The algorithm embodied in the permute() function should work on any list: 

    #!/usr/bin/perl -n

    # tsc-permute: permute each word of input

    permute([split], []);

    sub permute {

        my @items = @{ $_[0] };

        my @perms = @{ $_[1] };

        unless (@items) {

            print "@perms\n";

        } else {

            my(@newitems,@newperms,$i);

            foreach $i (0 .. $#items) {

                @newitems = @items;

                @newperms = @perms;

                unshift(@newperms, splice(@newitems, $i, 1));

                permute([@newitems], [@newperms]);

            }

        }

    }



How do I sort an array by (anything)?

Supply a comparison function to sort() (described in sort): 

    @list = sort { $a <=> $b } @list;

The default sort function is cmp, string comparison, which would sort (1, 2, 10) into (1, 10, 2). <=>, used above, is the numerical comparison operator. 

If you have a complicated function needed to pull out the part you want to sort on, then don't do it inside the sort function. Pull it out first, because the sort BLOCK can be called many times for the same element. Here's an example of how to pull out the first word after the first number on each item, and then sort those words case-insensitively. 

    @idx = ();

    for (@data) {

        ($item) = /\d+\s*(\S+)/;

        push @idx, uc($item);

    }

    @sorted = @data[ sort { $idx[$a] cmp $idx[$b] } 0 .. $#idx ];

Which could also be written this way, using a trick that's come to be known as the Schwartzian Transform: 

    @sorted = map  { $_->[0] }

              sort { $a->[1] cmp $b->[1] }

              map  { [ $_, uc((/\d+\s*(\S+)/ )[0] ] } @data;

If you need to sort on several fields, the following paradigm is useful. 

    @sorted = sort { field1($a) <=> field1($b) ||

                     field2($a) cmp field2($b) ||

                     field3($a) cmp field3($b)

                   }     @data;

This can be conveniently combined with precalculation of keys as given above. 

See http://www.perl.com/CPAN/doc/FMTEYEWTK/sort.html for more about this approach. 

See also the question below on sorting hashes. 



How do I manipulate arrays of bits?

Use pack() and unpack(), or else vec() and the bitwise operations. 

For example, this sets $vec to have bit N set if $ints[N] was set: 

    $vec = '';

    foreach(@ints) { vec($vec,$_,1) = 1 }

And here's how, given a vector in $vec, you can get those bits into your @ints array: 

    sub bitvec_to_list {

        my $vec = shift;

        my @ints;

        # Find null-byte density then select best algorithm

        if ($vec =~ tr/\0// / length $vec > 0.95) {

            use integer;

            my $i;

            # This method is faster with mostly null-bytes

            while($vec =~ /[^\0]/g ) {

                $i = -9 + 8 * pos $vec;

                push @ints, $i if vec($vec, ++$i, 1);

                push @ints, $i if vec($vec, ++$i, 1);

                push @ints, $i if vec($vec, ++$i, 1);

                push @ints, $i if vec($vec, ++$i, 1);

                push @ints, $i if vec($vec, ++$i, 1);

                push @ints, $i if vec($vec, ++$i, 1);

                push @ints, $i if vec($vec, ++$i, 1);

                push @ints, $i if vec($vec, ++$i, 1);

            }

        } else {

            # This method is a fast general algorithm

            use integer;

            my $bits = unpack "b*", $vec;

            push @ints, 0 if $bits =~ s/^(\d)// && $1;

            push @ints, pos $bits while($bits =~ /1/g);

        }

        return \@ints;

    }

This method gets faster the more sparse the bit vector is. (Courtesy of Tim Bunce and Winfried Koenig.) 



Why does defined() return true on empty arrays and hashes?

See defined in the 5.004 release or later of Perl. 



Data: Hashes (Associative Arrays)



How do I process an entire hash?

Use the each() function (see each) if you don't care whether it's sorted: 

    while ( ($key, $value) = each %hash) {

        print "$key = $value\n";

    }

If you want it sorted, you'll have to use foreach() on the result of sorting the keys as shown in an earlier question. 



What happens if I add or remove keys from a hash while iterating over it?

Don't do that. 



How do I look up a hash element by value?

Create a reverse hash: 

    %by_value = reverse %by_key;

    $key = $by_value{$value};

That's not particularly efficient. It would be more space-efficient to use: 

    while (($key, $value) = each %by_key) {

        $by_value{$value} = $key;

    }

If your hash could have repeated values, the methods above will only find one of the associated keys. This may or may not worry you. 



How can I know how many entries are in a hash?

If you mean how many keys, then all you have to do is take the scalar sense of the keys() function: 

    $num_keys = scalar keys %hash;

In void context it just resets the iterator, which is faster for tied hashes. 



How do I sort a hash (optionally by value instead of key)?

Internally, hashes are stored in a way that prevents you from imposing an order on key-value pairs. Instead, you have to sort a list of the keys or values: 

    @keys = sort keys %hash;    # sorted by key

    @keys = sort {

                    $hash{$a} cmp $hash{$b}

            } keys %hash;       # and by value

Here we'll do a reverse numeric sort by value, and if two keys are identical, sort by length of key, and if that fails, by straight ASCII comparison of the keys (well, possibly modified by your locale -- see the perllocale manpage). 

    @keys = sort {

                $hash{$b} <=> $hash{$a}

                          ||

                length($b) <=> length($a)

                          ||

                      $a cmp $b

    } keys %hash;



How can I always keep my hash sorted?

You can look into using the DB_File module and tie() using the $DB_BTREE hash bindings as documented in In Memory Databases. The Tie::IxHash module from CPAN might also be instructive. 



What's the difference between "delete" and "undef" with hashes?

Hashes are pairs of scalars: the first is the key, the second is the value. The key will be coerced to a string, although the value can be any kind of scalar: string, number, or reference. If a key $key is present in the array, exists($key) will return true. The value for a given key can be undef, in which case $array{$key} will be undef while $exists{$key} will return true. This corresponds to ($key, undef) being in the hash. 

Pictures help... here's the %ary table: 

          keys  values

        +------+------+

        |  a   |  3   |

        |  x   |  7   |

        |  d   |  0   |

        |  e   |  2   |

        +------+------+

And these conditions hold 

        $ary{'a'}                       is true

        $ary{'d'}                       is false

        defined $ary{'d'}               is true

        defined $ary{'a'}               is true

        exists $ary{'a'}                is true (perl5 only)

        grep ($_ eq 'a', keys %ary)     is true

If you now say 

        undef $ary{'a'}

your table now reads: 

          keys  values

        +------+------+

        |  a   | undef|

        |  x   |  7   |

        |  d   |  0   |

        |  e   |  2   |

        +------+------+

and these conditions now hold; changes in caps: 

        $ary{'a'}                       is FALSE

        $ary{'d'}                       is false

        defined $ary{'d'}               is true

        defined $ary{'a'}               is FALSE

        exists $ary{'a'}                is true (perl5 only)

        grep ($_ eq 'a', keys %ary)     is true

Notice the last two: you have an undef value, but a defined key! 

Now, consider this: 

        delete $ary{'a'}

your table now reads: 

          keys  values

        +------+------+

        |  x   |  7   |

        |  d   |  0   |

        |  e   |  2   |

        +------+------+

and these conditions now hold; changes in caps: 

        $ary{'a'}                       is false

        $ary{'d'}                       is false

        defined $ary{'d'}               is true

        defined $ary{'a'}               is false

        exists $ary{'a'}                is FALSE (perl5 only)

        grep ($_ eq 'a', keys %ary)     is FALSE

See, the whole entry is gone! 



Why don't my tied hashes make the defined/exists distinction?

They may or may not implement the EXISTS() and DEFINED() methods differently. For example, there isn't the concept of undef with hashes that are tied to DBM* files. This means the true/false tables above will give different results when used on such a hash. It also means that exists and defined do the same thing with a DBM* file, and what they end up doing is not what they do with ordinary hashes. 



How do I reset an each() operation part-way through?

Using keys %hash in scalar context returns the number of keys in the hash and resets the iterator associated with the hash. You may need to do this if you use last to exit a loop early so that when you re-enter it, the hash iterator has been reset. 



How can I get the unique keys from two hashes?

First you extract the keys from the hashes into arrays, and then solve the uniquifying the array problem described above. For example: 

    %seen = ();

    for $element (keys(%foo), keys(%bar)) {

        $seen{$element}++;

    }

    @uniq = keys %seen;

Or more succinctly: 

    @uniq = keys %{{%foo,%bar}};

Or if you really want to save space: 

    %seen = ();

    while (defined ($key = each %foo)) {

        $seen{$key}++;

    }

    while (defined ($key = each %bar)) {

        $seen{$key}++;

    }

    @uniq = keys %seen;



How can I store a multidimensional array in a DBM file?

Either stringify the structure yourself (no fun), or else get the MLDBM (which uses Data::Dumper) module from CPAN and layer it on top of either DB_File or GDBM_File. 



How can I make my hash remember the order I put elements into it?

Use the Tie::IxHash from CPAN. 

    use Tie::IxHash;

    tie(%myhash, Tie::IxHash);

    for ($i=0; $i<20; $i++) {

        $myhash{$i} = 2*$i;

    }

    @keys = keys %myhash;

    # @keys = (0,1,2,3,...)



Why does passing a subroutine an undefined element in a hash create it?

If you say something like: 

    somefunc($hash{"nonesuch key here"});

Then that element ``autovivifies''; that is, it springs into existence whether you store something there or not. That's because functions get scalars passed in by reference. If somefunc() modifies $_[0], it has to be ready to write it back into the caller's version. 

This has been fixed as of perl5.004. 

Normally, merely accessing a key's value for a nonexistent key does not cause that key to be forever there. This is different than awk's behavior. 



How can I make the Perl equivalent of a C structure/C++ class/hash or array of hashes or arrays?

Use references (documented in the perlref manpage). Examples of complex data structures are given in the perldsc manpage and the perllol manpage. Examples of structures and object-oriented classes are in the perltoot manpage. 



How can I use a reference as a hash key?

You can't do this directly, but you could use the standard Tie::Refhash module distributed with perl. 



Data: Misc



How do I handle binary data correctly?

Perl is binary clean, so this shouldn't be a problem. For example, this works fine (assuming the files are found): 

    if (`cat /vmunix` =~ /gzip/) {

        print "Your kernel is GNU-zip enabled!\n";

    }

On some systems, however, you have to play tedious games with ``text'' versus ``binary'' files. See binmode. 

If you're concerned about 8-bit ASCII data, then see the perllocale manpage. 

If you want to deal with multibyte characters, however, there are some gotchas. See the section on Regular Expressions. 



How do I determine whether a scalar is a number/whole/integer/float?

Assuming that you don't care about IEEE notations like ``NaN'' or ``Infinity'', you probably just want to use a regular expression. 

   warn "has nondigits"        if     /\D/;

    warn "not a natural number" unless /^\d+$/;             # rejects -3

    warn "not an integer"       unless /^-?\d+$/;           # rejects +3

   warn "not an integer"       unless /^[+-]?\d+$/;

   warn "not a decimal number" unless /^-?\d+\.?\d*$/;  # rejects .2

   warn "not a decimal number" unless /^-?(?:\d+(?:\.\d*)?|\.\d+)$/;

   warn "not a C float"

       unless /^([+-]?)(?=\d|\.\d)\d*(\.\d*)?([Ee]([+-]?\d+))?$/;

If you're on a POSIX system, Perl's supports the POSIX::strtod function. Its semantics are somewhat cumbersome, so here's a getnum wrapper function for more convenient access. This function takes a string and returns the number it found, or undef for input that isn't a C float. The is_numeric function is a front end to getnum if you just want to say, ``Is this a float?'' 

    sub getnum {

        use POSIX qw(strtod);

        my $str = shift;

        $str =~ s/^\s+//;

        $str =~ s/\s+$//;

        $! = 0;

        my($num, $unparsed) = strtod($str);

        if (($str eq '') || ($unparsed != 0) || $!) {

            return undef;

        } else {

            return $num;

        } 

    } 

    sub is_numeric { defined &getnum } 

Or you could check out http://www.perl.com/CPAN/modules/by-module/String/String-Scanf-1.1.tar.gz instead. The POSIX module (part of the standard Perl distribution) provides the strtol and strtod for converting strings to double and longs, respectively. 



How do I keep persistent data across program calls?

For some specific applications, you can use one of the DBM modules. See the AnyDBM_File manpage. More generically, you should consult the FreezeThaw, Storable, or Class::Eroot modules from CPAN. 



How do I print out or copy a recursive data structure?

The Data::Dumper module on CPAN is nice for printing out data structures, and FreezeThaw for copying them. For example: 

    use FreezeThaw qw(freeze thaw);

    $new = thaw freeze $old;

Where $old can be (a reference to) any kind of data structure you'd like. It will be deeply copied. 



How do I define methods for every class/object?

Use the UNIVERSAL class (see the UNIVERSAL manpage). 



How do I verify a credit card checksum?

Get the Business::CreditCard module from CPAN. 

perlfunc - Perl builtin functions 



DESCRIPTION

The functions in this section can serve as terms in an expression. They fall into two major categories: list operators and named unary operators. These differ in their precedence relationship with a following comma. (See the precedence table in the perlop manpage.) List operators take more than one argument, while unary operators can never take more than one argument. Thus, a comma terminates the argument of a unary operator, but merely separates the arguments of a list operator. A unary operator generally provides a scalar context to its argument, while a list operator may provide either scalar and list contexts for its arguments. If it does both, the scalar arguments will be first, and the list argument will follow. (Note that there can ever be only one list argument.) For instance, splice() has three scalar arguments followed by a list. 

In the syntax descriptions that follow, list operators that expect a list (and provide list context for the elements of the list) are shown with LIST as an argument. Such a list may consist of any combination of scalar arguments or list values; the list values will be included in the list as if each individual element were interpolated at that point in the list, forming a longer single-dimensional list value. Elements of the LIST should be separated by commas. 

Any function in the list below may be used either with or without parentheses around its arguments. (The syntax descriptions omit the parentheses.) If you use the parentheses, the simple (but occasionally surprising) rule is this: It LOOKS like a function, therefore it IS a function, and precedence doesn't matter. Otherwise it's a list operator or unary operator, and precedence does matter. And whitespace between the function and left parenthesis doesn't count--so you need to be careful sometimes: 

    print 1+2+4;        # Prints 7.

    print(1+2) + 4;     # Prints 3.

    print (1+2)+4;      # Also prints 3!

    print +(1+2)+4;     # Prints 7.

    print ((1+2)+4);    # Prints 7.

If you run Perl with the -w switch it can warn you about this. For example, the third line above produces: 

    print (...) interpreted as function at - line 1.

    Useless use of integer addition in void context at - line 1.

For functions that can be used in either a scalar or list context, nonabortive failure is generally indicated in a scalar context by returning the undefined value, and in a list context by returning the null list. 

Remember the following important rule: There is no rule that relates the behavior of an expression in list context to its behavior in scalar context, or vice versa. It might do two totally different things. Each operator and function decides which sort of value it would be most appropriate to return in a scalar context. Some operators return the length of the list that would have been returned in list context. Some operators return the first value in the list. Some operators return the last value in the list. Some operators return a count of successful operations. In general, they do what you want, unless you want consistency. 

An named array in scalar context is quite different from what would at first glance appear to be a list in scalar context. You can't get a list like (1,2,3) into being in scalar context, because the compiler knows the context at compile time. It would generate the scalar comma operator there, not the list construction version of the comma. That means it was never a list to start with. 

In general, functions in Perl that serve as wrappers for system calls of the same name (like chown(2), fork(2), closedir(2), etc.) all return true when they succeed and undef otherwise, as is usually mentioned in the descriptions below. This is different from the C interfaces, which return -1 on failure. Exceptions to this rule are wait(), waitpid(), and syscall(). System calls also set the special $! variable on failure. Other functions do not, except accidentally. 



Perl Functions by Category

Here are Perl's functions (including things that look like functions, like some keywords and named operators) arranged by category. Some functions appear in more than one place. 

Functions for SCALARs or strings 

chomp, chop, chr, crypt, hex, index, lc, lcfirst, length, oct, ord, pack, q/STRING/, qq/STRING/, reverse, rindex, sprintf, substr, tr///, uc, ucfirst, y/// 

Regular expressions and pattern matching 

m//, pos, quotemeta, s///, split, study, qr// 

Numeric functions 

abs, atan2, cos, exp, hex, int, log, oct, rand, sin, sqrt, srand 

Functions for real @ARRAYs 

pop, push, shift, splice, unshift 

Functions for list data 

grep, join, map, qw/STRING/, reverse, sort, unpack 

Functions for real %HASHes 

delete, each, exists, keys, values 

Input and output functions 

binmode, close, closedir, dbmclose, dbmopen, die, eof, fileno, flock, format, getc, print, printf, read, readdir, rewinddir, seek, seekdir, select, syscall, sysread, sysseek, syswrite, tell, telldir, truncate, warn, write 

Functions for fixed length data or records 

pack, read, syscall, sysread, syswrite, unpack, vec 

Functions for filehandles, files, or directories 

-X, chdir, chmod, chown, chroot, fcntl, glob, ioctl, link, lstat, mkdir, open, opendir, readlink, rename, rmdir, stat, symlink, umask, unlink, utime 

Keywords related to the control flow of your perl program 

caller, continue, die, do, dump, eval, exit, goto, last, next, redo, return, sub, wantarray 

Keywords related to scoping 

caller, import, local, my, package, use 

Miscellaneous functions 

defined, dump, eval, formline, local, my, reset, scalar, undef, wantarray 

Functions for processes and process groups 

alarm, exec, fork, getpgrp, getppid, getpriority, kill, pipe, qx/STRING/, setpgrp, setpriority, sleep, system, times, wait, waitpid 

Keywords related to perl modules 

do, import, no, package, require, use 

Keywords related to classes and object-orientedness 

bless, dbmclose, dbmopen, package, ref, tie, tied, untie, use 

Low-level socket functions 

accept, bind, connect, getpeername, getsockname, getsockopt, listen, recv, send, setsockopt, shutdown, socket, socketpair 

System V interprocess communication functions 

msgctl, msgget, msgrcv, msgsnd, semctl, semget, semop, shmctl, shmget, shmread, shmwrite 

Fetching user and group info 

endgrent, endhostent, endnetent, endpwent, getgrent, getgrgid, getgrnam, getlogin, getpwent, getpwnam, getpwuid, setgrent, setpwent 

Fetching network info 

endprotoent, endservent, gethostbyaddr, gethostbyname, gethostent, getnetbyaddr, getnetbyname, getnetent, getprotobyname, getprotobynumber, getprotoent, getservbyname, getservbyport, getservent, sethostent, setnetent, setprotoent, setservent 

Time-related functions 

gmtime, localtime, time, times 

Functions new in perl5 

abs, bless, chomp, chr, exists, formline, glob, import, lc, lcfirst, map, my, no, prototype, qx, qw, readline, readpipe, ref, sub*, sysopen, tie, tied, uc, ucfirst, untie, use 

* sub was a keyword in perl4, but in perl5 it is an operator, which can be used in expressions.
Functions obsoleted in perl5 dbmclose, dbmopen 



Alphabetical Listing of Perl Functions

-X 

run a file test 

abs 

absolute value function 

accept 

accept an incoming socket connect 

alarm 

schedule a SIGALRM 

atan2 

arctangent of Y/X 

bind 

binds an address to a socket 

binmode 

prepare binary files on old systems 

bless 

create an object 

caller 

get context of the current subroutine call 

chdir 

change your current working directory 

chmod 

changes the permissions on a list of files 

chomp 

remove a trailing record separator from a string 

chop 

remove the last character from a string 

chown 

change the owership on a list of files 

chr 

get character this number represents 

chroot 

make directory new root for path lookups 

close 

close file (or pipe or socket) handle 

closedir 

close directory handle 

connect 

connect to a remove socket 

continue 

optional trailing block in a while or foreach 

cos 

cosine function 

crypt 

one-way passwd-style encryption 

dbmclose 

breaks binding on a tied dbm file 

dbmopen 

create binding on a tied dbm file 

defined 

test whether a value, variable, or function is defined 

delete 

deletes a value from a hash 

die 

raise an exception or bail out 

do 

turn a BLOCK into a TERM 

dump 

create an immediate core dump 

each 

retrieve the next key/value pair from a hash 

endgrent 

be done using group file 

endhostent 

be done using hosts file 

endnetent 

be done using networks file 

endprotoent 

be done using protocols file 

endpwent 

be done using passwd file 

endservent 

be done using services file 

eof 

test a filehandle for its end 

eval 

catch exceptions or compile code 

exec 

abandon this program to run another 

exists 

test whether a hash key is present 

exit 

terminate this program 

exp 

raise e to a power 

fcntl 

file control system all 

fileno 

return file descriptor from filehandle 

flock 

lock an entire file with an advisory lock 

fork 

create a new process just like this one 

format 

declare a picture format with use by the write() function 

formline 

internal function used for formats 

getc 

get the next character from the filehandle 

getgrent 

get next group record 

getgrgid 

get group record given group user ID 

getgrnam 

get group record given group name 

gethostbyaddr 

get host record given its address 

gethostbyname 

get host record given name 

gethostent 

get next hosts record 

getlogin 

return who logged in at this tty 

getnetbyaddr 

get network record given its address 

getnetbyname 

get networks record given name 

getnetent 

get next networks record 

getpeername 

find the other hend of a socket connection 

getpgrp 

get process group 

getppid 

get parent process ID 

getpriority 

get current nice value 

getprotobyname 

get protocol record given name 

getprotobynumber 

get protocol record numeric protocol 

getprotoent 

get next protocols record 

getpwent 

get next passwd record 

getpwnam 

get passwd record given user login name 

getpwuid 

get passwd record given user ID 

getservbyname 

get services record given its name 

getservbyport 

get services record given numeric port 

getservent 

get next services record 

getsockname 

retrieve the sockaddr for a given socket 

getsockopt 

get socket options on a given socket 

glob 

expand filenames using wildcards 

gmtime 

convert UNIX time into record or string using Greenwich time 

goto 

create spaghetti code 

grep 

locate elements in a list test true against a given criterion 

hex 

convert a string to a hexadecimal number 

import 

patch a module's namespace into your own 

int 

get the integer portion of a number 

ioctl 

system-dependent device control system call 

join 

join a list into a string using a separator 

keys 

retrieve list of indices from a hash 

kill 

send a signal to a process or process group 

last 

exit a block prematurely 

lc 

return lower-case version of a string 

lcfirst 

return a string with just the next letter in lower case 

length 

return the number of bytes in a string 

link 

create a hard link in the filesytem 

listen 

register your socket as a server 

local 

create a temporary value for a global variable (dynamic scoping) 

localtime 

convert UNIX time into record or string using local time 

log 

retrieve the natural logarithm for a number 

lstat 

stat a symbolic link 

m// 

match a string with a regular expression pattern 

map 

apply a change to a list to get back a new list with the changes 

mkdir 

create a directory 

msgctl 

SysV IPC message control operations 

msgget 

get SysV IPC message queue 

msgrcv 

receive a SysV IPC message from a message queue 

msgsnd 

send a SysV IPC message to a message queue 

my 

declare and assign a local variable (lexical scoping) 

next 

iterate a block prematurely 

no 

unimport some module symbols or semantics at compile time 

oct 

convert a string to an octal number 

open 

open a file, pipe, or descriptor 

opendir 

open a directory 

ord 

find a character's numeric representation 

pack 

convert a list into a binary representation 

package 

declare a separate global namespace 

pipe 

open a pair of connected filehandles 

pop 

remove the last element from an array and return it 

pos 

find or set the offset for the last/next m//g search 

print 

output a list to a filehandle 

printf 

output a formatted list to a filehandle 

prototype 

get the prototype (if any) of a subroutine 

push 

append one or more elements to an array 

q/STRING/ 

singly quote a string 

qq/STRING/ 

doubly quote a string 

quotemeta 

quote regular expression magic characters 

qw/STRING/ 

quote a list of words 

qx/STRING/ 

backquote quote a string 

rand 

retrieve the next pseudorandom number 

read 

fixed-length buffered input from a filehandle 

readdir 

get a directory from a directory handle 

readlink 

determine where a symbolic link is pointing 

recv 

receive a message over a Socket 

redo 

start this loop iteration over again 

ref 

find out the type of thing being referenced 

rename 

change a filename 

require 

load in external functions from a library at runtime 

reset 

clear all variables of a given name 

return 

get out of a function early 

reverse 

flip a string or a list 

rewinddir 

reset directory handle 

rindex 

right-to-left substring search 

rmdir 

remove a directory 

s/// 

replace a pattern with a string 

scalar 

force a scalar context 

seek 

reposition file pointer for random-access I/O 

seekdir 

reposition directory pointer 

select 

reset default output or do I/O multiplexing 

semctl 

SysV semaphore control operations 

semget 

get set of SysV semaphores 

semop 

SysV semaphore operations 

send 

send a message over a socket 

setgrent 

prepare group file for use 

sethostent 

prepare hosts file for use 

setnetent 

prepare networks file for use 

setpgrp 

set the process group of a process 

setpriority 

set a process's nice value 

setprotoent 

prepare protocols file for use 

setpwent 

prepare passwd file for use 

setservent 

prepare services file for use 

setsockopt 

set some socket options 

shift 

remove the first element of an array, and return it 

shmctl 

SysV shared memory operations 

shmget 

get SysV shared memory segment identifier 

shmread 

read SysV shared memory 

shmwrite 

write SysV shared memory 

shutdown 

close down just half of a socket connection 

sin 

return the sin of a number 

sleep 

block for some number of seconds 

socket 

create a socket 

socketpair 

create a pair of sockets 

sort 

sort a list of values 

splice 

add or remove elements anywhere in an array 

split 

split up a string using a regexp delimiter 

sprintf 

formatted print into a string 

sqrt 

square root function 

srand 

seed the random number generator 

stat 

get a file's status information 

study 

optimize input data for repeated searches 

sub 

declare a subroutine, possibly anonymously 

substr 

get or alter a portion of a stirng 

symlink 

create a symbolic link to a file 

syscall 

execute an arbitrary system call 

sysread 

fixed-length unbuffered input from a filehandle 

system 

run a separate program 

syswrite 

fixed-length unbuffered output to a filehandle 

tell 

get current seekpointer on a filehandle 

telldir 

get current seekpointer on a directory handle 

tie 

bind a variable to an object class 

time 

return number of seconds since 1970 

times 

return elapsed time for self and child processes 

tr/// 

transliterate a string 

truncate 

shorten a file 

uc 

return upper-case version of a string 

ucfirst 

return a string with just the next letter in upper case 

umask 

set file creation mode mask 

undef 

remove a variable or function definition 

unlink 

remove one link to a file 

unpack 

convert binary structure into normal perl variables 

unshift 

prepend more elements to the beginning of a list 

untie 

break a tie binding to a variable 

use 

load in a module at compile time 

utime 

set a file's last access and modify times 

values 

return a list of the values in a hash 

vec 

test or set particular bits in a string 

wait 

wait for any child process to die 

waitpid 

wait for a particular child process to die 

wantarray 

get list vs array context of current subroutine call 

warn 

print debugging info 

write 

print a picture record 

y/// 

transliterate a string 

NAME

perlLoL - Manipulating Lists of Lists in Perl 



DESCRIPTION



Declaration and Access of Lists of Lists

The simplest thing to build is a list of lists (sometimes called an array of arrays). It's reasonably easy to understand, and almost everything that applies here will also be applicable later on with the fancier data structures. 

A list of lists, or an array of an array if you would, is just a regular old array @LoL that you can get at with two subscripts, like $LoL[3][2]. Here's a declaration of the array: 

    # assign to our array a list of list references

    @LoL = (

           [ "fred", "barney" ],

           [ "george", "jane", "elroy" ],

           [ "homer", "marge", "bart" ],

    );

    print $LoL[2][2];

  bart

Now you should be very careful that the outer bracket type is a round one, that is, a parenthesis. That's because you're assigning to an @list, so you need parentheses. If you wanted there not to be an @LoL, but rather just a reference to it, you could do something more like this: 

    # assign a reference to list of list references

    $ref_to_LoL = [

        [ "fred", "barney", "pebbles", "bambam", "dino", ],

        [ "homer", "bart", "marge", "maggie", ],

        [ "george", "jane", "alroy", "judy", ],

    ];

    print $ref_to_LoL->[2][2];

Notice that the outer bracket type has changed, and so our access syntax has also changed. That's because unlike C, in perl you can't freely interchange arrays and references thereto. $ref_to_LoL is a reference to an array, whereas @LoL is an array proper. Likewise, $LoL[2] is not an array, but an array ref. So how come you can write these: 

    $LoL[2][2]

    $ref_to_LoL->[2][2]

instead of having to write these: 

    $LoL[2]->[2]

    $ref_to_LoL->[2]->[2]

Well, that's because the rule is that on adjacent brackets only (whether square or curly), you are free to omit the pointer dereferencing arrow. But you cannot do so for the very first one if it's a scalar containing a reference, which means that $ref_to_LoL always needs it. 



Growing Your Own

That's all well and good for declaration of a fixed data structure, but what if you wanted to add new elements on the fly, or build it up entirely from scratch? 

First, let's look at reading it in from a file. This is something like adding a row at a time. We'll assume that there's a flat file in which each line is a row and each word an element. If you're trying to develop an @LoL list containing all these, here's the right way to do that: 

    while (<>) {

        @tmp = split;

        push @LoL, [ @tmp ];

    }

You might also have loaded that from a function: 

    for $i ( 1 .. 10 ) {

        $LoL[$i] = [ somefunc($i) ];

    }

Or you might have had a temporary variable sitting around with the list in it. 

    for $i ( 1 .. 10 ) {

        @tmp = somefunc($i);

        $LoL[$i] = [ @tmp ];

    }

It's very important that you make sure to use the [] list reference constructor. That's because this will be very wrong: 

    $LoL[$i] = @tmp;

You see, assigning a named list like that to a scalar just counts the number of elements in @tmp, which probably isn't what you want. 

If you are running under use strict, you'll have to add some declarations to make it happy: 

    use strict;

    my(@LoL, @tmp);

    while (<>) {

        @tmp = split;

        push @LoL, [ @tmp ];

    }

Of course, you don't need the temporary array to have a name at all: 

    while (<>) {

        push @LoL, [ split ];

    }

You also don't have to use push(). You could just make a direct assignment if you knew where you wanted to put it: 

    my (@LoL, $i, $line);

    for $i ( 0 .. 10 ) {

        $line = <>;

        $LoL[$i] = [ split ' ', $line ];

    }

or even just 

    my (@LoL, $i);

    for $i ( 0 .. 10 ) {

        $LoL[$i] = [ split ' ', <> ];

    }

You should in general be leery of using potential list functions in a scalar context without explicitly stating such. This would be clearer to the casual reader: 

    my (@LoL, $i);

    for $i ( 0 .. 10 ) {

        $LoL[$i] = [ split ' ', scalar(<>) ];

    }

If you wanted to have a $ref_to_LoL variable as a reference to an array, you'd have to do something like this: 

    while (<>) {

        push @$ref_to_LoL, [ split ];

    }

Now you can add new rows. What about adding new columns? If you're dealing with just matrices, it's often easiest to use simple assignment: 

    for $x (1 .. 10) {

        for $y (1 .. 10) {

            $LoL[$x][$y] = func($x, $y);

        }

    }

    for $x ( 3, 7, 9 ) {

        $LoL[$x][20] += func2($x);

    }

It doesn't matter whether those elements are already there or not: it'll gladly create them for you, setting intervening elements to undef as need be. 

If you wanted just to append to a row, you'd have to do something a bit funnier looking: 

    # add new columns to an existing row

    push @{ $LoL[0] }, "wilma", "betty";

Notice that I couldn't say just: 

    push $LoL[0], "wilma", "betty";  # WRONG!

In fact, that wouldn't even compile. How come? Because the argument to push() must be a real array, not just a reference to such. 



Access and Printing

Now it's time to print your data structure out. How are you going to do that? Well, if you want only one of the elements, it's trivial: 

    print $LoL[0][0];

If you want to print the whole thing, though, you can't say 

    print @LoL;         # WRONG

because you'll get just references listed, and perl will never automatically dereference things for you. Instead, you have to roll yourself a loop or two. This prints the whole structure, using the shell-style for() construct to loop across the outer set of subscripts. 

    for $aref ( @LoL ) {

        print "\t [ @$aref ],\n";

    }

If you wanted to keep track of subscripts, you might do this: 

    for $i ( 0 .. $#LoL ) {

        print "\t elt $i is [ @{$LoL[$i]} ],\n";

    }

or maybe even this. Notice the inner loop. 

    for $i ( 0 .. $#LoL ) {

        for $j ( 0 .. $#{$LoL[$i]} ) {

            print "elt $i $j is $LoL[$i][$j]\n";

        }

    }

As you can see, it's getting a bit complicated. That's why sometimes is easier to take a temporary on your way through: 

    for $i ( 0 .. $#LoL ) {

        $aref = $LoL[$i];

        for $j ( 0 .. $#{$aref} ) {

            print "elt $i $j is $LoL[$i][$j]\n";

        }

    }

Hmm... that's still a bit ugly. How about this: 

    for $i ( 0 .. $#LoL ) {

        $aref = $LoL[$i];

        $n = @$aref - 1;

        for $j ( 0 .. $n ) {

            print "elt $i $j is $LoL[$i][$j]\n";

        }

    }



Slices

If you want to get at a slice (part of a row) in a multidimensional array, you're going to have to do some fancy subscripting. That's because while we have a nice synonym for single elements via the pointer arrow for dereferencing, no such convenience exists for slices. (Remember, of course, that you can always write a loop to do a slice operation.) 

Here's how to do one operation using a loop. We'll assume an @LoL variable as before. 

    @part = ();

    $x = 4;

    for ($y = 7; $y < 13; $y++) {

        push @part, $LoL[$x][$y];

    }

That same loop could be replaced with a slice operation: 

    @part = @{ $LoL[4] } [ 7..12 ];

but as you might well imagine, this is pretty rough on the reader. 

Ah, but what if you wanted a two-dimensional slice, such as having $x run from 4..8 and $y run from 7 to 12? Hmm... here's the simple way: 

    @newLoL = ();

    for ($startx = $x = 4; $x <= 8; $x++) {

        for ($starty = $y = 7; $y <= 12; $y++) {

            $newLoL[$x - $startx][$y - $starty] = $LoL[$x][$y];

        }

    }

We can reduce some of the looping through slices 

    for ($x = 4; $x <= 8; $x++) {

        push @newLoL, [ @{ $LoL[$x] } [ 7..12 ] ];

    }

If you were into Schwartzian Transforms, you would probably have selected map for that 

    @newLoL = map { [ @{ $LoL[$_] } [ 7..12 ] ] } 4 .. 8;

Although if your manager accused of seeking job security (or rapid insecurity) through inscrutable code, it would be hard to argue. :-) If I were you, I'd put that in a function: 

    @newLoL = splice_2D( \@LoL, 4 => 8, 7 => 12 );

    sub splice_2D {

        my $lrr = shift;        # ref to list of list refs!

        my ($x_lo, $x_hi,

            $y_lo, $y_hi) = @_;

        return map {

            [ @{ $lrr->[$_] } [ $y_lo .. $y_hi ] ]

        } $x_lo .. $x_hi;

    }

DESCRIPTION

The single feature most sorely lacking in the Perl programming language prior to its 5.0 release was complex data structures. Even without direct language support, some valiant programmers did manage to emulate them, but it was hard work and not for the faint of heart. You could occasionally get away with the $m{$LoL,$b} notation borrowed from awk in which the keys are actually more like a single concatenated string "$LoL$b", but traversal and sorting were difficult. More desperate programmers even hacked Perl's internal symbol table directly, a strategy that proved hard to develop and maintain--to put it mildly. 

The 5.0 release of Perl let us have complex data structures. You may now write something like this and all of a sudden, you'd have a array with three dimensions! 

    for $x (1 .. 10) {

        for $y (1 .. 10) {

            for $z (1 .. 10) {

                $LoL[$x][$y][$z] =

                    $x ** $y + $z;

            }

        }

    }

Alas, however simple this may appear, underneath it's a much more elaborate construct than meets the eye! 

How do you print it out? Why can't you say just print @LoL? How do you sort it? How can you pass it to a function or get one of these back from a function? Is is an object? Can you save it to disk to read back later? How do you access whole rows or columns of that matrix? Do all the values have to be numeric? 

As you see, it's quite easy to become confused. While some small portion of the blame for this can be attributed to the reference-based implementation, it's really more due to a lack of existing documentation with examples designed for the beginner. 

This document is meant to be a detailed but understandable treatment of the many different sorts of data structures you might want to develop. It should also serve as a cookbook of examples. That way, when you need to create one of these complex data structures, you can just pinch, pilfer, or purloin a drop-in example from here. 

Let's look at each of these possible constructs in detail. There are separate sections on each of the following: 

· arrays of arrays 

· hashes of arrays 

· arrays of hashes 

· hashes of hashes 

· more elaborate constructs 

But for now, let's look at general issues common to all these types of data structures. 



REFERENCES

The most important thing to understand about all data structures in Perl -- including multidimensional arrays--is that even though they might appear otherwise, Perl @ARRAYs and %HASHes are all internally one-dimensional. They can hold only scalar values (meaning a string, number, or a reference). They cannot directly contain other arrays or hashes, but instead contain references to other arrays or hashes. 

You can't use a reference to a array or hash in quite the same way that you would a real array or hash. For C or C++ programmers unused to distinguishing between arrays and pointers to the same, this can be confusing. If so, just think of it as the difference between a structure and a pointer to a structure. 

You can (and should) read more about references in the perlref(1) man page. Briefly, references are rather like pointers that know what they point to. (Objects are also a kind of reference, but we won't be needing them right away--if ever.) This means that when you have something which looks to you like an access to a two-or-more-dimensional array and/or hash, what's really going on is that the base type is merely a one-dimensional entity that contains references to the next level. It's just that you can use it as though it were a two-dimensional one. This is actually the way almost all C multidimensional arrays work as well. 

    $list[7][12]                        # array of arrays

    $list[7]{string}                    # array of hashes

    $hash{string}[7]                    # hash of arrays

    $hash{string}{'another string'}     # hash of hashes

Now, because the top level contains only references, if you try to print out your array in with a simple print() function, you'll get something that doesn't look very nice, like this: 

    @LoL = ( [2, 3], [4, 5, 7], [0] );

    print $LoL[1][2];

  7

    print @LoL;

  ARRAY(0x83c38)ARRAY(0x8b194)ARRAY(0x8b1d0)

That's because Perl doesn't (ever) implicitly dereference your variables. If you want to get at the thing a reference is referring to, then you have to do this yourself using either prefix typing indicators, like ${$blah}, @{$blah}, @{$blah[$i]}, or else postfix pointer arrows, like $a->[3], $h->{fred}, or even $ob->method()->[3]. 



COMMON MISTAKES

The two most common mistakes made in constructing something like an array of arrays is either accidentally counting the number of elements or else taking a reference to the same memory location repeatedly. Here's the case where you just get the count instead of a nested array: 

    for $i (1..10) {

        @list = somefunc($i);

        $LoL[$i] = @list;       # WRONG!

    }

That's just the simple case of assigning a list to a scalar and getting its element count. If that's what you really and truly want, then you might do well to consider being a tad more explicit about it, like this: 

    for $i (1..10) {

        @list = somefunc($i);

        $counts[$i] = scalar @list;

    }

Here's the case of taking a reference to the same memory location again and again: 

    for $i (1..10) {

        @list = somefunc($i);

        $LoL[$i] = \@list;      # WRONG!

    }

So, what's the big problem with that? It looks right, doesn't it? After all, I just told you that you need an array of references, so by golly, you've made me one! 

Unfortunately, while this is true, it's still broken. All the references in @LoL refer to the very same place, and they will therefore all hold whatever was last in @list! It's similar to the problem demonstrated in the following C program: 

    #include <pwd.h>

    main() {

        struct passwd *getpwnam(), *rp, *dp;

        rp = getpwnam("root");

        dp = getpwnam("daemon");

        printf("daemon name is %s\nroot name is %s\n",

                dp->pw_name, rp->pw_name);

    }

Which will print 

    daemon name is daemon

    root name is daemon

The problem is that both rp and dp are pointers to the same location in memory! In C, you'd have to remember to malloc() yourself some new memory. In Perl, you'll want to use the array constructor [] or the hash constructor {} instead. Here's the right way to do the preceding broken code fragments: 

    for $i (1..10) {

        @list = somefunc($i);

        $LoL[$i] = [ @list ];

    }

The square brackets make a reference to a new array with a copy of what's in @list at the time of the assignment. This is what you want. 

Note that this will produce something similar, but it's much harder to read: 

    for $i (1..10) {

        @list = 0 .. $i;

        @{$LoL[$i]} = @list;

    }

Is it the same? Well, maybe so--and maybe not. The subtle difference is that when you assign something in square brackets, you know for sure it's always a brand new reference with a new copy of the data. Something else could be going on in this new case with the @{$LoL[$i]}} dereference on the left-hand-side of the assignment. It all depends on whether $LoL[$i] had been undefined to start with, or whether it already contained a reference. If you had already populated @LoL with references, as in 

    $LoL[3] = \@another_list;

Then the assignment with the indirection on the left-hand-side would use the existing reference that was already there: 

    @{$LoL[3]} = @list;

Of course, this would have the ``interesting'' effect of clobbering @another_list. (Have you ever noticed how when a programmer says something is ``interesting'', that rather than meaning ``intriguing'', they're disturbingly more apt to mean that it's ``annoying'', ``difficult'', or both? :-) 

So just remember always to use the array or hash constructors with [] or {}, and you'll be fine, although it's not always optimally efficient. 

Surprisingly, the following dangerous-looking construct will actually work out fine: 

    for $i (1..10) {

        my @list = somefunc($i);

        $LoL[$i] = \@list;

    }

That's because my() is more of a run-time statement than it is a compile-time declaration per se. This means that the my() variable is remade afresh each time through the loop. So even though it looks as though you stored the same variable reference each time, you actually did not! This is a subtle distinction that can produce more efficient code at the risk of misleading all but the most experienced of programmers. So I usually advise against teaching it to beginners. In fact, except for passing arguments to functions, I seldom like to see the gimme-a-reference operator (backslash) used much at all in code. Instead, I advise beginners that they (and most of the rest of us) should try to use the much more easily understood constructors [] and {} instead of relying upon lexical (or dynamic) scoping and hidden reference-counting to do the right thing behind the scenes. 

In summary: 

    $LoL[$i] = [ @list ];       # usually best

    $LoL[$i] = \@list;          # perilous; just how my() was that list?

    @{ $LoL[$i] } = @list;      # way too tricky for most programmers



CAVEAT ON PRECEDENCE

Speaking of things like @{$LoL[$i]}, the following are actually the same thing: 

    $listref->[2][2]    # clear

    $$listref[2][2]     # confusing

That's because Perl's precedence rules on its five prefix dereferencers (which look like someone swearing: $ @ * % &) make them bind more tightly than the postfix subscripting brackets or braces! This will no doubt come as a great shock to the C or C++ programmer, who is quite accustomed to using *a[i] to mean what's pointed to by the i'th element of a. That is, they first take the subscript, and only then dereference the thing at that subscript. That's fine in C, but this isn't C. 

The seemingly equivalent construct in Perl, $$listref[$i] first does the deref of $listref, making it take $listref as a reference to an array, and then dereference that, and finally tell you the i'th value of the array pointed to by $LoL. If you wanted the C notion, you'd have to write ${$LoL[$i]} to force the $LoL[$i] to get evaluated first before the leading $ dereferencer. 



WHY YOU SHOULD ALWAYS use strict
If this is starting to sound scarier than it's worth, relax. Perl has some features to help you avoid its most common pitfalls. The best way to avoid getting confused is to start every program like this: 

    #!/usr/bin/perl -w

    use strict;

This way, you'll be forced to declare all your variables with my() and also disallow accidental ``symbolic dereferencing''. Therefore if you'd done this: 

    my $listref = [

        [ "fred", "barney", "pebbles", "bambam", "dino", ],

        [ "homer", "bart", "marge", "maggie", ],

        [ "george", "jane", "elroy", "judy", ],

    ];

    print $listref[2][2];

The compiler would immediately flag that as an error at compile time, because you were accidentally accessing @listref, an undeclared variable, and it would thereby remind you to write instead: 

    print $listref->[2][2]



DEBUGGING

Before version 5.002, the standard Perl debugger didn't do a very nice job of printing out complex data structures. With 5.002 or above, the debugger includes several new features, including command line editing as well as the x command to dump out complex data structures. For example, given the assignment to $LoL above, here's the debugger output: 

    DB<1> x $LoL

    $LoL = ARRAY(0x13b5a0)

       0  ARRAY(0x1f0a24)

          0  'fred'

          1  'barney'

          2  'pebbles'

          3  'bambam'

          4  'dino'

       1  ARRAY(0x13b558)

          0  'homer'

          1  'bart'

          2  'marge'

          3  'maggie'

       2  ARRAY(0x13b540)

          0  'george'

          1  'jane'

          2  'elroy'

          3  'judy'



CODE EXAMPLES

Presented with little comment (these will get their own manpages someday) here are short code examples illustrating access of various types of data structures. 



LISTS OF LISTS



Declaration of a LIST OF LISTS

 @LoL = (

        [ "fred", "barney" ],

        [ "george", "jane", "elroy" ],

        [ "homer", "marge", "bart" ],

      );



Generation of a LIST OF LISTS

 # reading from file

 while ( <> ) {

     push @LoL, [ split ];

 }

 # calling a function

 for $i ( 1 .. 10 ) {

     $LoL[$i] = [ somefunc($i) ];

 }

 # using temp vars

 for $i ( 1 .. 10 ) {

     @tmp = somefunc($i);

     $LoL[$i] = [ @tmp ];

 }

 # add to an existing row

 push @{ $LoL[0] }, "wilma", "betty";



Access and Printing of a LIST OF LISTS

 # one element

 $LoL[0][0] = "Fred";

 # another element

 $LoL[1][1] =~ s/(\w)/\u$1/;

 # print the whole thing with refs

 for $aref ( @LoL ) {

     print "\t [ @$aref ],\n";

 }

 # print the whole thing with indices

 for $i ( 0 .. $#LoL ) {

     print "\t [ @{$LoL[$i]} ],\n";

 }

 # print the whole thing one at a time

 for $i ( 0 .. $#LoL ) {

     for $j ( 0 .. $#{ $LoL[$i] } ) {

         print "elt $i $j is $LoL[$i][$j]\n";

     }

 }



HASHES OF LISTS



Declaration of a HASH OF LISTS

 %HoL = (

        flintstones        => [ "fred", "barney" ],

        jetsons            => [ "george", "jane", "elroy" ],

        simpsons           => [ "homer", "marge", "bart" ],

      );



Generation of a HASH OF LISTS

 # reading from file

 # flintstones: fred barney wilma dino

 while ( <> ) {

     next unless s/^(.*?):\s*//;

     $HoL{$1} = [ split ];

 }

 # reading from file; more temps

 # flintstones: fred barney wilma dino

 while ( $line = <> ) {

     ($who, $rest) = split /:\s*/, $line, 2;

     @fields = split ' ', $rest;

     $HoL{$who} = [ @fields ];

 }

 # calling a function that returns a list

 for $group ( "simpsons", "jetsons", "flintstones" ) {

     $HoL{$group} = [ get_family($group) ];

 }

 # likewise, but using temps

 for $group ( "simpsons", "jetsons", "flintstones" ) {

     @members = get_family($group);

     $HoL{$group} = [ @members ];

 }

 # append new members to an existing family

 push @{ $HoL{"flintstones"} }, "wilma", "betty";



Access and Printing of a HASH OF LISTS

 # one element

 $HoL{flintstones}[0] = "Fred";

 # another element

 $HoL{simpsons}[1] =~ s/(\w)/\u$1/;

 # print the whole thing

 foreach $family ( keys %HoL ) {

     print "$family: @{ $HoL{$family} }\n"

 }

 # print the whole thing with indices

 foreach $family ( keys %HoL ) {

     print "family: ";

     foreach $i ( 0 .. $#{ $HoL{$family} } ) {

         print " $i = $HoL{$family}[$i]";

     }

     print "\n";

 }

 # print the whole thing sorted by number of members

 foreach $family ( sort { @{$HoL{$b}} <=> @{$HoL{$a}} } keys %HoL ) {

     print "$family: @{ $HoL{$family} }\n"

 }

 # print the whole thing sorted by number of members and name

 foreach $family ( sort {

                            @{$HoL{$b}} <=> @{$HoL{$a}}

                                        ||

                                    $a cmp $b

            } keys %HoL )

 {

     print "$family: ", join(", ", sort @{ $HoL{$family} }), "\n";

 }



LISTS OF HASHES



Declaration of a LIST OF HASHES

 @LoH = (

        {

            Lead     => "fred",

            Friend   => "barney",

        },

        {

            Lead     => "george",

            Wife     => "jane",

            Son      => "elroy",

        },

        {

            Lead     => "homer",

            Wife     => "marge",

            Son      => "bart",

        }

  );



Generation of a LIST OF HASHES

 # reading from file

 # format: LEAD=fred FRIEND=barney

 while ( <> ) {

     $rec = {};

     for $field ( split ) {

         ($key, $value) = split /=/, $field;

         $rec->{$key} = $value;

     }

     push @LoH, $rec;

 }

 # reading from file

 # format: LEAD=fred FRIEND=barney

 # no temp

 while ( <> ) {

     push @LoH, { split /[\s+=]/ };

 }

 # calling a function  that returns a key,value list, like

 # "lead","fred","daughter","pebbles"

 while ( %fields = getnextpairset() ) {

     push @LoH, { %fields };

 }

 # likewise, but using no temp vars

 while (<>) {

     push @LoH, { parsepairs($_) };

 }

 # add key/value to an element

 $LoH[0]{pet} = "dino";

 $LoH[2]{pet} = "santa's little helper";



Access and Printing of a LIST OF HASHES

 # one element

 $LoH[0]{lead} = "fred";

 # another element

 $LoH[1]{lead} =~ s/(\w)/\u$1/;

 # print the whole thing with refs

 for $href ( @LoH ) {

     print "{ ";

     for $role ( keys %$href ) {

         print "$role=$href->{$role} ";

     }

     print "}\n";

 }

 # print the whole thing with indices

 for $i ( 0 .. $#LoH ) {

     print "$i is { ";

     for $role ( keys %{ $LoH[$i] } ) {

         print "$role=$LoH[$i]{$role} ";

     }

     print "}\n";

 }

 # print the whole thing one at a time

 for $i ( 0 .. $#LoH ) {

     for $role ( keys %{ $LoH[$i] } ) {

         print "elt $i $role is $LoH[$i]{$role}\n";

     }

 }



HASHES OF HASHES



Declaration of a HASH OF HASHES

 %HoH = (

        flintstones => {

                lead      => "fred",

                pal       => "barney",

        },

        jetsons     => {

                lead      => "george",

                wife      => "jane",

                "his boy" => "elroy",

        },

        simpsons    => {

                lead      => "homer",

                wife      => "marge",

                kid       => "bart",

        },

 );



Generation of a HASH OF HASHES

 # reading from file

 # flintstones: lead=fred pal=barney wife=wilma pet=dino

 while ( <> ) {

     next unless s/^(.*?):\s*//;

     $who = $1;

     for $field ( split ) {

         ($key, $value) = split /=/, $field;

         $HoH{$who}{$key} = $value;

     }

 # reading from file; more temps

 while ( <> ) {

     next unless s/^(.*?):\s*//;

     $who = $1;

     $rec = {};

     $HoH{$who} = $rec;

     for $field ( split ) {

         ($key, $value) = split /=/, $field;

         $rec->{$key} = $value;

     }

 }

 # calling a function  that returns a key,value hash

 for $group ( "simpsons", "jetsons", "flintstones" ) {

     $HoH{$group} = { get_family($group) };

 }

 # likewise, but using temps

 for $group ( "simpsons", "jetsons", "flintstones" ) {

     %members = get_family($group);

     $HoH{$group} = { %members };

 }

 # append new members to an existing family

 %new_folks = (

     wife => "wilma",

     pet  => "dino",

 );

 for $what (keys %new_folks) {

     $HoH{flintstones}{$what} = $new_folks{$what};

 }



Access and Printing of a HASH OF HASHES

 # one element

 $HoH{flintstones}{wife} = "wilma";

 # another element

 $HoH{simpsons}{lead} =~ s/(\w)/\u$1/;

 # print the whole thing

 foreach $family ( keys %HoH ) {

     print "$family: { ";

     for $role ( keys %{ $HoH{$family} } ) {

         print "$role=$HoH{$family}{$role} ";

     }

     print "}\n";

 }

 # print the whole thing  somewhat sorted

 foreach $family ( sort keys %HoH ) {

     print "$family: { ";

     for $role ( sort keys %{ $HoH{$family} } ) {

         print "$role=$HoH{$family}{$role} ";

     }

     print "}\n";

 }

 # print the whole thing sorted by number of members

 foreach $family ( sort { keys %{$HoH{$b}} <=> keys %{$HoH{$a}} } keys %HoH ) {

     print "$family: { ";

     for $role ( sort keys %{ $HoH{$family} } ) {

         print "$role=$HoH{$family}{$role} ";

     }

     print "}\n";

 }

 # establish a sort order (rank) for each role

 $i = 0;

 for ( qw(lead wife son daughter pal pet) ) { $rank{$_} = ++$i }

 # now print the whole thing sorted by number of members

 foreach $family ( sort { keys %{ $HoH{$b} } <=> keys %{ $HoH{$a} } } keys %HoH ) {

     print "$family: { ";

     # and print these according to rank order

     for $role ( sort { $rank{$a} <=> $rank{$b} }  keys %{ $HoH{$family} } ) {

         print "$role=$HoH{$family}{$role} ";

     }

     print "}\n";

 }



MORE ELABORATE RECORDS



Declaration of MORE ELABORATE RECORDS

Here's a sample showing how to create and use a record whose fields are of many different sorts: 

     $rec = {

         TEXT      => $string,

         SEQUENCE  => [ @old_values ],

         LOOKUP    => { %some_table },

         THATCODE  => \&some_function,

         THISCODE  => sub { $_[0] ** $_[1] },

         HANDLE    => \*STDOUT,

     };

     print $rec->{TEXT};

     print $rec->{LIST}[0];

     $last = pop @ { $rec->{SEQUENCE} };

     print $rec->{LOOKUP}{"key"};

     ($first_k, $first_v) = each %{ $rec->{LOOKUP} };

     $answer = $rec->{THATCODE}->($arg);

     $answer = $rec->{THISCODE}->($arg1, $arg2);

     # careful of extra block braces on fh ref

     print { $rec->{HANDLE} } "a string\n";

     use FileHandle;

     $rec->{HANDLE}->autoflush(1);

     $rec->{HANDLE}->print(" a string\n");



Declaration of a HASH OF COMPLEX RECORDS

     %TV = (

        flintstones => {

            series   => "flintstones",

            nights   => [ qw(monday thursday friday) ],

            members  => [

                { name => "fred",    role => "lead", age  => 36, },

                { name => "wilma",   role => "wife", age  => 31, },

                { name => "pebbles", role => "kid",  age  =>  4, },

            ],

        },

        jetsons     => {

            series   => "jetsons",

            nights   => [ qw(wednesday saturday) ],

            members  => [

                { name => "george",  role => "lead", age  => 41, },

                { name => "jane",    role => "wife", age  => 39, },

                { name => "elroy",   role => "kid",  age  =>  9, },

            ],

         },

        simpsons    => {

            series   => "simpsons",

            nights   => [ qw(monday) ],

            members  => [

                { name => "homer", role => "lead", age  => 34, },

                { name => "marge", role => "wife", age => 37, },

                { name => "bart",  role => "kid",  age  =>  11, },

            ],

         },

      );



Generation of a HASH OF COMPLEX RECORDS

     # reading from file

     # this is most easily done by having the file itself be

     # in the raw data format as shown above.  perl is happy

     # to parse complex data structures if declared as data, so

     # sometimes it's easiest to do that

     # here's a piece by piece build up

     $rec = {};

     $rec->{series} = "flintstones";

     $rec->{nights} = [ find_days() ];

     @members = ();

     # assume this file in field=value syntax

     while (<>) {

         %fields = split /[\s=]+/;

         push @members, { %fields };

     }

     $rec->{members} = [ @members ];

     # now remember the whole thing

     $TV{ $rec->{series} } = $rec;

     ###########################################################

     # now, you might want to make interesting extra fields that

     # include pointers back into the same data structure so if

     # change one piece, it changes everywhere, like for examples

     # if you wanted a {kids} field that was an array reference

     # to a list of the kids' records without having duplicate

     # records and thus update problems.

     ###########################################################

     foreach $family (keys %TV) {

         $rec = $TV{$family}; # temp pointer

         @kids = ();

         for $person ( @{ $rec->{members} } ) {

             if ($person->{role} =~ /kid|son|daughter/) {

                 push @kids, $person;

             }

         }

         # REMEMBER: $rec and $TV{$family} point to same data!!

         $rec->{kids} = [ @kids ];

     }

     # you copied the list, but the list itself contains pointers

     # to uncopied objects. this means that if you make bart get

     # older via

     $TV{simpsons}{kids}[0]{age}++;

     # then this would also change in

     print $TV{simpsons}{members}[2]{age};

     # because $TV{simpsons}{kids}[0] and $TV{simpsons}{members}[2]

     # both point to the same underlying anonymous hash table

     # print the whole thing

     foreach $family ( keys %TV ) {

         print "the $family";

         print " is on during @{ $TV{$family}{nights} }\n";

         print "its members are:\n";

         for $who ( @{ $TV{$family}{members} } ) {

             print " $who->{name} ($who->{role}), age $who->{age}\n";

         }

         print "it turns out that $TV{$family}{lead} has ";

         print scalar ( @{ $TV{$family}{kids} } ), " kids named ";

         print join (", ", map { $_->{name} } @{ $TV{$family}{kids} } );

         print "\n";

     }



Database Ties

You cannot easily tie a multilevel data structure (such as a hash of hashes) to a dbm file. The first problem is that all but GDBM and Berkeley DB have size limitations, but beyond that, you also have problems with how references are to be represented on disk. One experimental module that does partially attempt to address this need is the MLDBM module. Check your nearest CPAN site as described in the perlmodlib manpage for source code to MLDBM. 



SEE ALSO

perlref(1), perllol(1), perldata(1), perlobj(1) 

